
Systems Research Center

DEC's business and technology objectives require a strong research program. The Systems
Research Center (SRC) and three other research laboratories are committed to filling that need.

SRC began recruiting its first research scientists in l984 – their charter, to advance the state of
knowledge in all aspects of computer systems research. Our current work includes exploring
high-performance personal computing, distributed computing, programming environments,
system modelling techniques, specification technology, and tightly-coupled multiprocessors.

Our approach to both hardware and software research is to create and use real systems so that we
can investigate their properties fully. Complex systems cannot be evaluated solely in the abstract.
Based on this belief, our strategy is to demonstrate the technical and practical feasibility of our
ideas by building prototypes and using them as daily tools. The experience we gain is useful in
the short term in enabling us to refine our designs, and invaluable in the long term in helping us
to advance the state of knowledge about those systems. Most of the major advances in
information systems have come through this strategy, including time-sharing, the ArpaNet, and
distributed personal computing.

SRC also performs work of a more mathematical flavor which complements our systems
research. Some of this work is in established fields of theoretical computer science, such as the
analysis of algorithms, computational geometry, and logics of programming. The rest of this
work explores new ground motivated by problems that arise in our systems research.

DEC has a strong commitment to communicating the results and experience gained through
pursuing these activities. The Company values the improved understanding that comes with
exposing and testing our ideas within the research community. SRC will therefore report results
in conferences, in professional journals, and in our research report series. We will seek users for
our prototype systems among those with whom we have common research interests, and we will
encourage collaboration with university researchers.

Robert W. Taylor, Director

Authors’ abstract

This is a reference manual for Trestle, a Modula-3 toolkit for the X window system.
Trestle is a collection of interfaces structured around a central abstract type: a “virtual
bitmap terminal” or VBT, which represents a share of the workstation’s screen,
keyboard, and mouse—a thing comparable to the viewers, windows, or widgets of other
systems.

Trestle is included in SRC Modula-3 version 2.0, which is available via public ftp.

Trestle includes a fairly standard set of interactors, including menus, buttons, “container”
classes that provide overlapping or tiled subwindows, and “leaf” windows that display
text or other data. This reference manual also specifies the interfaces that allow you to
create your own window classes. Knowledge of X is not required.

A Trestle window is an object whose behavior is determined by its methods. For
example, a window’s response to a mouse click is determined by calling its mouse
method. This is fast becoming the standard architecture for toolkits, but Trestle carries it
further than most. For example, you can change the way a Trestle window paints by
overriding its paint method; this is useful for sophisticated effects like groupware.

Trestle provides a novel strategy for writing applications that are independent of the type
of display screen they are running on. For example, it is easy to write a Trestle
application that can be moved back and forth between a color display and a monochrome
display where the application will look good on both.

1 Introduction

This report is a programmer’s reference manual for Trestle, a Modula-3 window system
toolkit.

Trestle has been implemented over two underlying window systems: X [5] and
the native Firefly window system developed at SRC [6]. Other implementations are
possible, but at present the only widely available implementation is Trestle-on-X.

To use Trestle-on-X, you need a Modula-3 compiler and an X server for your
system. The Trestle code is an application library layered on top of Xlib, the standard
X client library. Trestle applications obey X’s ICCCM protocol for cooperating with
the window manager and other applications, so you can use your favorite window
manager and mix Trestle applications freely with other X applications.

The reference manual is self-contained but non-tutorial; you would do well to read
theTrestle Tutorialfirst (Chapter 7 of [4]). We assume you are familiar with Modula-3
[1, 2, 4].

The Trestle abstraction. A Trestle.T is a connection to a window system. The
window system is assumed to have a keyboard, a pointing device, and one or more
display screens. For example, in Trestle-on-X, aTrestle.T is implemented by a
connection to an X server.

Each screen is a raster display, whose image is stored in a frame buffer containing a
rectangular array of pixels. Changing the contents of a frame buffer is calledpainting,
since it changes the image displayed to the user. The different screens can be of
different types (e.g., color or black and white).

Trestle imposes anhv -coordinate system on each display screen, in which the
h coordinate increases from left to right and thev coordinate increases from top to
bottom. TheTrestle interface allows you to determine the number of screens and
also their types, dimensions, and the positions of their coordinate origins.

We will call the pointing device the mouse, although it might be a stylus or other
instrument. The mouse generally has one or more buttons that the user can click down
and up.

The system displays acursor, a small arrow or other image that points at some
pixel of some screen. By moving the mouse the user can move the cursor around the
screen or from one screen to another. Applications can change the shape of the cursor
to convey information to the user.

The completeTrestle interface is described in Section 3.

The VBT abstraction. The key abstraction in Trestle is the “virtual bitmap terminal”
or VBT. A VBT represents a share of the keyboard, mouse, and displays.VBTs are
comparable to the windows, widgets, and viewers of other systems.

An application is generally organized as a tree ofVBTs, with the rootVBT
representing the top-level application window. The internal nodes are calledsplitVBTs
or parentVBTs: they divide their screens between one or more childVBTs according
to some layout depending on the class of split. At the leaves of the tree areVBTs that
contain no subwindows.

A typical application consists of a number of leafVBTs whose behavior is specific
to that application, together with some more leafVBTs that provide buttons, scrollbars,
and other “interactors”, all held together by a tree of splits that define the geometric
layout. A split with only one child is called afilter. For example, aBorderedVBT is
a filter that adds a border around the child’s screen. A split that can have more than
one child is called aproper split. For example, anHVSplit is a split in which the
children are laid out horizontally or vertically.

Sections 4, 5, and 6 describe Trestle’s built-in proper splits, filters, and leaves.
To obtain a share of a Trestle display, an application creates aVBTand “installs” it

with the procedureTrestle.Install ,which allocates some portion of some display
to theVBT, within which the application can paint. TheVBT is said to beinstalledand
is called atop-level window. The size and position of top-level windows depends on
the arguments toTrestle.Install and on the whim of the window manager.

A VBTimposes anhv -coordinatesystem on its screen. A top-levelVBT’s coordinate
system need not be the same as the coordinate system of the screen on which it is
installed. The translation between the two coordinate systems can be determined
through theTrestle.ScreenOf procedure.

In a split VBT, the translation between the parent and child coordinate systems
depends on the class of the split. Trestle provides one filter (TranslateVBT ) whose
sole purpose is to position the child coordinate system origin at the northwest corner of
the child’s domain, since this is convenient for some applications. All the other built-in
splits make the child coordinate system agree with the parent coordinate system, since
this is usually the most convenient.

Information flows through aVBT in two directions. Painting commands travel
from the leaves of the tree towards the root. Events like mouse clicks and cursor
positions travel from the root towards the leaves. AVBT is an object with methods
for handling events; to deliver an event to aVBT, the system invokes the appropriate
method. TheVBT interface in Section 2 specifies the event-handling methods and the
painting procedures.

The screen of aVBTis forgetful; that is, its contents can be lost at any time, at which
point the system activates itsrepaint method, which is expected to repaint what has
been lost. Similarly, the height, width, and coordinate origin of aVBT’s screen can
change at any time, in which case the system activates itsreshape method. Finally,
the type of the pixels in aVBT’s screen can change (e.g., from color to monochrome),
in which case the system activates itsrescreen method. These events reflect the fact
that the user of the window system can expose portions of a top-level window, reshape
top-level windows, and move top-level windows from one display to another.

Selections and event-time. From the user’s point of view, a selection is a highlighted
occurrence of text or other data that can be made in a window via some gesture, such
as sweeping with the mouse. Selections are supported to make it easy for users to cut
and paste text and other data between windows. A particular selection is always in at
most one window at a time, namely the “owner” of the selection. If a selection is in no
window at all, its owner isNIL .

From the programmer’s point of view, the selection owner is aVBT-valued variable
shared between all applications. The procedureVBT.Acquire is used to acquire a
selection. Whenever aVBTacquires a selection, the previous owner is notified, so that
it can take down any highlighting or other feedback. AnyVBT can own a selection,
not just a top-level window.

The proceduresVBT.Read andVBT.Write are used to read or write the value
of the selection. Calls toRead andWrite are implemented by locating the selection
owner (which could be in the same address space as the caller toRead or Write ,
or in a different address space) and activating itsread or write method, which is
responsible for doing the work. The selection values communicated byRead and
Write can be of any type that can be pickled (see Section 3.6 ofSystems Programming
with Modula-3[4]); in particular, they can be of typeTEXT.

TheVBT to which user keystrokes are directed is called thekeyboard focus. Some
window managers define the focus to be the window containing the cursor; other
window managers move the focus in response to mouse clicks. Trestle applications
work with either kind of window manager.

Trestle classifies the keyboard focus as a selection, since it is a globalVBT-valued
variable that can be acquired and released. If you want to receive keystrokes, you must
acquire the focus. If this succeeds, you should provide some feedback to the user, for
example by displaying a blinking caret. (Even if the window manager is identifying
the top-level window containing the focus, you should still let the user know which
subwindow contains the focus.) When you are notified that you have lost the focus,
you should take down the feedback.

It is also possible to send any selection owner a “miscellaneous code”, which
will be delivered by activating themisc method in the owner. For example, the
way that Trestle notifies a window that it no longer owns a selection is by sending
it a miscellaneous code of typeLost . Miscellaneous codes are also used for other
purposes; for example, to notify windows that they have been deleted.

The event-time protocol. There are many potential race conditions involving selec-
tions. For example, suppose that the user clicks in windowA, expecting it to acquire
the keyboard focus. But windowA is slow—perhaps it is paging or blocked in a call to
a server that is being debugged—and does not respond. So the user clicks in another
window B, which acquires the keyboard focus, and types away. A few minutes later,
windowA comes to life and grabs the keyboard focus. Suddenly and unexpectedly the
user’s typing is redirected toA instead ofB. Similar race conditions can occur with
selections other than the keyboard focus—for example, you select a file name, then
activate adelete command by clicking, then wonder how long you must wait before
it is safe to make another selection.

Trestle uses theevent-time protocolto deal with these race conditions. This means
that Trestle keeps track of thecurrent event time, which is the timestamp of the last
keystroke, mouseclick, or miscellaneous code that has been delivered to anyVBT.
Attempts to read, write, or acquire a selection must be accompanied by a timestamp,
and if this timestamp does not agree with the current event time, the attempt fails. This

guarantees that onlyVBTs that are responding to the user’s latest action canaccess the
selections.

When Trestle activates a window’s method to deliver it an event, it generally
waits for the method to return before it delivers any events to any other windows.
This gives the window a fair chance to use the time stamp in the event toaccess the
selections. However, if the method takes an unreasonably long time—more than a
few seconds—Trestle may give up on the window and start delivering events to other
windows anyway.

As a consequence, if you must do a long-running computation in response to a user
event, then you should fork the computation in a separate thread and return from the
method promptly, to avoid delaying the user, who may want to click in another window.
You should also do any operations that requireaccessing the selections from the main
thread before the method returns, since an event-time operation in the forked thread
will fail if the user has continued typing or clicking during the forked computation.

The geometry interfaces. The interfacesAxis , Point , Rect , Region , Trape-
zoid , andPath are explained in Section 8. In brief,Axis.T.Hor andAxis.T.Ver
name the horizontal and vertical coordinate axes; aPoint.T (or simply apoint) is a
pair of integers representing a point in the plane; aRect.T is a rectangle of points
whose sides are parallel to the coordinate axes; aRegion.T is an arbitrary set of
points represented as a sorted array of rectangles; aTrapezoid.T is a set of points
bounded by two horizontal lines and two lines with arbitrary slopes; and aPath.T is
a path in the plane represented by a sequence of straight and curved segments.

Resources. A pixmapis a rectangular array of pixels. Abitmapis a pixmap in which
the pixels are one bit deep. For example, a large pixmap could represent a photographic
image; a small bitmap could represent a cursor shape. Trestle also uses a pixmap to
represent the infinite texture that results from tiling the plane with translations of the
pixmap. Thus whether a pixmap represents an infinite texture or a bounded image
depends only on the context in which it is used.

A font is a typeface suitable for painting text.
A painting operationis an operation code for changing the values of pixels in the

frame buffer of a display screen.
Pixmaps, cursor shapes, fonts, and painting operations are collectively called

resources. Resources come in bothscreen-independentandscreen-dependentforms.
A screen-independent resource varies with the screentype to produce a similar effect on
all types of screens. For example, two important screen-independent painting operations
arePaintOp.Fg andPaintOp.Bg , which set pixels to a screen’s foreground and
background colors. In contrast, a screen-dependent resource is useful only on a
particular screentype. If it is used on aVBTwith the wrong type of screen, the system
won’t crash, but the effect will be non-deterministic—a screen-dependent painting
operation that blackens a pixel on a black-and-white screen might set a 24-bit pixel to
chartreuse on a true-color screen.

Screen-independent resources are convenient, but screen-dependent resources are
sometimes necessary for exploiting the capabilities of specific display hardware.

The screen-independent resource types are calledPixmap.T ,Cursor.T ,Font.T ,
and PaintOp.T . The interfaces where these types are defined also provide proce-
dures for generating useful resources. For example,PaintOp.FromRGB will pro-
duce a screen-independent painting operation that sets a pixel to a particular color;
Font.FromName will produce a screen-independent font given the name of the
typeface.

The corresponding screen-dependent resources areScrnPixmap.T , ScrnCur-
sor.T , ScrnFont.T , andScrnPaintOp.T . The interfaces where these types are
defined also specify the representations of the raw values—the layout of pixmaps in
memory, the attributes of fonts, and similar details that all sane people prefer to avoid.

Converting a screen-independent resource into the corresponding screen-dependent
resource for a particular type of screen is calledresolvingthe resource. ThePalette
interface will give you a screen-independent resource if you give it a closure for
resolving the resource. You can therefore use thePalette interface to construct your
own screen-independent resources. For example, you could produce a (PaintOp.T ,
Font.T ) pair that produces red Times Roman text on a color display and black italic
text on a black-and-white display; or aPixmap.T that selects between a low and a
high resolution bitmap depending on the screen resolution.

The closure for resolving the resource will be invoked automatically when a top-
level window moves to a new screentype for the first time. The closure will be passed
an argument of typeScreenType.T , which represents a type of display screen. A
ScreenType.T determines the depth of the screen’s pixels (e.g., one or eight), the
method for associating a color with a pixel value (e.g., color-mapped or true-color), the
set of allowed operations on its pixels, and the repositories for screen-dependent fonts,
cursors, and pixmaps that can be used on the screen.

Implementing your own splits. Most applications can be built by using Trestle’s built-
in splits and leaves, together with one or more leafVBTs specific to the application. If
you are programming a more sophisticated application, you may want to augment the
built-in splits with some of your own. Section 9 introduces the interfaces that allow
you to do this.

To implement a leafVBT, you only have to supply methods to handle the events
that flow down the tree (from the root to the leaves). To implement a splitVBT, you
also have to supply methods to handle the information that flows up the tree, such
as painting commands or commands to change the cursor shape. TheVBTClass
interface declares these methods and presents their specifications.

Very few splits override the method for painting, since the default behavior, which
is to clip to the child’s domain and relay the painting to the parent, is usually what is
desired. But some splits do override this method: for example, theZSplit , whose
child windows are allowed to overlap one another, has a paint method that clips its
children’s painting to the visible parts of their domains. And the top level window has a
painting method that translatesVBTpainting commands into X painting commands and
relays them to the X server. The interfacesBatch , BatchUtil , andPaintPrivate
reveal the details necessary to override painting methods.



  

6 1 INTRODUCTION

The remainder of the reference manual consists of complete Modula-3 interfaces
printed in typewriter font and interspersed with commentary printed in roman font.
Some of the commentary is in the form of “pseudo-Modula-3” program fragments,
which are also printed in typewriter font.

The Trestle release that accompanies SRC Modula-3 version 2.0 contains several
interfaces that are not documented in this reference manual. For example, the
VTextVBT interface provides editable textVBTs and theTrestleAux interface
allows you to set window manager parameters and do strange things to top-level
windows. The specifications for these interfaces are directly in the Modula-3 interface
files.



   

7

2 The VBT interface

A VBT.T (or simply aVBT) is the basic window abstraction of the Trestle system.

INTERFACE VBT;

IMPORT Word, Axis, Point, Rect, Region, Trapezoid,
Path, Pixmap, Cursor, Font, PaintOp, ScrnPixmap;

2.1 The public methods

A VBT is represented as an object with a private prefix and twelve public methods,
which define the way theVBT responds to events. Here are the type declarations that
reveal the public methods, while concealing the private prefix:

TYPE
T <: Public;
Public = Prefix OBJECT

METHODS
<* LL.sup = mu * >
mouse(READONLY cd: MouseRec);
position(READONLY cd: PositionRec);
redisplay();
misc(READONLY cd: MiscRec);
key(READONLY cd: KeyRec);
discard();
<* LL.sup = mu.SELF * >
reshape(READONLY cd: ReshapeRec);
rescreen(READONLY cd: RescreenRec);
repaint(READONLY rgn: Region.T);
shape(ax: Axis.T; n: CARDINAL): SizeRange;
<* LL.sup <= mu * >
read(sel: Selection; tc: CARDINAL): Value

RAISES {Error};
write(sel: Selection; val: Value; tc: CARDINAL)

RAISES {Error};
END;

Prefix <: ROOT;

For example, if the user reshapes a window, Trestle will call the window’s reshape
method; if the user exposes some part of the window, Trestle will call the window’s
repaint method. The remainder of theVBT interface specifies the methods in detail.
The pragmas aboutLL are explained in the section on locking level, below.

You should never call aVBT’s methods directly. TheVBTClass interface provides
wrapper procedures that call the methods indirectly.

2.2 Screens and domains

Every VBT has ascreenthat associates a pixel value with each integer lattice point.
We writev[p] to denote the value of the pixel at pointp of the screen of theVBT v.
Changing the pixel values in aVBT’s screen is calledpainting.

The part of aVBT’s screen that is visible to the user—or that would be visible if
other windows weren’t in the way—is called thedomainof theVBT:

PROCEDURE Domain(v: T): Rect.T; < * LL.sup < v * >

Return the rectangular extent of the visible part ofv ’s screen.

The domain is an arbitrary rectangle: it can be empty, the coordinate origin can be
anywhere inside or outside it, and it does not necessarily correspond to the position of
the window on the physical display screen.

Whenv is reshaped,Domain(v) changes from one rectangle to another. During
this transformation Trestle tries to save the old screen until the new screen is fully
repainted: thus in the midst of reshaping,v[p] can be useful for some pointsp outside
Domain(v) . At other times, Trestle keeps track ofv[p] only for pointsp inside
Domain(v) .

The pragmaLL.sup < v is explained in the next section.

2.3 Locking level

The global mutexmuserializes operations that affect the tree ofVBTs:

VAR mu: MUTEX;

In addition, everyVBT includes a private mutex that serializes operations on theVBT
itself. The private mutex of aVBT is revealed in theVBTClass interface, not in this
interface.

The order in which a thread is allowed to acquire these locks is called the “locking
order”. It is defined by these two rules:

� The globalmuprecedes everyVBT.

� EveryVBTprecedes its parent.

The “locking level” of a thread, orLL for short, is the set of locks that the thread
has acquired. The expressionLL.sup denotes the maximum of the locks inLL . (The
locking order is partial, butLL.sup will be defined for any thread in a correct program,
since threads acquire locks in ascending order.)

Each procedure declaration in the Trestle system includes a pragma specifying the
locking level at which a thread can legally call the procedure. For example, the pragma
LL.sup < v on theDomain procedure allows a thread to callDomain with no locks,
or with mu locked, or with descendants ofv locked, but forbids calling it with any
otherVBTs locked.

Similarly, eachpublic data field and method of an object has a locking level. In
both cases, a locking level pragma applies to all the fields or methods between it and
the next pragma. These pragmas may contain the special identifierSELF, which refers
to the object itself.

The locking level for a method is identical to the locking level for a procedure: it
specifies the locking level at which a thread can legally call the method. For example,
whenever themouse, position , redisplay , misc , key , or discard methods of
a VBTare called, the locking level satisfiesLL.sup = mu .

The locking level for a writable data field is of the form

LL >= {mu1, ..., muN}.

This specifies that in order to write the field, a thread must hold all of the locksmu1
throughmuN. As a consequence, a thread can read the field if it holds any of the locks.

(In a locking level pragma, the ordering symbols>=, <=, <, and> are overloaded
to denote either set containment or lock order, depending on context. For example,LL
>= {mu, v} indicates that the thread has bothmuandv locked, whileLL.sup <=
mu indicates that all locks held by the thread precedemuin the locking order.)

A data field may also be commentedCONST, meaning that it is readonly after
initialization and therefore can be read with no locks at all.

There is one more special notation related to locking levels: aVBT v can hold a
“share” of the global lockmu; its share is denoted bymu.v . This is explained in the
section of this interface that specifies thereshape method.

All the procedures in the Trestle system restore the caller’s locking level when they
return. For example, callingDomain(v) has no net effect on a thread’s locking level.

2.4 ScreenTypes

Pixel values are integers. The color associated with a pixel value is determined
in some manner that depends on thescreentypeof the VBT. A value st of type
VBT.ScreenType represents a screentype:

TYPE
ScreenType <: ScreenTypePublic;
ScreenTypePublic = OBJECT ( * CONST* )

depth: INTEGER;
color: BOOLEAN;
res: ARRAY Axis.T OF REAL

END;

The integerst.depth is the number of bits per pixel in screens of typest . The
booleanst.color is TRUEif the pixels are colored,FALSE if they are black and
white or gray-scale. The arrayst.res gives the horizontal and vertical resolution of
the screen in pixels per millimeter for desk-top displays, or in visually equivalent units
for other displays.

The screentype of a newly-allocatedVBT is NIL ; it becomes non-NIL only when
theVBT is connected to a window system.

Here are two procedures for reading the screentype of aVBT and for converting
distances to screen coordinates:

PROCEDURE ScreenTypeOf(v: T): ScreenType;
<* LL.sup < v * >

Return the screentype ofv .

PROCEDURE MMToPixels(v: T; mm: REAL; ax: Axis.T)
: REAL; < * LL.sup < v * >

Return the number of pixels that correspond tommmillimeters onv ’s screentype
in the axisax ; or return0 if v ’s screentype isNIL .

TheScreenType interface reveals more details, for example, about color maps.

2.5 Splits and leaves

User interfaces are usually constructed from a tree ofVBTs whose root is the “top-level
window” known to the window manager.VBTs are classified into two main subtypes
based on their positions in the tree:

TYPE
Split <: T;
Leaf <: T;

PROCEDURE Parent(v: T): Split; < * LL.sup < v * >

Returnv ’s parent, orNIL if v has no parent.

A Split (also called a parentVBT) divides its screen up among its children according
to some layout policy that depends on the class of split. Each pixel of the parent
screen represents a pixel of one of the childVBTs, which is said to control that pixel.
For example, overlapping windows are provided by a class of split called aZSplit ,
for which the children are ordered bottom to top, and each pixelv[p] of the parent
domain is controlled by the top-most child whose domain includesp.

See theSplit interface for common operations on splits (e.g., enumerating
children).

A Leaf is a VBT in which the twelve public methods make theLeaf ignore all
events, be indifferent about its shape, and do nothing when discarded. It is provided as
a starting point: you can define a useful subtype ofLeaf by overriding the methods
that are relevant to the new class.

Almost all subtypes ofVBTare subtypes of eitherSplit or Leaf .

2.6 Timestamps, modifiers, mouse buttons, and cursor positions

The following types are used in several of the event methods:

TYPE
TimeStamp = Word.T;

Modifier =
{Shift, Lock, Control, Option,

Mod0, Mod1, Mod2, Mod3,
MouseL, MouseM, MouseR,
Mouse0, Mouse1, Mouse2, Mouse3, Mouse4};

Button = [Modifier.MouseL..Modifier.Mouse4];

Modifiers = SET OF Modifier;

ScreenID = INTEGER;

CursorPosition = RECORD
pt: Point.T;
screen: ScreenID;
gone, offScreen: BOOLEAN;

END;

CONST
Buttons = Modifiers{FIRST(Button)..LAST(Button)};

Trestle has an internal unsigned clock register that is incremented every few millisec-
onds. When Trestle reports a mouse or keyboard event to aVBT, it also reports the
value of the clock register when the event occurred, which is called thetimestampof
the event. Timestamps serve as unique identifiers for the associated events. Also,
the absolute time interval between two events can be computed by subtracting their
timestamps withWord.Minus and multiplying byTrestle.TickTime() , which
is the absolute interval between clock ticks.

A few keys on the keyboard are defined to bemodifiers, like Shift , Control ,
andOption . When Trestle reports a mouse or keyboard event to aVBT, it also reports
the set of modifier keys and buttons that were down when the event occurred. Thus
the application can distinguish shifted mouse clicks from unshifted mouse clicks, for
example.

The modifierShift is reported if either of the keyboard’s shift keys is down;
similarly for Control andOption . The modifierLock is reported if the lock key
is locked down. If the keyboard has a key labelledlock but this key does not have
mechanical alternate action, then the modifierLock reflects the simulated state of the
lock key (that is, alternate presses of the lock key turn the modifier on or off). Trestle
does not define whether it reports up and down transitions for lock keys while the
modifier is set.

Some Trestle servers interpret other keys as modifiers: the type definitionaccom-
modates up to four additional modifiers,Mod0 throughMod3.

The mouse buttons are reported as modifiers. The naming of the first three buttons
assumes a three-button mouse; in general it is assumed that there are at most eight
buttons.

When Trestle reports a mouse position event to aVBT v, it also reports a valuecp
of typeCursorPosition . The pointcp.pt is the position of the cursor; the integer
cp.screen identifies the screen of the window system where the event occurred;
and cp.offScreen is TRUE if the position is on a different screen thanv , and
FALSE otherwise. Ifcp.offScreen is FALSE, then cp.pt is in v ’s coordinate
system, otherwisecp.pt is in the coordinate system ofcp.screen . The boolean
cp.gone is TRUEif v doesn’t control the positioncp.pt , andFALSE if it does. If
cp.offScreen is TRUE, then so iscp.gone . A position is controlled by aVBT
w if a mouse-click at that position would ordinarily be delivered tow. All positions
controlled by aVBTare in its domain; every pixel in the domain of a split is controlled
by at most one child of that split. You should think of the positions controlled by a
VBTas the visible positions in its domain.

2.7 The mouse method

Trestle calls aVBT’s mouse method to report mouse clicks. The method will be called
with LL.sup = mu , and takes an argument of typeMouseRec.

TYPE MouseRec = RECORD
whatChanged: Button;
time: TimeStamp;
cp: CursorPosition;
modifiers: Modifiers;
clickType: ClickType;
clickCount: INTEGER;

END;

ClickType =
{FirstDown, OtherDown, OtherUp, LastUp};

The method callv.mouse(cd) indicates that the mouse buttoncd.whatChanged
went down or up at timecd.time and cursor positioncd.cp .

The fieldcd.clickType is FirstDown if the button went down when no other
buttons were down,OtherDown if it went down when some other button(s) were
already down,LastUp if it went up when all other buttons were up, andOtherUp if
it went up when some other button(s) were still down.

The fieldcd.modifiers reflects the state of the modifiers (either just before or
just after the button transition; it is not specified which).

If cd.clickType is FirstDown , thencd.cp.gone will be FALSE.
The fieldcd.clickCount is the number of preceding transitions of the button

that were near in time and space. For example,clickCount=3 on the final up
transition of a double click. Some Trestle implementations have auxilliary interfaces
that allow you to set the amount of time and mouse motion allowed.

2.8 The mouse focus rule

A split relays mouse clicks to whichever child of the split controls the pixel at the

position of the click—more or less. If this rule were applied blindly, a child could
receive a down-click and never receive the corresponding up-click, which would make
it impossible to program many user interfaces that involve dragging. Therefore the
actual rule is more complicated.

Each splitsp contains a variablemouseFocus(sp) , which records the child of
the split that has received a transition of type FirstDown but not yet received a
subsequent transition of typeLastUp . If there is no such child,mouseFocus(sp) is
NIL . The splitsp relays theMouseRec cd by the “mouse focus rule”:

IF some childch controlscd.cp THEN

w := ch;

w.mouse(cd)

ELSE

w := NIL

END;

IF cd.clickType = ClickType.FirstDown THEN

mouseFocus(sp) := w

ELSE

IF mouseFocus(sp) # NIL AND mouseFocus(sp) # w THEN

cd.cp.gone := TRUE;

mouseFocus(sp).mouse(cd)

END;

IF cd.clickType = ClickType.LastUp THEN

mouseFocus(sp) := NIL

END

END

The mouse focus is guaranteed to receive all button transitions until the last button
comes up, no matter where it occurs.

2.9 The position method

Trestle calls aVBT’s position method to report cursor positions. The method will be
called withLL.sup = mu , and takes an argument of typePositionRec .

TYPE PositionRec = RECORD
cp: CursorPosition;
time: TimeStamp;
modifiers: Modifiers;

END;

The method callv.position(cd) indicates that at the timecd.time the cursor
positionwascd.cp and the set of modifiers keys that were down wascd.modifiers .

The next section explains how to control the delivery of cursor positions.

2.10 Tracking the cursor by setting cages

EveryVBT v contains a fieldcage(v) , which represents a set of cursor positions. As
long as the cursor’s position is insidev ’s cage, Trestle won’t report the position tov .
As soon as the cursor’s position moves outsidecage(v) , Trestle reports the position
to v , after first resettingv ’s cage to contain all cursor positions. Resetting the cage
inhibits further reporting of cursor positions: to continue tracking, the position method
must set a new cage.

TYPE
Cage = RECORD

rect: Rect.T;
inOut: InOut;
screen: ScreenID;

END;
InOut = SET OF BOOLEAN;

CONST
AllScreens: ScreenID = -1;

The cagecg contains the cursor positioncp if

� cp.pt is in cg.rect ,

� cp.gone is in cg.inOut , and

� eithercg.screen = AllScreens or cg.screen = cp.screen .

Trestle imposes the restriction on cages that ifcg.screen = AllScreens , then
cg.rect must beRect.Full or Rect.Empty , and if cg contains no cursor
positions, then it must be equal as a record toEmptyCage (which is declared below).
For example, here are some useful cages:

CONST
GoneCage =

Cage{Rect.Full, InOut{TRUE}, AllScreens};
InsideCage =

Cage{Rect.Full, InOut{FALSE}, AllScreens};
EverywhereCage =

Cage{Rect.Full, InOut{FALSE, TRUE}, AllScreens};
EmptyCage =

Cage{Rect.Empty, InOut{}, AllScreens};

GoneCage contains all cursor positions that are “gone”; set it on aVBT to wait for
the cursor to be over a position controlled by theVBT. The cageInsideCage is the
complement ofGoneCage: it contains all positions that theVBT controls. The cage
EverywhereCage contains all cursor positions, andEmptyCage contains none.

Here is the procedure for setting the cage of aVBT:

PROCEDURE SetCage(v: T; READONLY cg: Cage);
<* LL.sup < v * >

Setcage(v) to the intersection ofcage(v) with cg .

In the usual case,SetCage is called fromv ’s position method, at which pointv ’s cage
is EverywhereCage and therefore the intersection just comes out tocg . In unusual
cases, it will be found that intersecting the new cage with the old is what is required.

The procedureCageFromPosition is helpful for tracking the cursor continu-
ously. By settingCageFromPosition(cp) in response toeach cursor position cp ,
you can track the cursor as long as it moves within yourVBT. There are two additional
optional boolean arguments: settingtrackOutside allows you to track the cursor
over the whole screen containing theVBT; settingtrackOffScreen allows you to
track the cursor even onto other screens:

PROCEDURE CageFromPosition(
READONLY cp: CursorPosition;
trackOutside, trackOffScreen: BOOLEAN := FALSE)
: Cage; < * LL arbitrary * >

CageFromPosition(cp) returns the cage that contains only the position
cp ; or GoneCage if either cp.gone or cp.offScreen is TRUEand the
corresponding argument is not.

More precisely,CageFromPosition is equivalent to:

IF NOT cp.gone OR
trackOutside AND NOT cp.offScreen OR
trackOffScreen

THEN
RETURNthe cage containing only the positioncp

ELSIF cp.offScreen AND trackOutside THEN
RETURN Cage{Rect.Full, InOut{FALSE,TRUE}, cp.screen}

ELSE
RETURN GoneCage

END

Finally, the following two procedures are occasionally useful:

PROCEDURE Outside(
READONLY cp: CursorPosition; READONLY c: Cage)
: BOOLEAN; <* LL arbitrary * >

Return whether the positioncp is outside the cagecg .

PROCEDURE CageFromRect(READONLY r: Rect.T;
READONLY cp: CursorPosition): Cage; < * LL arbitrary * >

ReturnCage{r, InOut{cp.gone}, cp.screen} .

The effect ofSetCage(v, CageFromRect(r, cp)) is to suspend cursor positions
as long as the cursor stays inside the rectangler and has the same value ofgone ascp
does. This is useful when sweeping text selections, for example.

Splits relay cursor positions to their children. If several of the children are tracking
the cursor at the same time, the order in which positions are relayed to the different
children can be important. The order is determined by the following rule, which
specifies the way a splitsp forwards aPositionRec cd to its children (the variable
current(sp) is the child that controls the last cursor position seen bysp ):

IF some childch controlscd.cp THEN
w := ch

ELSE
w := NIL

END;
goneCd := cd;
goneCd.cp.gone := TRUE;
IF w # current(sp) THEN

Deliver(current(sp), goneCd)
END;
FOR all ch other thanwandcurrent(sp) DO

Deliver(ch, goneCd)
END;
IF w # NIL THEN Deliver(w, cd) END;
current(sp) := w

where

Deliver(v, cd) =
IF Outside(cd.cp, cage(v)) THEN

cage(v) := EverywhereCage;
v.position(cd)

END

A split maintains its cage to be a subset of the intersection of its children’s cages, so
that it will receive any cursor positions that it owes its children.

2.11 The key method

Trestle calls aVBT’s key method to report keystrokes. The method will be called with
LL.sup = mu , and takes an argument of typeKeyRec .

TYPE
KeyRec = RECORD

whatChanged: KeySym;
time: TimeStamp;
wentDown: BOOLEAN;



   

2.12 The redisplay method 17

modifiers: Modifiers;
END;

KeySym = INTEGER;

CONST
NoKey: KeySym = 0;

The method callv.key(cd) indicates that the keycd.whatChanged went up or
down at timecd.time . The booleancd.wentDown is true if the key went down;
false if it went up. The setcd.modifiers reflects the state of the modifiers (either
just before or just after the transition; it is not specified which).

A KeySym represents a symbol on a key of the keyboard. For example, there are
separateKeySyms for upper and lower case letters. The interfacesLatin1Key and
KeyboardKey specify theKeySym codes for many symbols that occur on standard
keyboards. These interfaces are shipped with SRC Trestle but are not included in the
printed version of the reference manual. The codes are chosen to agree with the X
Keysym codes (see X Window System, Scheifler et al., [5] Appendix E).

If the keyboard, like most keyboards, has two symbols on some of the keys, then
the KeySym for the down transition and later up transition might be different. For
example, if the user pushes the left shift key, then thez /Z key, and then releases the
keys in the same order, Trestle would report these four transitions:

left shift down, modifiers = {} or {Shift}
Z down, modifiers = {Shift}
left shift up, modifiers = {} or {Shift}
z up, modifiers = {}

Although the same physicalZ/z key went down and up, the down transition is reported
for theZ KeySymand the up transition is reported for thez KeySym.

The constantNoKey is simply an unusedKeySymcode.
To get Trestle to deliver keystrokes to aVBT, you make theVBT the owner of the

keyboard focus by calling the procedureVBT.Acquire .

2.12 The redisplay method

A typical VBT has a “display invariant” that defines what its screen looks like as a
function of its state. When the state changes, the display invariant is reestablished by
updating the screen.

When a series of changes are made, each of which invalidates the display invariant,
it is undesirable to update the screen after every change. For example, if the border
width and the border texture of aBorderedVBT both change, it is better not to paint
the intermediate state.

Therefore, Trestle keeps track of a set ofVBTs that have been “marked for
redisplay”. Procedures that invalidate aVBT’s display invariant mark theVBT instead
of updating thescreen directly. Trestleautomatically schedules acall to theredisplay
method of every marked window (unless the window’s screentype isNIL ). The method

takes no arguments: the callv.redisplay() must reestablishv ’s display invariant.
It will be called withLL.sup = mu .

The default redisplay method for aLeaf calls the reshape method with an empty
saved rectangle.

There are several procedures related to redisplay:

PROCEDURE Mark(v: T); < * LL.sup < v * >

Mark v for redisplay.

PROCEDURE IsMarked(v: T): BOOLEAN; < * LL.sup < v * >

ReturnTRUEif v is marked for redisplay.

PROCEDURE Unmark(v: T); < * LL.sup < v * >

If v is marked for redisplay, unmark it.

A marked window is automatically unmarked when it is redisplayed, reshaped, or
rescreened. Thus theUnmark procedure is rarely needed.

2.13 The reshape method

Trestle calls aVBT’s reshape method to report changes in its domain. The method will
be called withLL.sup = mu.v (as explained below), and takes an argument of type
ReshapeRec .

TYPE ReshapeRec = RECORD
new, prev, saved: Rect.T;
marked: BOOLEAN

END;

The method callv.reshape(cd) indicates that the domain ofv has changed from
cd.prev to cd.new . The rectanglecd.saved is the subset of the previous domain
that Trestle has preserved for the client in case it is of use in painting the new domain.
This is the only case in which Trestle tries to save portions of aVBT’s screen outside
its domain. After the reshape method returns, Trestle will generally forget the old parts
of the screen. The booleancd.marked indicates whetherv was marked when it was
reshaped; in any case,v is automatically unmarked as it is reshaped.

If new = Rect.Empty then the window is no longer visible (for example, this
happens when the window is iconized). Any background threads that are painting
should be stopped, since their efforts are useless.

The default reshape method for aLeaf calls therepaint method to repaint the
whole new domain.

When the reshape method is called,muis locked, and it will remain locked until the
method returns. However, Trestle may lockmuand then reshape, repaint, or rescreen
severalVBTs concurrently, so you can’t assume that an activation of your reshape
method excludes the activation of anotherVBT’s reshape, repaint, or rescreen method.

This locking level will be referred to asv ’s share ofmu, and writtenmu.v . Holding
mu is logically equivalent to holdingmu.v for everyv . Consequently,mu.v < mu
in the locking order. Holdingmu.v does not suffice to call a procedure that requires
muto be locked; on the other hand you cannot lockmuwhile holdingmu.v , since this
would deadlock.

2.14 The rescreen method

Trestle calls aVBT’s rescreen method to report changes to its screentype. The method
will be called withLL.sup = mu.v , and takes an argument of typeRescreenRec .

TYPE RescreenRec = RECORD
prev: Rect.T;
st: ScreenType;
marked: BOOLEAN;

END;

The method callv.rescreen(cd) indicates that the screentype ofv has changed to
cd.st and that its domain has changed fromcd.prev to Rect.Empty . (Typically
the VBT will be reshaped to a non-empty domain on the new screentype.) It is
possible thatcd.st=NIL . The booleancd.marked indicates whetherv was marked
when it was rescreened; in any case,v is automatically unmarked as it is rescreened.
VBT.Leaf.rescreen reshapesv to empty.

2.15 The repaint method

Trestle calls aVBT’s repaint method to report that part of its screen has been exposed
and must be repainted. The method will be called withLL.sup = mu.v , and takes
an argument of typeRegion.T .

There are some subtleties if you are scrolling (that is, copying bits from one part of
the screen to another) at the same time that Trestle is activating your repaint method.
To explain them we will become more formal and precise.

EveryVBT v has a “bad region”bad(v) . For each pointp that is inDomain(v)
and not inbad(v) , the pixelv[p] is displayed to the user; that is, ifvis[p] denotes
what is actually visible at pixelp, then we have the basic invariant

vis[p] = v[p] for all p controlled byv and outsidebad(v)

Trestle can expandbad(v) at any time, as though cosmic rays had damaged the pixels.
Wheneverbad(v) contains pixels that are controlled byv , Trestle will call v ’s

repaint method by settingexposed(v) (the “exposed region” ofv ) to include all such
pixels, and then executing the following code:

< bad(v) := the set differencebad(v) - exposed(v);
FOR p in exposed(v) DO v[p] := vis[p] END >;

v.repaint(exposed(v));
exposed(v) := the empty set

That is, as a pixelp is removed frombad(v) and added toexposed(v) , the screen
v[p] is changed tovis[p] , so that the basic invariant is maintained. You can
imagine that the cosmic ray’s damage has now reachedv[p] , not justvis[p] . The
angle brackets indicate that the shrinking ofbad(v) and the damaging ofv[p] occur
atomically, so that the basic invariant is maintained. (In particular, the basic invariant
is true whenever you call the procedureVBT.Scroll , where you can find more about
the bad region and the exposed region.)

Sometimes it is convenient to do all painting from the repaint method; in which
case the following procedure is useful:

PROCEDURE ForceRepaint(v: T; READONLY rgn: Region.T);
<* LL.sup < v * >

Setbad(v) := Region.Join(rgn, bad(v)) . If the resultingbad(v) is
non-empty, schedule an activation ofv ’s repaint method.

2.16 About painting in general

Trestle’s painting procedures all follow the same pattern. The arguments to the
procedure specify:

� a destination, which is a set of pixels in aVBT’s screen. For example, the
destination could be a rectangle, a trapezoid, a shape bounded by a curved path,
or a region.

� a source, which is conceptually an infinite array of pixels, not necessarily of the
same depth as those on the screen. For example, the source could be a texture, a
text string in some font, an explicit bitmap or image, or theVBT’s screen itself.

� anoperation, which is a function that takes a destination pixel value and a source
pixel value and produces a destination pixel value. For example, the operation
could be planewiseXOR.

The effect of the painting procedure is to apply the operation to each pixel in the
destination region. That is, ifv is theVBT, the effect of the painting procedure is to
setv[p] := op(v[p], s[p]) for each pointp in the destination, whereop is the
operation,v[p] is the pixel at pointp of v ’s screen, ands[p] is the source pixel at
pointp.

Two useful operations arePaintOp.Bg andPaintOp.Fg , defined by

PaintOp.Bg(d, s) = the screen’s background pixel
PaintOp.Fg(d, s) = the screen’s foreground pixel

These operations ignore their arguments; they seteach destination pixel to a constant
value, regardless of its previous value or the source value. The actual background and
foreground pixels vary from screentype to screentype; you can think ofBg as white
andFg as black (unless you prefer video-reversed screens).

Another useful operation isPaintOp.Copy , defined by

PaintOp.Copy(d, s) = s

For example,PaintOp.Copy can be used to paint an eight-bit pixmap source on an
eight-bit pixmap screen. It would be an error to usePaintOp.Copy with a one-bit
source and an eight-bit screen—the system wouldn’t crash, but anything could happen
to the destination pixels.s

For more painting operations, see thePaintOp interface.

2.17 Scrolling (copying one part of the screen to another)

PROCEDURE Scroll(
v: Leaf;
READONLY clip: Rect.T;
READONLY delta: Point.T;
op: PaintOp.T := PaintOp.Copy); < * LL.sup < v * >

Translate a rectangle ofv ’s screen bydelta and use it as a source for the
operationop applied to each destination pixel in the clipping rectangleclip .

The Scroll procedure usesv ’s screen as source. It can therefore be used to
copy pixels from one part ofv ’s screen to another. Any operation can be used for
combining the translated pixels with the destination pixels, but the operation defaults
to PaintOp.Copy .

The source rectangle can be computed fromclip by subtractingdelta . More
precisely,Scroll(v, clip, delta, op) is equivalent to:

for each pair of pointsp, q such that
p is in clip,
p = q + delta , and
q is in Domain(v)

simultaneously assign
v[p] := op(v[p], v[q]);
if q is in exposed(v) andp is not,

or if q is in bad(v)
then addp to bad(v)

By “simultaneously” it is meant that the pairsp, q are enumerated in an order so that
no destination pixel of an early pair corresponds to a source pixel of any later pair.

Recall the bad region and exposed regionbad(v) and exposed(v) from the
description of the repaint method.

If you do all your painting from within therepaint , reshape , andredisplay
methods, then you can ignore the subtleties involving thebad(v) andexposed(v) .
But if you have any asynchronous threads that callScroll , you have to be careful.
For example, suppose you do all your painting from a concurrent worker thread, and
arrange for your repaint and reshape methods to simply add entries to the worker
thread’s queue recording the painting that must be done. Then you must be careful to
avoid the following sequence of events:

� The worker thread removes from its work queue an item indicating that it must
repaint some regionA, and determines that the best way to do this is to scroll some
other regionB.

� The repaint method is activated with exposed regionB; it addsB to the work
queue and returns. As it returns, the system sets theVBT’s bad and exposed
regions to be empty. (See the description of therepaint method.)

� The worker thread copies the garbage fromB into A.

Eventually the worker thread will get around to repaintingB, but the damage toA
will never be repaired.

To avoid this race condition, the repaint method should convey the bad region to the
worker thread by a separate communication path, rather than simply put it the ordinary
work queue. The worker thread can thus avoid using bad bits as the source of scroll
operations.

Of course it is possible for the scrolling to happen after therepaint method is
called but before the method has conveyed the bad region to the worker thread. There
is no way to prevent this sequence of events, but there is no need to, either: in this case
the source of the scroll operation will be in the exposed region (since therepaint
method has not yet returned), and therefore (by the specification above) the call to
Scroll will expand the bad region. This will eventually lead to the repaint method
being activated a second time, repairing the damage.

In short, in order to allow concurrent painting, we do not clear the exposed region
until therepaint method returns, and we specify that a scroll from aq in bad(v) or
exposed(v) to ap that is not inbad(v) invalidates the destination.

Notice that a scroll fromexposed(v) to exposed(v) does not invalidate the
destination. This allows the repaint method to paint a portion ofexposed(v) and
then scroll that portion to other parts ofexposed(v) —unusual, but legal.

2.18 Painting textures

This section describes procedures for texturing rectangles, regions, and trapezoids.

PROCEDURE PaintTexture(
v: Leaf;
READONLY clip: Rect.T;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T;
READONLY delta := Point.Origin); < * LL.sup < v * >

Paint the rectangleclip with the texturesrc+delta using the operationop .

A textureis an infinite periodic pixmap. A texturetxt is represented by a pixmap
src with a finite non-empty rectangular domainDomain(src) ; the rule is thattxt
is the result of tiling the plane with translates of the pixmapsrc . Using the convenient
procedureRect.Mod we can state this rule as:txt[p] = src[Rect.Mod(p,
Domain(src))] .

The texturesrc+delta is the translation of the texturesrc by the vectordelta .
Putting this all together,PaintTexture(v, clip, op, src, delta) is

equivalent to:

for each pair of pointsp, q such that
p is in clip and
p = q + delta

assign
v[p] := op(v[p], src[Rect.Mod(q, Domain(src))]).

Note that settingdelta to Point.Origin causes the texture to be aligned in an
absolute coordinate system independent of the domain of the window (which helps to
make textures in different windows match), while setting it to the northwest corner of
v ’s domain causes the texture to be aligned in the window’s coordinate system (which
allows a window to be reshaped by scrolling the old domain into the new).

If src ’s domain is empty, the effect is undefined but limited to the clipping region.
The default paint operation for PaintTexture isBgFg, defined by

PaintOp.BgFg(d, 0) = the screen’s background pixel
PaintOp.BgFg(d, 1) = the screen’s foreground pixel

This paint operation is only appropriate ifsrc is one-bit deep; the effect is to copy the
source to the destination, interpreting 0 as background and 1 as foreground.

PROCEDURE PaintTint(
v: Leaf;
READONLY clip: Rect.T;
op: PaintOp.T); < * LL.sup < v * >

Paint the rectangleclip with the texturePixmap.Solid using the operation
op .

For example,PaintTint(v, clip, PaintOp.Bg) paintsclip with the back-
ground color, andPaintTint(v, clip, PaintOp.Fg) paints clip with the
foreground color.

PROCEDURE PolyTint(
v: Leaf;
READONLY clip: ARRAY OF Rect.T;
op: PaintOp.T); < * LL.sup < v * >

Paint each rectangleclip[i] in order with the texturePixmap.Solid using
the operationop .

PROCEDURE PolyTexture(
v: Leaf;
READONLY clip: ARRAY OF Rect.T;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T;

READONLY delta := Point.Origin); < * LL.sup < v * >

Paint each rectangleclip[i] in order with the texturesrc+delta using the
operationop .

PROCEDURE PaintRegion(
v: Leaf;
READONLY rgn: Region.T;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T := Pixmap.Solid;
READONLY delta := Point.Origin); < * LL.sup < v * >

Paint the regionrgn with the texturesrc+delta using the operationop .

PROCEDURE PaintTrapezoid(
v: Leaf;
READONLY clip: Rect.T;
READONLY trap: Trapezoid.T;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T := Pixmap.Solid;
READONLY delta := Point.Origin); < * LL.sup < v * >

Paint the intersection ofclip andtrap with the texturesrc+delta using the
operationop .

2.19 Filling and stroking paths

Trestle also supports PostScript-like graphics operations [3]:

TYPE
WindingCondition = {Odd, NonZero};
EndStyle = {Round, Butt, Square};
JoinStyle = {Round, Bevel, Miter};

PROCEDURE Fill(
v: Leaf;
READONLY clip: Rect.T;
path: Path.T;
wind := WindingCondition.NonZero;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T := Pixmap.Solid;
READONLY delta := Point.Origin); < * LL.sup < v * >

Paint the intersection ofclip and the region entwined bypath with the texture
src+delta using the operationop .

The pointp is entwined bypath if the winding number ofpath aroundp satisfies the
winding conditionwind . To ensure that the winding number is defined even for the
points on the path, the path is regarded as translated north by� and west by�2, where�
is infinitesimal.

PROCEDURE Stroke(
v: Leaf;
READONLY clip: Rect.T;
path: Path.T;
width: CARDINAL := 0;
end := EndStyle.Round;
join := JoinStyle.Round;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T := Pixmap.Solid;
READONLY delta := Point.Origin); < * LL.sup < v * >

Paint the intersection ofclip and the stroke determined bypath , end , and
join with the texturesrc+delta using the operationop .

The exact results ofStroke are different on different Trestle implementations. The
approximate specification is like PostScript:

If end = Round and join = Round , the path is drawn by a circular brush of
diameterwidth that traverses the path.

If end = Butt , then the ends of unclosed trails in the path are stroked by a line
segment of lengthwidth centered and perpendicular to the path in the neighborhood
of the endpoint. Ifend = Square , the path is extended at the endpoint by a straight
line segment of lengthwidth/2 tangent to the path and a butt end is drawn.

If join = Bevel , the joint between two patches is constructed by usingButt
endstyles for them and then filling the triangular notch that remains. Ifjoin =
Miter , then instead of just filling the triangular notch, the outer edges of the two lines
are extended to meet at a point, and the resulting quadrilateral is filled.

If width = 0 , join is ignored andend determines whether the final endpoint
of an open subpath should be drawn: ifend is Butt , the final endpoint is omitted,
otherwise it is drawn.

Finally, there is a convenience procedure for stroking a path containing a single
straight line segment:

PROCEDURE Line(
v: Leaf;
READONLY clip: Rect.T;
p, q: Point.T;
width: CARDINAL := 0;
end := EndStyle.Round;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T := Pixmap.Solid;
READONLY delta := Point.Origin); < * LL.sup < v * >

Like Stroke applied to the path containing the segment(p,q) .

2.20 Painting pixmaps

The following procedure paints a pixmap without replicating it into an infinite texture:

PROCEDURE PaintPixmap(
v: Leaf;
READONLY clip: Rect.T := Rect.Full;
op: PaintOp.T := PaintOp.BgFg;
src: Pixmap.T;
READONLY delta: Point.T); < * LL.sup < v * >

Translate the pixmapsrc by delta and paint it on the screen ofv , using the
operationop and clipping to the rectangleclip .

More precisely,PaintPixmap(v, clip, op, src, delta) is equivalent to

for each pair of pointsp, q such that
p is in clip,
q is in Domain(src) , and
p = q + delta,

assign
v[p] := op(v[p], src[q])

Since aPixmap.T is a screen-independent resource, you can’t read its domain without
specifying theVBT it is to be used on:

PROCEDURE PixmapDomain(v: T; pix: Pixmap.T): Rect.T;
<* LL.sup < v * >

Return the domain ofpix on the screentype ofv .

It is also possible to paint screen-dependent pixmaps:

PROCEDURE PaintScrnPixmap(
v: Leaf;
READONLY clip: Rect.T := Rect.Full;
op: PaintOp.T := PaintOp.Copy;
src: ScrnPixmap.T;
READONLY delta: Point.T); < * LL.sup < v * >

Like PaintPixmap , but with a screen-dependent pixmap instead of a
screen-independent pixmap.

If src does not have an appropriate screentype forv , the effect of the procedure is
undefined but limited to the clipping region.

Because Trestle batches painting operations, the pixmapsrc must be regarded as
still in use afterPaintScrnPixmap returns. If you wish to free the pixmap by calling
src.free() , you should first callVBT.Sync(v) .

2.21 Painting text

The text painting procedures take an optional array of displacements, whose entries
have the following type:

TYPE
DeltaH = [-512 .. 511];
Displacement =

RECORD index: CARDINAL; dh: DeltaH END;

A displacementd causes all characters whose index in the text isd.index or greater
to be displacedd.dh pixels to the right. The first character has index0. Thed.index
values in an array of displacements must benon-decreasing.

PROCEDURE PaintText(
v: Leaf;
READONLY clip: Rect.T := Rect.Full;
READONLY pt: Point.T;
fnt: Font.T := Font.BuiltIn;
t: TEXT;
op: PaintOp.T := PaintOp.TransparentFg;
READONLY dl := ARRAY OF Displacement{});

<* LL.sup < v * >

Paint the textt onto the screen ofv , starting at positionpt , using the fontfnt ,
the operationop , and the displacement listdl .

The arguments toPaintText must satisfy at least one of the following two conditions:

� the background operation is transparent; that is,op(p, 0) = p for any pixelp,
or

� the font is self-clearing (see below) anddl is empty.

If neither condition is true, the effect ofPaintText is implementation-dependent, but
is confined to the clipping rectangle.

TheScrnFont interface defines the properties of fonts. Here we introduce names
for the properties needed to explainPaintText . If f is a font andch is a character,
then

� printWidth(ch, f) is the printing width ofch ; that is, the amount to
increment the reference point whench is printed in fontf ;

� bits(ch, f) is the bitmap forch in f , which is positioned withch ’s reference
point at the origin;

� height(ch, f) is the height ofch above the baseline; that is, the number of
rows of bits(ch, f) whosev -coordinate is at most zero; anddepth(ch,
fnt) is the number of rows ofbits(ch, f) whosev -coordinate exceeds zero;

� ascent(f) anddescent(f) are the logical extent off above and below the
baseline. Some characters may extend higher or lower.

A font is self-clearingif

� each character’s height and depth equal the font’s ascent and descent, and

� each character’sprintWidth equals the width of its bitmap and each character’s
reference point is at the west boundary of its bitmap (oreach character’s
printWidth equals the negative of the width of its bitmap and each character’s
reference point is at the east boundary of its bitmap).

The call toPaintText is equivalent to the following loop:

rp := pt;
i := 0;
LOOP

IF dl # NIL THEN
FOR j := 0 TO HIGH(dl^) DO

IF dl[j].index = i THEN INC(rp.h, dl[j].dh) END
END

END;
IF i = Text.Length(t) THEN EXIT END;
PaintPixmap(v, clip, op, bits(t[i], fnt), rp);
rp.h := rp.h + PrintWidth(t[i], fnt);
i := i + 1

END

The following two procedures are useful for computing the sizes of texts. Since fonts
are screen-independent, they take theVBTwhose screentype is to be used:

PROCEDURE BoundingBox
(v: Leaf; txt: TEXT; fnt: Font.T): Rect.T;
<* LL.sup < v * >

Return the bounding box of the texttxt if it were painted at the origin on the
screen ofv .

More precisely, letr be the smallest rectangle that contains the bounding boxes of the
characters oftxt if txt were painted onv in the fontfnt with txt ’s reference point
at the origin. ThenBoundingBox returns a rectangle with the same horizontal extent
asr , but whose height and depth are the maximum height and depth of any character
in the font.

PROCEDURE TextWidth
(v: Leaf; txt: TEXT; fnt: Font.T): INTEGER;
<* LL.sup < v * >

Return the sum of the printing widths of the characters intxt in the fontfnt .

TextWidth returns the displacement of the reference point that would occur ift
were painted onv in font fnt . It may differ from the width ofBoundingBox(txt,
fnt) , since the printing width of the last character can be different from the width of
its bounding box, and the reference point for the first character might not be at the left
edge oftxt ’s bounding box.

You can paint characters out of an array instead of aTEXT:

PROCEDURE PaintSub(
v: Leaf;
READONLY clip: Rect.T := Rect.Full;
READONLY pt: Point.T;
fnt: Font.T := Font.BuiltIn;
READONLY chars: ARRAY OF CHAR;
op: PaintOp.T := PaintOp.TransparentFg;
READONLY dl := ARRAY OF Displacement{});
<* LL.sup < v * >

Like PaintText applied to the characters inchars .

2.22 Synchronization of painting requests

To improve painting performance, Trestle combines painting commands into batches,
and sends them to the server a batch at a time.

Most applications can ignore the batching, but the procedures in this section can be
of use in applications where the timing of paint operations is critical.

For example, when replacing one line of text with another in a non-self-clearing
font, the old text must be erased before the new text is painted. If the painting command
that erases the old text happens to fall at the end of a batch, there may be a delay of
several milliseconds between the time it affects the screen and the time the following
paint text command affects the screen, which can produce an undesirable flickering
effect. The chances of this happening can be greatly reduced by enclosing the two
commands in agroup, using the following two procedures:

PROCEDURE BeginGroup(v: Leaf; sizeHint: INTEGER := 0);
<* LL.sup < v * >

Begin a group of painting commands.

PROCEDURE EndGroup(v: Leaf); < * LL.sup < v * >

End the current group of painting commands.

If a group of painting commands are bracketed byBeginGroup and EndGroup ,
Trestle will try to avoid introducing delays between the commands, such as might
otherwise be introduced by batching. Trestle assumes that you will generate the
painting commands and theEndGroup in rapid succession.

Increasing the value ofsizeHint may improve atomicity, at the cost of throughput.
The maximum useful value ofsizeHint is the total size in bytes of the painting
commands in the group, which you can compute using the interfacePaintPrivate .

PROCEDURE Sync(v: Leaf); < * LL.sup < v * >

Wait until all painting commands issued tov prior to the call toSync have been
executed.

2.23 Screen capture

PROCEDURE Capture(
v: T;
READONLY clip: Rect.T;
VAR (* out * ) br: Region.T)
: ScrnPixmap.T; < * LL.sup < v * >

Return a pixmap containing the part ofv ’s screen in the rectanglerect .

The screentype of the result will be the same as the screentype ofv . Because aVBT’s
screen is forgetful, it may be impossible to read the requested region. In this casebr
is set to contain all positions of pixels that were not copied. Naturally, Trestle makes
br as small as it can. If none of the bits are available, the result may beNIL .

2.24 Controlling the cursor shape

EveryVBT v contains a fieldcursor(v) , which is set with the following procedure:

PROCEDURE SetCursor(v: T; cs: Cursor.T);
<* LL.sup < v * >

Setcursor(v) to cs .

A split displays the cursor of its mouse focus, or of its current child if its mouse focus
is NIL . Only if the cursor of the relevant child isCursor.DontCare or if there is no
relevant child does the split display its own cursor.

To be more precise, the shape of the cursor over the top level windowv is
determined by the following recursive procedure:

GetCursor(v) =
IF NOT ISTYPE(v, Split) THEN

RETURN cursor(v)
ELSE

IF mouseFocus(v) # NIL THEN
cs := GetCursor(mouseFocus(v))

ELSIF current(v) # NIL THEN
cs := GetCursor(current(v))

ELSE
cs := Cursor.DontCare

END;
IF cs = Cursor.DontCare THEN

RETURN cursor(v)
ELSE

RETURN cs
END

END

2.25 Selections

Trestle maintains an internal table of named selections, which initially contains several
selections of general use, and which can be extended by users:

TYPE Selection = RECORD sel: CARDINAL END;

PROCEDURE GetSelection(name: TEXT): Selection;
<* LL arbitrary * >

Return the selection with the given name, creating it if necessary.

PROCEDURE SelectionName(s: Selection): TEXT;
<* LL arbitrary * >

Return the name used to creates , or NIL if s is unknown.

VAR (* CONST* )
NilSel: Selection ( * := GetSelection("NilSel") * );
Forgery: Selection ( * := GetSelection("Forgery") * );
KBFocus: Selection ( * := GetSelection("KBFocus") * );
Target: Selection ( * := GetSelection("Target") * );
Source: Selection ( * := GetSelection("Source") * );

NilSel andForgery are reserved for Trestle’s internal use. The owner ofKBFocus
(the keyboard focus) is theVBT that receives keystrokes.

We offer the following suggestions for the use of target and source selections:

� The target selection. If text, this should be underlined black or reverse video. The
selection gesture should not require modifiers like shift or control.

� The source selection. If text, this should be underlined gray. The source gesture
should be a modified version of the gesture for making the target selection.

An operation like “copy” should replace the target selection with the value of the
source selection.

The following exception declaration provides for the errors that can occur in dealing
with selections.

EXCEPTION Error(ErrorCode);

TYPE ErrorCode =
{EventNotCurrent, TimeOut, Uninstalled, Unreadable,

Unwritable, UnownedSelection, WrongType};

Explanation of error codes:

� EventNotCurrent : Raised by attempts to access a selection with an event time
that is not current.

� TimeOut : If you attempt to read or write a selection, and the selection owner’s
method does not return for an unreasonably long time, then Trestle stops waiting
and raises this exception.

� Uninstalled : Raised by event-time operations on uninstalledVBTs; that is, on
VBTs none of whose ancestors have been connected to a window system by one
of the installation procedures in theTrestle interface.

� Unreadable ,Unwritable : Raised by attempts to read an unreadable selection,
or write an unwritable selection.

� UnownedSelection : Raised by attempts to read,write, or deliver miscellaneous
codes to the owner of an unowned selection.

� WrongType : Raised by attempts to read or write a selection with a type not
supported by the selection owner.

2.26 Acquiring and releasing selection ownership

PROCEDURE Acquire(
v: T;
s: Selection;
t: TimeStamp)

RAISES {Error}; < * LL.sup < v * >

Makev the owner of selections , provided thatt is the current event.

If Acquire(v, s, t) is successful, the previous owner of the selection will receive a
miscellaneous code of typeLost (even if the owner isv ). The window system affected
is the one to whichv is connected. The possible error codes areEventNotCurrent
andUninstalled .

PROCEDURE Release(v: T; s: Selection);
<* LL.sup < v * >

If the current owner ofs is v , then aLost code is queued for delivery tov and
the owner ofs becomesNIL

The window system affected is the one to whichv is connected.Release is a no-op
if the current owner is notv or if v is not installed.

2.27 The miscellaneous method

Trestle calls aVBT’s misc method to deliver miscellaneous codes. The method will
be called withLL.sup = mu , and takes an argument of typeMiscRec .

Trestle maintains an internal table of named miscellaneous code types, which
initially contains several types of general interest, and which can be extended by users.

TYPE MiscRec = RECORD
type: MiscCodeType;

detail: MiscCodeDetail;
time: TimeStamp;
selection: Selection;

END;

MiscCodeType = RECORD typ: CARDINAL END;
MiscCodeDetail = ARRAY [0 .. 1] OF INTEGER;

PROCEDURE GetMiscCodeType(name: TEXT): MiscCodeType;
<* LL arbitrary * >

Return the MiscCodeType with the given name, creating it if necessary.

PROCEDURE MiscCodeTypeName(type: MiscCodeType): TEXT;
<* LL arbitrary * >

Return the name used to creates , or NIL if s is unknown.

CONST
NullDetail = MiscCodeDetail {0, ..};

VAR (* CONST* )
Deleted: MiscCodeType;
Disconnected: MiscCodeType;
TakeSelection: MiscCodeType;
Lost: MiscCodeType;
TrestleInternal: MiscCodeType;

These “variables” are really constants for the following codes:

GetMiscCodeType("Deleted")
GetMiscCodeType("Disconnected")
GetMiscCodeType("TakeSelection")
GetMiscCodeType("Lost")
GetMiscCodeType("TrestleInternal")

The method callv.misc(cd) sendsv the misc code relevant tocd.selection as
part of the eventcd.time . The meaning of thetype anddetail fields is up to the
application, except for the following.

A Deleted code is delivered to a top-level window when it is explicitly deleted
from its server, either by a user command to the window manager or under program
control. A Disconnected code is delivered to a top-level window when it is
disconnected from its server, either because the server crashed or because the network
connection was lost. ATakeSelection code is delivered to a top-level window
when the user has gestured that it would like the window to acquire the indicated
selection; most often the keyboard focus. (The nature of the gesture is between the
user and the window manager. Many applications also acquire the keyboard focus in
response to mouse clicks.) ALost code withselection = s will be delivered to
a window when it loses ownership ofs . TrestleInternal codes are reserved for
the implementation.

The timestamp in aTakeSelection code is the timestamp for the current
event and is therefore valid for event-time operations. The timestamps inDeleted ,
Disconnected , andLost codes are not. The selection field is relevant inLost and
TakeSelection codes; it is irrelevant inDeleted andDisconnected codes.

2.28 Sending miscellaneous codes

You can send a miscellanous code to the owner of a selection by using the following
procedure:

PROCEDURE Put(
v: T;
s: Selection;
t: TimeStamp;
type: MiscCodeType;
READONLY detail := NullDetail)

RAISES {Error}; < * LL.sup < v * >

Create aMiscRec with the given fields and enqueue it for delivery to the owner
of selections , if t is the current event-time.

The window system affected is the one to whichv is connected. The possible error
codes areEventNotCurrent , Uninstalled , andUnownedSelection . If the
selection is unowned it is possible that thePut will be silently ignored.

2.29 Circumventing event-time

The following procedure offers an escape from the event-time protocol. For example,
a long-running thread that has no idea what the current event time is can forge a
miscellaneous code to itself and use its timestamp to acquire the keyboard focus. (Your
users may not like it if you do this.)

PROCEDURE Forge(
v: T;
type: MiscCodeType;
READONLY detail := NullDetail)

RAISES {Error}; < * LL.sup < v * >

Create aMiscRec with the giventype anddetail fields, with selection field
Forgery , and with a newly created timestamp and enqueue it for delivery tov .

The timestamp will be valid for event-time operations (provided that it is used
promptly). Forging codes that have meaning to the window manager (e.g., aDeleted
code) could have unexpected effects if they are delivered to installed windows or their
descendants. The only possible error code isUninstalled .

2.30 Communicating selection values

When you read the value of a Trestle selection you get a result of typeValue :

TYPE
Value <: Value_Public;
Value_Public =

OBJECT METHODS toRef(): REFANY RAISES {Error} END;

Call thetoRef method to convert theValue into aREFANY.
The simplest way to construct aValue is with the following procedure:

PROCEDURE FromRef(r: REFANY): Value;
<* LL.sup <= mu * >

Return aValue v such thatv.toRef() is equal to the result of pickling and
unpicklingr .

On a system without pickles, the valuer must have typeTEXT. If r does not have type
TEXT, any exceptions raised by pickling lead to checked run-time errors.

Using FromRef leads to synchronous transmission of selection values—that is,
the value is transferred as part of the call toRead or Write . To get asynchronous
behavior, allocate your ownValues and override thetoRef method. Trestle will
transmit theValue to the other application, and only when that application calls the
toRef method will yourtoRef method be called.

The toRef method in aValue will be called withLL.sup <= mu . The toRef
method can raise the errorUnreadable if, for example, the address space of the
selection owner has been destroyed. It can also raise the errorWrongType if the
underlyingREFANYcannot be represented in the address space calling the method; this
can only happen with non-TEXTselections.

The procedureReady tests whether a value is synchronous or asynchronous:

PROCEDURE Ready(v: Value): BOOLEAN; < * LL.sup <= mu * >

ReturnTRUEif calling v.toRef() will return quickly; returnFALSEif calling
v.toRef() might be slow or block.

Finally, here are the procedures for reading and writing selections:

PROCEDURE Read(
v: T;
s: Selection;
t: TimeStamp;
tc: INTEGER := -1)
: Value

RAISES {Error}; < * LL.sup <= mu * >

Return the value of selections as a reference of typetc , if t is the current
event-time.

If tc = -1 , Read uses the typecode forTEXT. The window system affected is the one
to whichv is connected. TheKBFocus selection is always unreadable. If the selection
owner’s read method is erroneous, calling thetoRef method of the returnedValue

may produce a reference with a typecode other thantc . The possible error codes are
EventNotCurrent , Uninstalled , Unreadable , WrongType , TimeOut , and
UnownedSelection .

PROCEDURE Write(
v: T;
s: Selection;
t: TimeStamp;
val: Value;
tc: INTEGER := -1)

RAISES {Error}; < * LL.sup <= mu * >

Replace the selections with the valuev , which encodes a reference with
typecodetc , assumingt is the current event-time.

If tc = -1 , Write uses the typecode forTEXT. The window system affected is
the one to whichv is connected. TheKBFocus selection is always unwritable.
The possible error codes areEventNotCurrent , Uninstalled , Unwritable ,
TimeOut , andWrongType .

2.31 The read and write methods

Trestle calls aVBT’s read and write methods toaccess any selections that it owns. The
method will be called withLL.sup <= mu (see below).

The signature of the read method is

(s: Selection; tc: CARDINAL): Value RAISES {Error}

Trestle callsv.read(s, tc) wheneverv is the owner of selections and some
application passess and tc to Read. The method should return the value of
the selection, or raiseError(Unreadable) if for some reason the value cannot
be delivered, orError(WrongType) if the selection cannot be converted to the
requested type. The methods will be called withLL.sup <= mu ; in fact, if the caller
of Read is in the same address space,LL for the method call is the same asLL for the
caller ofRead, elseLL for the method call is{} .

The signature of the write method is

(s: Selection; val: Value; tc: CARDINAL)
RAISES {Error}

Trestle callsv.write(s, val, tc) wheneverv is the owner of selections and
some application passess , val , and tc to Write . The method should replace the
selection with the value ofval , or raise the exception with error codeUnwritable
if for some reason the selection is not writable, or with error codeWrongType if
the selection cannot be written with the requested type. Trestle does not enforce
any consistency betweentc and the typecode of the referenceval.toRef() . For
example, ifval.toRef() is NIL , the meaning could be determined bytc . The
locking level is the same as for the read method.

While a read or write method is active in a descendant of an installed window,
Trestle will block the delivery to that window of any mouse or key events, misc codes,
or cursor positions. If the computations are long, it is therefore preferable to do them
asynchronously, to avoid blocking the user.

2.32 Controlling the shape of a VBT

The preferred shape of aVBT is represented by a pair of records of typeSizeRange ,
one for each axis:

TYPE SizeRange = RECORD lo, pref, hi: CARDINAL END;

CONST DefaultShape =
SizeRange{lo := 0, pref := 0, hi := 99999};

If a VBT’s preferred shape in the axisax is theSizeRange sh , then the desirable
sizes for theVBT in axisax range fromsh.lo to sh.hi-1 , and its preferred size is
sh.pref .

A SizeRange sh is illegal unlesssh.lo <= sh.pref < sh.hi .
When a parentVBTdivides its screen up between its children, it tries to satisfy its

children’s shape requirements, which it finds by calling the children’s shape method.
The signature of the shape method is

(ax: Axis.T; n: CARDINAL): SizeRange

The behavior of the shape method depends on whethern is zero. The call
v.shape(ax, 0) returns the preferred shape forv in theax axis, assuming nothing
is known about its size in the other axis. Ifn#0 , the callsh := v.shape(ax, n)
returns the preferred shape forv in theax axis assuming thatv ’s size in the other axis
is n. When the method is called,LL.sup = mu.v .

It is a checked runtime error for a shape method to return an illegal size range. A
common error is to return an illegal size range withsh.lo = sh.hi .

The child must not assume that its shape requirement is satisfied, since, for example,
the requirements of a split’s children can be inconsistent.

The defaultshape method for aLeaf returnsDefaultShape .
When the preferred shape of aVBTchanges, you should callNewShape:

PROCEDURE NewShape(v: T);
<* LL.sup >= mu.v AND LL.sup < v * >

Notify v ’s parent that its preferred size range may have changed.

Typically, the parent will mark itself, and any change will take effect at the time of
the next redisplay. Notice that the locking level allowsNewShape to be called from a
reshape or rescreen method; it can also be called from a thread that hasmulocked.

2.33 Putting properties on a VBT

Associated with each window is a “property set”, which is a set of non-nil traced
references.

PROCEDURE PutProp(v: T; ref: REFANY); < * LL.sup < v * >

Add ref to v ’s property set, replacing any existing reference of the same type
asref . This is a checked runtime error ifref is NIL .

PROCEDURE GetProp(v: T; tc: INTEGER): REFANY;
<* LL.sup < v * >

Return the element ofv ’s property set with typecodetc , or NIL if no such
element exists.

PROCEDURE RemProp(v: T; tc: INTEGER); < * LL.sup < v * >

Remove the element with typecodetc from v ’s property set, if one exists.

2.34 Discarding a VBT

It is good form to callVBT.Discard(v) whenv is about to be garbage-collected:

PROCEDURE Discard(v: T); < * LL.sup = mu * >

Prepare for and callv.discard() .

The discard method will be called withLL.sup = mu , and takes no argument. The
method should perform any class-dependent cleanup that is needed. The default discard
method is a no-op.

END VBT.

3 The Trestle interface

TheTrestle interface provides routines for connecting to window systems; installing,
decorating, and moving top-level windows, and performing related operations.

INTERFACE Trestle;

IMPORT VBT, Rect, Point, Region, ScrnPixmap,
TrestleComm;

TYPE
T <: ROOT;

A Trestle.T identifies an instance of a window system. All the routines in this
interface that take aTrestle.T accept the valueNIL , which represents the default
window system obtained by callingConnect(NIL) .

PROCEDURE Install(
v: VBT.T;
applName: TEXT := NIL;
inst: TEXT := NIL;
windowTitle: TEXT := NIL;
iconTitle: TEXT := NIL;
bgColorR: REAL := -1.0;
bgColorG: REAL := -1.0;
bgColorB: REAL := -1.0;
iconWindow: VBT.T := NIL;
trsl: T := NIL)

RAISES {TrestleComm.Failure}; < * LL.sup <= VBT.mu * >

Initiate the installation ofv as a decorated top-level window of the window
systemtrsl .

Install may return before the installation is complete.Install is a checked
runtime error ifv is not detached, or ifv is in the process of being installed. The
position of the window on the screen depends on the window manager.

The textapplName is the application name; it defaults to the application name
from the process environment.

The text inst distinguishes windows with the same application name. For
example, a text editor might use the full path name of the file being edited as the
instance. The default is the value of the environment variableWINSTANCE.

Trestle does not require that the pair(applName, inst) be unique, but session
management tools will work more smoothly if it is.

The textwindowTitle will be placed in the window’s title bar when the window
is not iconic. It defaults to the concatenation ofapplName , a space, andinst , or just
to applName if inst is NIL .

The icon for the window will contain the texticonTitle together withicon-
Window (if it is not NIL ). For example,iconWindow might be a smallBitmapVBT .

Some window managers ignoreiconWindow . The default foriconTitle is inst ,
or applName if inst is NIL .

The triplebgColorR , bgColorG , bgColorB specify the red, green, and blue
components of the background color for the window and icon titles. If they are
defaulted, the window manager’s default background color will be used; if they are not
defaulted they should be between0.0 and1.0 . Some window managers ignore the
background color.

An installed window’s maximum, minimum, and preferred size will be reported to
the window manager, initially and whenever they change. However, aStableVBT
filter is inserted aboveeach installed window, so that a new preferred size will not be
reported if the window’s current size satisfies the new max and min constraints. Use
StableVBT.Disable to force a new preferred size.

It is a checked runtime error if eitherv or iconWindow is already installed.
Installing awindow inserts oneor morefilters above it, including aHighlightVBT ,

a StableVBT , and filters that make screen-independent resources work.

PROCEDURE AwaitDelete(v: VBT.T); < * LL = {} * >

Wait until v is deleted or disconnected from whatever window system it is
installed on.

AwaitDelete will not return until after theDeleted or Disconnected code has
been delivered and processed by the window. It is a noop ifv is already deleted or is
not installed.

PROCEDURE Delete(v: VBT.T); < * LL.sup = VBT.mu * >

Deletev from wherever it is installed.

Delete automatically releases any selections owned byv or any ofv ’s descendants.
BeforeDelete(v) returns, lost codes will be delivered for any such selections. If
v owned the mouse focus,v will also receive a synthesized mouse transition of type
LastUp . Thenv will receive aDeleted code, and finallyDelete will return. At
this pointv is disconnected and can be re-installed.

PROCEDURE Decorate(
v: VBT.T;
instance: TEXT := NIL;
windowTitle: TEXT := NIL;
iconTitle: TEXT := NIL;
bgColorR: REAL := -1.0;
bgColorG: REAL := -1.0;
bgColorB: REAL := -1.0;
applName: TEXT := NIL;
iconWindow: VBT.T := NIL)

RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Change the decorations ofv to the given values

Any parameter that is defaulted will not be changed, unless v has beenAttached
since it was last decorated, in which case the default value is computed as inInstall .
Decorate is a noop ifv is not an installed decorated window.

PROCEDURE GetDecoration(v: VBT.T;
VAR instance, windowTitle, iconTitle, applName: TEXT;
VAR bgColorR, bgColorG, bgColorB: REAL;
VAR iconWindow: VBT.T): BOOLEAN; < * LL.sup = VBT.mu * >

If v is decorated, fetchv ’s decorations, and returnTRUE. Otherwise, return
FALSE.

3.1 Window placement

PROCEDURE Attach(v: VBT.T; trsl: T := NIL)
RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Attachv to the window systemtrsl , leaving it invisible.

Attach is like Install , except (1) the locking level is different, (2) the attachment
is completed beforeAttach returns, (3) the window becomes undecorated, and (4) the
window remains invisible until you callOverlap , Iconize , or MoveNear . Before
calling one of these, most clients will want to callDecorate .

PROCEDURE Overlap(
v: VBT.T;
id: ScreenID;
READONLY nw: Point.T)

RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Move the northwest corner ofv to the pointnw on the screenid .

If v is undecorated, this produces a window with no title bar or border, and the user
will probably not be able to move, iconize or delete the window; this is a bad idea
unless you’re implementing pop-up or pull-down menus.

PROCEDURE Iconize(v: VBT.T)
RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Make the windowv become iconic.

PROCEDURE MoveNear(v, w: VBT.T)
RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Move the windowv to be near the windoww.

The exact effect ofMoveNear depends on the window manager. Ifw is NIL or is not
installed wherev is, thenMoveNear will attempt to bringv to the attention of the
user; in particular, ifv is an overlapping window,v will be brought to the top; ifv is

an icon, it will be deiconized; ifv is in the invisible state produced byAttach , it will
be opened in some visible place.

Overlap , Iconize , and MoveNear are all no-ops ifv is not installed. The
effects ofIconize andMoveNear are undefined for undecorated windows.

PROCEDURE InstallOffscreen(
v: VBT.T;
width, height: CARDINAL;
preferredScreenType: VBT.ScreenType)

RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Give v a domain with the given dimensions in the off-screen memory of the
window system to which it is attached.

InstallOffscreen rescreensv to preferredScreenType , or something as
much like it as supported for off-screen windows. The windowv must be in the
floating state produced byAttach . The usual purpose is to paint onv and then use
VBT.Capture to retrieve the contents of its screen as a pixmap. You should delete
v when you are done with it. Untilv is deleted, you should not pass it toOverlap ,
Iconize , MoveNear or InstallOffscreen .

3.2 Enumerating and positioning screens

A window system may have multiple screens. Each screen is identified by an integer.

TYPE ScreenID = INTEGER;

CONST NoScreen: ScreenID = -1;

TYPE ScreenOfRec = RECORD
id: ScreenID;
q: Point.T;
trsl: T;
dom: Rect.T

END;

PROCEDURE ScreenOf(
v: VBT.T; READONLY p: Point.T)
: ScreenOfRec; < * LL.sup < v * >

Return information about wherev is installed.

If v is an installed window then afterres := ScreenOf(v, p) we have

� res.id is theID of the screen currently containingv ;

� res.q is the point in screen coordinates that corresponds to the pointp in window
coordinates;

� res.trsl is the window system on whichv is installed; and

� res.dom is the domain of the screenres.id .

The pointp need not be in the domain ofv . If v is not installed, thenres.trsl
will be NIL , res.id will be NoScreen , and the other fields will be arbitrary. If
the window manager is movingv between screens whenScreenOf is called, then
res.id will be NoScreen andres.dom andres.q will be arbitrary.

TYPE
Screen = RECORD

id: ScreenID;
dom: Rect.T;
delta: Point.T;
type: VBT.ScreenType

END;
ScreenArray = REF ARRAY OF Screen;

PROCEDURE GetScreens(trsl: T := NIL): ScreenArray
RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Return an array of descriptors of the screens of the window systemtrsl .

For eachScreen s in the returned array, the rectangles.dom is the domain of the
VBTat the root of the screen. The screens all lie in a global coordinate system, within
which the user moves the cursor. The pointp in screen coordinates corresponds to the
point p+s.delta in global coordinates. (Some window systems don’t support this;
in which cases.delta will be set toPoint.Origin for all screens.) The value
s.type is the screentype of the screen’s rootVBT. GetScreens returnsNIL if the
window system has no screens.

3.3 Reading pixels from a screen

PROCEDURE Capture(
id: ScreenID;
READONLY clip: Rect.T;
VAR (* out * ) br: Region.T;
trsl: T := NIL)
: ScrnPixmap.T

RAISES {TrestleComm.Failure};
<* LL.sup = VBT.mu * >

Read the contents ofclip from screenid of trsl .

Capture(id, clip, br, trsl) is likeVBT.Capture(r, clip, br) ,where
r is theVBTat the root of screenid of the window systemtrsl .

3.4 Checking on recent input activity

PROCEDURE AllCeded(trsl: T := NIL): BOOLEAN
RAISES {TrestleComm.Failure}; < * LL.sup = VBT.mu * >

Return whether there is pending input fromtrsl .

If a program callsAllCeded(t) andTRUEis returned, then there are no mouse clicks
or keystrokes on their way to any top-level windows installed by the program ont . For
example, when the VT100 terminal emulator has observed a key-down and waited for
half a second and observed no key-up and concludes that it should go into auto-repeat
mode, it verifies thatAllCeded returnsTRUEto make sure that the up transition is not
on its way, to avoid erroneously entering auto-repeat mode.

PROCEDURE TickTime(trsl: T := NIL): INTEGER;
<* LL.sup <= VBT.mu * >

Return the number of microseconds perVBT.TimeStamp , in events reported to
VBTs connected to the window systemtrsl .

3.5 Connecting to a window system

PROCEDURE Connect(inst: TEXT := NIL): T
RAISES {TrestleComm.Failure}; < * LL.sup <= VBT.mu * >

Connect to the window system namedinst .

In general, the format and interpretation ofinst are implementation-dependent. Here
are the rules when using an X server:

If inst is NIL , it defaults to the value of the environment variableDISPLAY,
unless this variable is undefined, in which case it defaults to:0 .

The syntax ofinst should be:

<machine name>(":" | "::")<number>("" | "." <number>)

where<machine name> is an arbitrary string of characters (possibly empty) and
<number> is a non-negative decimal integer. It denotes an X serveraccording to the
rules on page 27 of the second edition ofX Window System, by Scheifler et. al., Digital
Press, 1990 [5].

For example,nemesia:0 denotes the first window system on the machine
nemesia , and:0 denotes the first window system on the machine callingConnect .

The exception is raised if the designated window system doesn’t exist or cannot be
connected to.

END Trestle.

4 Splits

4.1 The Split interface

The Split interface provides the functionality that is common to all splits; for example,
enumerating and deleting children.

This interface is for clients of splits; see theVBTClass and ProperSplit
interfaces for information about implementing your own split classes.

INTERFACE Split;

IMPORT VBT, Point, Rect;

TYPE T = VBT.Split;

EXCEPTION NotAChild;

A Split.T is a VBT that divides its screen up between one or more childVBTs.
The children of a split are ordered; they can be enumerated with theSucc andPred
procedures:

PROCEDURE Succ(v: T; ch: VBT.T): VBT.T
RAISES {NotAChild}; < * LL >= {VBT.mu} * >

Return the child ofv that follows the childch .

The successor ofNIL is the first child; the successor of the last child isNIL ; the
successor ofNIL is NIL if there are no children. The exceptionNotAChild is raised
if ch is not a child ofv .

PROCEDURE Pred(v: T; ch: VBT.T): VBT.T
RAISES {NotAChild}; < * LL >= {VBT.mu} * >

Return the child ofv that precedes the childch .

More precisely,Pred(v,ch) = x iff Succ(v,x) = ch . All of Trestle’s standard
splits implementSucc andPred in constant time.

PROCEDURE NumChildren(v: T): CARDINAL;
<* LL >= {VBT.mu} * >

Return the number of children ofv .

PROCEDURE Nth(v: T; n: CARDINAL): VBT.T;
<* LL >= {VBT.mu} * >

Return the child ofv with indexn.

More precisely,Nth(v, n) is the child ofv with n predecessors, orNIL if v has at
mostn children. Warning: for Trestle’s standard splits,Nth requires time proportional
to n, so it would be wasteful to enumerate the children by calling it repeatedly; use
Succ instead.

PROCEDURE Index(v: T; ch: VBT.T): CARDINAL
RAISES {NotAChild}; < * LL >= {VBT.mu} * >

Return the index ofv ’s child ch .

Index(v, ch) is the valuen such thatNth(v, n) = ch . Index(v, NIL)
equalsNumChildren(v) .

PROCEDURE Locate(v: T; READONLY pt: Point.T): VBT.T;
<* LL.sup = VBT.mu * >

Return the child ofv that controls the pointpt , or NIL if there is no such child.

PROCEDURE Delete(v: T; ch: VBT.T) RAISES {NotAChild};
<* LL.sup = VBT.mu * >

Delete the childch of the splitv , detachch , and markv for redisplay.

PROCEDURE Replace(v: T; ch, new: VBT.T)
RAISES {NotAChild}; < * LL.sup = VBT.mu * >

Replace childch of v with new, detachch (which must not beNIL ), and mark
v for redisplay.

PROCEDURE Insert(v: T; pred, new: VBT.T)
RAISES {NotAChild}; < * LL.sup = VBT.mu * >

Add new as a child ofv following pred .

Some split classes can accomodate only abounded number of children (for example,
filters). If Insert(v, pred, new) is applied to a splitv that cannotaccomodate
an additional child, thenpred (or the original first child, ifpred = NIL ) is deleted
from the split and discarded. The precise semantics are defined by the individual splits.
Insert raisesNotAChild if pred isn’t a child ofv , and is a checked run-time error
if new isn’t detached.

PROCEDURE Move(v: T; pred, ch: VBT.T)
RAISES {NotAChild}; < * LL.sup = VBT.mu * >

Move childch of v to follow pred . Bothch and (if non-NIL ) pred must be
children ofv .

PROCEDURE AddChildArray(v: T;
READONLY new: ARRAY OF VBT.T);

<* LL.sup = VBT.mu * >

Insert the non-NIL elements ofnew at the end of thev ’s list of children.

AddChildArray is equivalent to

pred := Pred(v, NIL);
FOR i := 0 TO LAST(new) DO

IF new[i] # NIL THEN

InsertAfter(v, pred, new[i]);
pred := new[i]

END
END

PROCEDURE AddChild(v: T;
v0, v1, v2, v3, v4, v5, v6, v7, v8, v9: VBT.T := NIL);

<* LL.sup = VBT.mu * >

Add the given children tov .

AddChild is equivalent to

AddChildArray(v,
ARRAY OF VBT.T{v0, v1, ..., v9})

END Split.

4.2 The ZSplit interface

A ZSplit.T is a parent window with overlapping child windows.
Each child has a stacking order given (conceptually) by az coordinate. A pixel

of the parent’s screen that is in the domain of more than one child is controlled by
whichever of these children is highest in thez coordinate. The portions of the domains
of the children that extend outside the parent domain will be clipped.

Split.Succ enumerates the children from top to bottom.
The bottom child is called thebackground. An initial background can be specified

when theZSplit is created; usually it remains the background throughout the life of
theZSplit . Usually the background has the same domain as the parent, and therefore
controls all pixels that are not controlled by any other child. In the unusual case that
the background child has a domain different from the parent domain, there may be
some parent pixels that are not controlled by any child. TheZSplit will ignore these
pixels when asked to repaint.

The shape of aZSplit is the shape of its background child (if it has no children its
shape is the default shape for aVBT). When the preferred shape of a non-background
child changes, theZSplit reshapes the child to its new preferred shape, preserving its
offset, which is the vector between the northwest corners of the parent and child.

INTERFACE ZSplit;

IMPORT VBT, Rect, Split, Point;

TYPE
T <: Public;
Private <: Split.T;
Public = Private OBJECT METHODS

<* LL <= VBT.mu * >

init(bg: VBT.T := NIL;
saveBits := FALSE;
parlim: INTEGER := -1): T

END;

The callv.init(...) initializesv as aZSplit .

It is only legal to call theinit method for a newly-allocatedZSplit (as in the
definition of the procedureNewbelow) or from theinit method of a subclass. This
restriction applies to all theinit methods in Trestle, although it will not be repeated
for each one.

The ZSplit will be given the initial background childbg if bg#NIL ; it will
be given no children ifbg=NIL . If bg is non-NIL it will be mapped initially. If
saveBits is TRUE, the split will try to save the children’s old bits when reformatting;
if the children don’t use them anyway, it is faster to letsaveBits default toFALSE.
The value ofparlim is the minimum area of a child for which a separate thread will
be forked to reshape or repaint it; if it is-1 , it is set to an appropriate default (see the
VBTTuning interface).

PROCEDURE New(
bg: VBT.T := NIL;
saveBits := FALSE;
parlim: INTEGER := -1)
: T; < * LL <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

4.2.1 Inserting children

The defaultSplit.Insert call is rarely useful for aZSplit : it inserts the new child
at the parent’s northwest corner, unmapped.Split.AddChild is even less useful,
since it adds children as the background, which is almost certainly not what you want.
The following procedures are more useful for inserting children into aZSplit :

PROCEDURE InsertAfter(
v: T;
pred, ch: VBT.T;
READONLY dom: Rect.T;
alsoMap: BOOLEAN := TRUE) RAISES {Split.NotAChild};

<* LL.sup = VBT.mu * >

Insertch as a new child ofv with domaindom, and markv for redisplay.

The new child is inserted immediately after (that is, below)pred ; if pred=NIL the
new child is inserted first (that is, on top). If the height or width ofdomdoes not satisfy
ch ’s size contraints, then the height and width of the child are projected into range; its

offset is preserved. This is a checked runtime error ifch is not detached. IfalsoMap
is TRUE, ch is mapped, otherwise it is unmapped.

It is occasionally useful to insert a new child below all existing children except the
background, in which case the following procedure is handy:

TYPE Altitude = {Top, Bot};

PROCEDURE Insert(
v: T;
ch: VBT.T;
READONLY dom: Rect.T;
alt := Altitude.Top;
alsoMap: BOOLEAN := TRUE); < * LL.sup = VBT.mu * >

Insert ch at the top ifalt = Altitude.Top ; insert ch just above the
background ifalt = Altitude.Bot .

That is,Insert is equivalent to

IF alt = Altitude.Top THEN
pred := NIL

ELSE
pred := Split.Pred(v, Split.Pred(v, NIL))

END;
InsertAfter(v, pred, ch, dom, alsoMap)

Finally, instead of providing the new child’s domain it can be useful to provide only
the northwest corner and let the child’s domain be determined by its shape constraints:

PROCEDURE InsertAt(
v: T;
ch: VBT.T;
nw: Point.T;
alt := Altitude.Top;
alsoMap: BOOLEAN := TRUE); < * LL.sup = VBT.mu * >

Insertch with its preferred shape and its northwest corner atnw. Thealt and
alsoMap parameters are interpreted as inInsert .

4.2.2 Moving, lifting, and lowering children

PROCEDURE Move(ch: VBT.T; READONLY dom: Rect.T);
<* LL.sup = VBT.mu * >

Change the domain ofch to bedomand markch ’s parent for redisplay.

If the height or width ofdomdo not satisfych ’s size constraints, then they are projected
into range, preserving the northwest corner ofdom. The stacking order ofch is not
changed.Move is a checked runtime error ifch ’s parent is not aZSplit . Note that
this has nothing to do withSplit.Move , unlike the next procedure.

PROCEDURE Lift(ch: VBT.T; alt := Altitude.Top);
<* LL.sup = VBT.mu * >

Lift ch to the top or lower it to be just above the background, depending onalt .
Lift is equivalent to:

v := VBT.Parent(ch);
IF alt = Altitude.Top THEN

pred := NIL
ELSE

pred := Split.Pred(v, Split.Pred(v, NIL))
END;
Split.Move(v, pred, ch)

4.2.3 Mapping and unmapping children

You can unmapa child of a ZSplit , which reshapes the child to be empty after
recording the child’s shape and offset. When you latermap the child, the recorded
shape and offset are restored. An unmapped child is rescreened when the parent is
rescreened, and its recorded shape and offset are updated when the parent is reshaped,
just like the domains of the mapped children.

PROCEDURE Unmap(ch: VBT.T); < * LL.sup = VBT.mu * >

If ch is mapped, unmap it and mark its parent for redisplay.

PROCEDURE Map(ch: VBT.T); < * LL.sup = VBT.mu * >

If ch is unmapped, map it and mark its parent for redisplay.

PROCEDURE IsMapped(ch: VBT.T): BOOLEAN;
<* LL.sup = VBT.mu * >

ReturnTRUEif ch is mapped andFALSEif ch is unmapped.

Map,Unmap, andIsMapped are checked runtime errors ifch ’s parent is not aZSplit .

4.2.4 Getting domains

PROCEDURE GetDomain(ch: VBT.T): Rect.T;
<* LL.sup = VBT.mu * >

Return the effective domain ofch .

The effective domain is the same as the normal domain, except (1) if the parent has
been marked for redisplay,GetDomain returns the domain thatch will receive when
the redisplay happens, or (2) if the domain of the parent isRect.Empty ,GetDomain
returns the domainch would receive if the parent were reshaped to its lastnon-empty
domain, or (3) if the child is unmapped,GetDomain returns the domain the child
would have if it were mapped.

GetDomain is a checked runtime error if the parent ofch is not aZSplit .

PROCEDURE GetParentDomain(v: T): Rect.T;
<* LL.sup = VBT.mu * >

Return the last non-empty value ofv.domain , or Rect.Empty if v.domain
has always been empty.

4.2.5 Moving children when the parent is reshaped

You can supply procedures to control what happens to the children when aZSplit
is reshaped. If you don’t supply a procedure, the default behavior is as follows: the
initial background child is always reshaped to have the same domain as the parent. The
other children are reshaped so as to preserve their shape and their offsets (even if this
makes them extend outside the parent domain). The rule is different if the parent is
reshaped toRect.Empty : in this case theZSplit records its children’s shapes and
offsets and reshapes them all toRect.Empty . When theZSplit is later reshaped to
a non-empty domain, it reshapes the initial background child to have the same domain
as the parent, and restores the saved dimensions and offsets of the other children.

In the unusual case that the initial background child is deleted, subsequent
background children do not automatically inherit the special reshaping behavior of the
initial background child.

To override the default behavior, useSetReshapeControl :

PROCEDURE SetReshapeControl(
ch: VBT.T;
rc: ReshapeControl); < * LL.sup = VBT.mu * >

Set the reshape control object for the childch to berc .

TYPE ReshapeControl = OBJECT METHODS
apply(ch:VBT.T; READONLY old, new, prev: Rect.T)
: Rect.T < * LL.sup = VBT.mu.ch * >

END;

SetReshapeControl arranges that whenever theZSplit parentv of ch is reshaped
from domainold to domainnew, then if the previous domain ofch is prev , the new
domain ofch will becomerc.apply(ch, old, new, prev) (if this rectangle
doesn’t satisfych ’s size constraints, its height and width will be projected into range,
preserving its offset).

These methods of theReshapeControl objects may be called concurrently for
different children. (This is why the apply method has only a share ofVBT.mu.) The
stacking order is not changed by reshaping.

When aZSplit child is replaced bySplit.Replace , the new child inherits the
old child’s reshape control object.

SetReshapeControl is a checked runtime error if the parent ofch is not a
ZSplit .

If the ZSplit is reshaped toRect.Empty , it will reshape its children to
Rect.Empty without calling their reshape control methods. Similarly, if the parent is
subsequently reshaped to its original rectangle, it will restore the children’s previous
domains without calling the methods.

One useful reshape control method provided by this interface isChainReshape ,
in which some set of the child’s west, east, north, and south edges are “chained” to the
corresponding edges of the parent. Chaining an edge means that the distance between
the child edge and the corresponding parent edge will be preserved. For example, if
both the west and east edges are chained, then the child’s horizontal extent will be
inset into the parent’s horizontal extent by fixed amounts on both sides. For another
example, suppose that the the east edge is chained and the west edge is not. In this
case the distance between the east edges of the child and parent will be preserved, but
the west edge of the child will move so as to preserve the width of the child. The north
and south edges control the vertical extent in a similar manner.

TYPE
Ch = {W, E, N, S};
ChainSet = SET OF Ch;
ChainReshapeControl = ReshapeControl OBJECT

chains: ChainSet
OVERRIDES

apply := ChainedReshape
END;

VAR (* CONST* )
NoChains, WChains, EChains, WEChains, NChains,
WNChains, ENChains, WENChains, SChains,
WSChains, ESChains, WESChains, NSChains,
WNSChains, ENSChains, WENSChains: ChainReshapeControl;

The “variables” above are constants for the following reshape control objects:

NEW(ChainReshapeControl, chains := ChainSet{}),

NEW(ChainReshapeControl, chains := ChainSet{Ch.W}),

...

NEW(ChainReshapeControl,

chains := ChainSet{Ch.W,Ch.E,Ch.N,Ch.S})

PROCEDURE ChainedReshape(
self: ChainReshapeControl;
ch: VBT.T;
READONLY oldParentDomain, newParentDomain,

oldChildDomain: Rect.T): Rect.T;

Return the rectangle that results from chaining each edge inself.chains
to the corresponding edge of the parent domain, and leaving the other edges
unconstrained.

If both edges in a dimension are chained, the offset and extent of the child will both
vary to satisfy the chain constraints; if one edge is chained, the offset will vary and the
extent will be fixed; if both edges are unchained, the offset and the extent will both be
fixed.

For example, the default behavior for the initial background child isWENSChains,
and the default behavior for all other children isWNChains.

One final reshape control method is sometimes useful:

PROCEDURE ScaledReshape(
self: ReshapeControl;
ch: VBT.T;
READONLY oldParentDomain, newParentDomain,

oldChildDomain: Rect.T) : Rect.T;

Return the integer approximation to the rectangle that results from scaling the
old child domain to occupy the same relative position of the changing parent
domain.

VAR (* CONST* ) Scaled: ReshapeControl;

This “variable” is really a constant for the following reshape control object:

NEW(ReshapeControl, apply := ScaledReshape)

END ZSplit.

4.3 The HVSplit interface

An HVSplit.T is a parent window that splits its screen into a row or column of child
windows, depending on theaxisof the split.

If the axis is horizontal,Split.Succ enumerates the children from west to east;
if the axis is vertical, it enumerates them from north to south.

An HVSplit can beadjustableor unadjustable, a property that affects the way its
space is divided between its children.

Thesizeof a child is the extent of its domain in the axis of split, thecross-sizeis
its extent in the other axis. For example, for a vertical split, a child’s size is its height
and its cross-size is its width.

The children of anHVSplit all have the same cross-size as the parent. To
determine the sizes of the children, theHVSplit begins by computing the range of
desirable sizes and the preferred size for each child by calling its shape method, passing
the method the cross-size, so that, for example, the height of a child of a vertical split
can depend on its width. At this point there are several cases.

If the sum of the minimum sizes of the children is greater than the size of the parent,
then the split is said to beoverfull. In this case the children are considered in order and
given their minimum sizes, as long as there is room. The first child that doesn’t fit is
given all the space that’s left, and the remaining children are given size zero.

If the split is not overfull, then the children are stretched according to the TeX model
of boxes and glue. The details depend on whether the split is adjustable or unadjustable.
For an adjustable split,each child’sstretchabilityis its maximum desirable size minus
its current size, and itsshrinkability is its current size minus its minimum desirable
size. If the size of the parent is increased by some amountX, then the sizes of the
children are increased by amounts that total toX and are proportional to the children’s
stretchabilities. Similarly, if the size of the parent is decreased by some amountX, then
the sizes of the children are decreased by amounts that total toX and are proportional
to the children’s shrinkabilities.

For a non-adjustable split, all the children’s sizes are first set to their preferred
sizes, and then they are stretched or shrunk the same as an adjustable split. Thus for a
non-adjustable spliteach redistribution of space depends only on the children’s shape
methods, not on their current sizes.

A non-adjustable split is best if the layout can be controlled purely by stretchability
and shrinkability. If the layout is also changed under user or program control, an
adjustable split is required. For example, in a column of editable text windows, you
should make the vertical split adjustable, since if the user makes one window big, and
then the parent changes size slightly, you do not want the big window child to snap
back to being small. On the other hand if you are using a horizontal split to center a
ButtonVBT between two stretchyTextureVBTs , you should make it unadjustable,
since in this case you always want to compute the division of space from the children’s
original shapes.

If the sum of the maximum sizes of the children is less than the size of the parent,
the split is said to beunderfull. There are no special rules for the underfull case: the
TeX stretching algorithm is used without change. This produces a state in which the
children are stretched larger than their maximum sizes.

A split is infeasibleif it is overfull or underfull, andfeasibleotherwise.
The shape of anHVSplit is computed as follows: its maximum, minimum, and

preferred sizes are obtained by adding up the corresponding values of its children.
The cross-size range is the intersection of the cross-size ranges of its children (if
this intersection is empty, the children’s maximum cross-sizes are increased until the
intersection is non-empty). The preferred cross-size ofv is the maximum of the
preferred cross-sizes of its children, projected intov ’s cross-size range.

INTERFACE HVSplit;

IMPORT VBT, Split, Axis, Rect, Interval;

TYPE
T <: Public;
Private <: Split.T;

Public = Private OBJECT METHODS
<* LL.sup <= VBT.mu * >
init(hv: Axis.T;

saveBits := FALSE;
parlim := -1;
adjustable := TRUE): T

END;

The callv.init(...) initializesv as anHVSplit with axishv and no children.
If saveBits is TRUE, the implementation will try to save the children’s old bits

when reshaping; if the children don’t use them anyway, it is faster to letsaveBits
default toFALSE. The value ofparlim is the minimum area of a child for which
a separate thread will be forked toreshape or repaint ; if it is -1 , it is set to an
appropriate default (see theVBTTuning interface).

PROCEDURE New(
hv: Axis.T;
saveBits := FALSE;
parlim := -1;
adjustable := TRUE): T;

<* LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE AxisOf(v: T): Axis.T;
<* LL.sup = VBT.mu * >

Return the axis ofv .

4.3.1 Inserting children

See theSplit interface to insert and reorder children.

PROCEDURE Cons(
hv: Axis.T;
ch0, ch1, ch2, ch3, ch4,

ch5, ch6, ch7, ch8, ch9: VBT.T := NIL;
saveBits := FALSE;
parlim := -1;
adjustable := TRUE): T; < * LL.sup = VBT.mu * >

Create anHVSplit with axishv and childrench0 , ch1 , ....

Cons is equivalent to the following:

result := New(hv, saveBits, parlim, adjustable);
Split.AddChild(result, ch0, ch1, ..., ch9);
RETURN result

PROCEDURE ConsArray(
hv: Axis.T;
READONLY ch: ARRAY OF VBT.T;
saveBits := FALSE;
parlim := -1;
adjustable := TRUE): T; < * LL.sup = VBT.mu * >

Create anHVSplit with axishv and childrench[0] , ch[1] , ....

ConsArray ignores anyNILs in the arraych . It is equivalent to:

VAR result := New(hv, saveBits, parlim, adjustable);
BEGIN

Split.AddChildArray(result, ch);
RETURN result

END

4.3.2 Adjusting the division of space

Thedivision point after a childis the sum of the sizes of all children up to and including
the child.

PROCEDURE Adjust(v: T; ch: VBT.T; totsz: INTEGER)
RAISES {Split.NotAChild}; < * LL.sup = VBT.mu * >

Change the sizes of the children ofv so that the division point afterch is as
close tototsz as possible, and markv for redisplay.

Adjust respects the size constraints on the children, and resizes children near the
division point in preference to children far from the division point. For example, a
sufficiently small adjustment will be made by resizing onlych and its successor. An
adjustment large enough to make one of these children reach its max or min size will
also resize the neighbor of that child, and so forth.

Adjust is a no-op if the split is infeasible or non-adjustable.

PROCEDURE FeasibleRange(v: T; ch: VBT.T): Interval.T
RAISES {Split.NotAChild}; < * LL.sup = VBT.mu * >

Return the interval of feasible positions for the division point afterch .

PROCEDURE AvailSize(v: T): CARDINAL;
<* LL.sup = VBT.mu * >

Return the largest size of a child that can be inserted intov without makingv
infeasible.

If the split is infeasible,AvailSize returns 0 andFeasibleRange returns the empty
interval. Both procedures assume the total size available is the total of all child sizes.

END HVSplit.

4.4 The PackSplit interface

A PackSplit.T is a parent window whose children are packed into multiple rows or
columns, depending on theaxisof the split.

If the axis is horizontal, the children are packed into rows from west to east, moving
south to a new row when the current row fills up. This is the normal style used in
placing words in a paragraph.

If the axis is vertical, the children are packed into columns from north to south,
moving east to a new column when the current column fills up. This is the normal style
used in placing paragraphs in a newspaper article.

A PackSplit always gives its children their preferred height and width, even if
this makes them extend outside the parent domain (in which case they will be clipped).

If the axis is horizontal, the children in any given row have their north edges
aligned, and all children that are first in their row have their west edges aligned with the
west edge of the parent. A child will be horizontally clipped if its requested horizontal
size exceeds the parent’s horizontal size; in this case the child will be alone in its row.

If the axis is vertical, the children in any given column have their west edges
aligned, and all children that are first in their column have their north edge aligned with
the north edge of the parent. A child will be vertically clipped if its requested vertical
size exceeds the parent’s vertical size; in this case the child will be alone in its column.

Thesizeof a window is the extent of its domain in the axis of thePackSplit ; its
cross-sizeis its extent in the other axis.

The range of desirable sizes and the preferred size of aPackSplit are just the
default for aVBT. The shape method uses the size to determine the cross-size that is
just large enough to pack in all the children at their preferred sizes, and returns as its
range of desirable cross-sizes a singleton interval containing only this cross-size.

INTERFACE PackSplit;

IMPORT VBT, PaintOp, Pixmap, Axis;

TYPE
T <: Public;
Private <: VBT.Split;
Public = Private OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(hv := Axis.T.Hor;

hgap, vgap := 1.5;
txt: Pixmap.T := Pixmap.Solid;
op: PaintOp.T := PaintOp.Bg;
nwAlign := FALSE;
saveBits := FALSE): T

END;

The callv.init(...) initializesv as an empty packsplit with axishv .

For a horizontalPackSplit , hgap is the gap to leave between children in each row;
vgap is the gap to leave between rows. For a verticalPackSplit , vgap is the gap
to leave between children in each column;hgap is the gap to leave between columns.
The gaps are specified in millimeters.

The area not covered by children is painted using the painting operationop and the
texturetxt+delta , wheredelta is the origin unlessnwAlign is set toTRUE, in
which casedelta will be set to the northwest corner ofv .

If saveBits is TRUE, the implementation will try to save the children’s old
bits when reshaping; if the children don’t use the old bits, it is more efficient to let
saveBits default toFALSE.

PROCEDURE New(
hv := Axis.T.Hor;
hgap, vgap := 1.5;
txt: Pixmap.T := Pixmap.Solid;
op: PaintOp.T := PaintOp.Bg;
nwAlign := FALSE;
saveBits := FALSE): T; < * LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE Set(
v: T;
txt: Pixmap.T;
op: PaintOp.T := PaintOp.BgFg;
nwAlign := FALSE); < * LL.sup = VBT.mu * >

Change the texture displayed byv and markv for redisplay.

PROCEDURE Get(
v: T;
VAR txt: Pixmap.T;
VAR op: PaintOp.T;
VAR nwAlign: BOOLEAN
); < * LL.sup = VBT.mu * >

Fetch the texture displayed byv .

PROCEDURE AxisOf(v: T): Axis.T; < * LL.sup <= VBT.mu * >

Return the axis ofv .

PROCEDURE HGap(v: T): REAL; < * LL.sup <= VBT.mu * >

Return thehgap of v .

PROCEDURE VGap(v: T): REAL; < * LL.sup <= VBT.mu * >

Return thevgap of v .

END PackSplit.

4.5 The TSplit interface

A TSplit.T is a parent window that giving its entire screen to one child at a time.
The child being displayed is called thecurrent child. The current child can beNIL , in
which case theTSplit ignores all events.

INTERFACE TSplit;

IMPORT VBT, Split;

TYPE
T <: Public;
Private <: Split.T;
Public = Private OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(fickle := TRUE): T

END;

The callv.init(fickle) initialize v as an emptyTSplit .
If fickle is TRUE, then the shape ofv will be the shape of its current child, or

a VBT’s default shape if the current child isNIL . If fickle is FALSE, then in each
axis the size range ofv will be the intersection of the size ranges of its children (if
this intersection is empty, the children’s maxsizes are increased until the intersection is
non-empty). The preferred size ofv is the the maximum of the preferred sizes of its
children, projected intov ’s size range. Ifv has no children, its shape is aVBT’s default
shape.

PROCEDURE SetCurrent(v: T; ch: VBT.T)
RAISES {Split.NotAChild}; < * LL.sup = VBT.mu * >

Set the current child ofv to bech and markv for redisplay.

PROCEDURE GetCurrent(v: T): VBT.T; < * LL.sup = VBT.mu * >

Return the current child ofv .

PROCEDURE Cons(ch0, ch1, ch2, ch3, ch4: VBT.T := NIL;
fickle := TRUE): T; < * LL.sup = VBT.mu * >

Create aTSplit with childrench0 , ch1 , ....

Cons is equivalent to

v := NEW(T).init(fickle);
Split.AddChild(v, ch0, ch1, ch2, ch3, ch4);
IF ch0 # NIL THEN SetCurrent(v, ch0) END;
RETURN v

END TSplit.

5 Filters

5.1 The Filter interface

A Filter.T is aSplit.T with at most one child.

INTERFACE Filter;

IMPORT Split, VBT;

TYPE
T <: Public;
Public = Split.T OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(ch: VBT.T): T

END;

The callv.init(ch) initializesv as a filter with childch and returnsv .

Split.Move on a filter is a noop.Split.Insert replaces the child, if any, and
detaches it.

PROCEDURE Child(v: T): VBT.T;
<* LL.sup = VBT.mu * >

Return the child ofv , or NIL if there is no child.

Filter.Child(v) is equivalent toSplit.Succ(v, NIL) .

PROCEDURE Replace(v: T; ch: VBT.T): VBT.T;
<* LL.sup = VBT.mu * >

Replacev ’s child by ch , detach and returnv ’s old child, and markv for
redisplay.

Filter.Replace is similar toSplit.Replace , except that it returns the old child
instead of taking the old child as an argument, and ifch is NIL it is similar to
Split.Delete .

END Filter.

5.2 The BorderedVBT interface

A BorderedVBT.T is a filter whose parent’s screen consists of the child’s screen
surrounded by a border. The parent’s shape is determined from the child’s shape by
adding the border size.

INTERFACE BorderedVBT;

IMPORT VBT, Filter, PaintOp, Pixmap;

TYPE
T <: Public;
Public = Filter.T OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(ch: VBT.T;

size: REAL := Default;
op: PaintOp.T := PaintOp.BgFg;
txt: Pixmap.T := Pixmap.Solid): T

END;

The callv.init(...) initializesv as aBorderedVBT with child ch and marksv
for redisplay.

The border size is given in millimeters. The border will be painted in the
untranslated texturetxt using the paint opop .

CONST Default = 0.5;

PROCEDURE New(
ch: VBT.T;
size: REAL := Default;
op: PaintOp.T := PaintOp.BgFg;
txt: Pixmap.T := Pixmap.Solid)
: T; < * LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE SetSize(v: T; newSize: REAL);
<* LL.sup = VBT.mu * >

Change the size of the border ofv to newSize millimeters and markv for
redisplay.

PROCEDURE SetColor(
v: T;
op: PaintOp.T;
txt := Pixmap.Solid);

<* LL.sup = VBT.mu * >

Change theop andtexture of v and markv for redisplay.

PROCEDURE Get(
v: T;
VAR size: REAL;
VAR op: PaintOp.T;
VAR txt: Pixmap.T); < * LL.sup = VBT.mu * >

Fetchv ’s parameters.

END BorderedVBT.

5.3 The RigidVBT interface

A RigidVBT.T is a filter whose size range is set explicitly, independently of its child’s
size range. In spite of its name, it size range does not have to be fixed to a single value.

All dimensions in this interface are specified in millimeters.

INTERFACE RigidVBT;

IMPORT VBT, Filter, Axis;

TYPE
T <: Public;
Public = Filter.T OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(ch: VBT.T; sh: Shape): T

END;

TYPE
SizeRange = RECORD lo, pref, hi: REAL END;
Shape = ARRAY Axis.T OF SizeRange;

The callv.init(...) initializesv as a rigidVBTwith child ch and shapesh .
A RigidVBT.SizeRange is like aVBT.SizeRange , but in millimeters instead

of pixels, usingREALs instead ofINTEGERs, and the range is[lo..hi] instead of
[lo..hi-1] .

PROCEDURE New(ch: VBT.T; sh: Shape): T;

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE FromHV(
ch: VBT.T;
hMin, vMin: REAL;
hMax, vMax, hPref, vPref: REAL := -1.0) : T;
<* LL.sup <= VBT.mu * >

Return aRigidVBT with child ch and the given shape.

If hMax or hPref are defaulted, they are assumed to be the same ashMin , and
similarly for vMax, vPref andvMin . That is,FromHV is equivalent to:

IF hMax = -1.0 THEN hMax := hMin END;
IF vMax = -1.0 THEN vMax := vMin END;
IF hPref = -1.0 THEN hPref := hMin END;
IF vPref = -1.0 THEN vPref := vMin END;
RETURN New(ch,

Shape{SizeRange{h, hMax, hPref},
SizeRange{v, vMax, vPref}})

END RigidVBT.



   

5.4 The HighlightVBT interface 63

5.4 The HighlightVBT interface

A HighlightVBT.T is a filter that highlights a rectangular outline over its child.
The parent screen is obtained from the child screen by texturing an outline inset in

a rectangle, using an inverting painting operation.
The parent keeps its screen correct as the child paints. Since the parent screen is

always correct, it never needs to mark itself for redisplay.

INTERFACE HighlightVBT;

IMPORT VBT, Rect, Point, Filter, Pixmap, PaintOp;

TYPE
T <: Public;
Public = Filter.T OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(ch: VBT.T;

op: PaintOp.T := PaintOp.TransparentSwap;
txt: Pixmap.T := Pixmap.Gray;
READONLY delta := Point.T{h := 0, v := 1}): T

END;

The callv.init(ch, ...) initializesv as aHighlightVBT with child ch and the
given parameters, and returnsv .

The highlight rectangle is initially empty. The filter brings up the highlight by
calling

VBT.PaintTexture(v, highlight region, op, txt, delta)

and brings down the highlight the same way; therefore the painting operation must be
its own inverse for the filter to work correctly.

The default values for the texture and delta are such that the highlightwill be visible
over white, black, or the standard gray texture. (If delta were(0,0) instead of(0,1) ,
the highlight would look fine over white or black but would be barely noticeable over
standard gray.)

PROCEDURE New(
ch: VBT.T;
op := PaintOp.TransparentSwap;
txt: Pixmap.T := Pixmap.Gray;
READONLY delta := Point.T{h := 0, v := 1} ) : T ;

<* LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE Find(v: VBT.T): T; < * LL.sup = VBT.mu * >

Return the lowest (possibly improper) ancestor ofv that is aHighlightVBT.T
or NIL if there isn’t one.

PROCEDURE SetRect(
v: VBT.T;
READONLY rect: Rect.T;
inset: CARDINAL := 2);

<* LL.sup = VBT.mu * >

Set the rectangle and inset ofFind(v) to the given values.

The inset is given in pixels, not in millimeters.

PROCEDURE SetTexture(
v: VBT.T;
txt: Pixmap.T;
READONLY delta := Point.Origin;
op := PaintOp.TransparentSwap);

<* LL.sup = VBT.mu * >

Set thetxt , delta , andop of Find(v) to the given values.

PROCEDURE Get(
v: VBT.T;
VAR rect: Rect.T;
VAR inset: CARDINAL;
VAR txt: Pixmap.T;
VAR delta: Point.T;
VAR op: PaintOp.T): BOOLEAN; < * LL.sup = VBT.mu * >

Fetch the parameters for theHighlightVBT abovev , and returnTRUE. If v
has no such ancestor, returnFALSE.

PROCEDURE Invert(v: VBT.T;
READONLY r: Rect.T;
inset: CARDINAL); < * LL.sup = VBT.mu * >

Highlight the outline inset into the rectangler with width inset , using a solid
texture.

Invert operates onFind(v) . It is equivalent to:

SetTexture(v, Pixmap.Solid);
SetRect(v, r, inset)

SetRect , SetTexture , andInvert are no-ops ifFind(v) is NIL .

END HighlightVBT.

5.5 The TranslateVBT interface

A TranslateVBT.T is a filter that maintains a translation between the coordinate
systems of the child and parent such that the child’s coordinate system has its origin



   

5.6 Buttons 65

at the northwest corner of the child domain. The child can beNIL , in which case the
TranslateVBT ignores all events.

INTERFACE TranslateVBT;

IMPORT VBT, Filter;

TYPE T <: Filter.T;

The callv.init(ch) initializesv as aTranslateVBT with child ch .

PROCEDURE New(ch: VBT.T): T; < * LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

END TranslateVBT.

5.6 Buttons

A ButtonVBT.T is a filter with an associated action procedure that is called when the
user clicks on the button or makes some other appropriate gesture.

Different subtypes ofButtonVBTs invoke the action procedure on different user
gestures, but allButtonVBTs have the three methodspre , post , andcancel . They
all interpret user gestures in such a way that the sequence of calls will be in the regular
expression

( (pre cancel) | (pre action post) ) *

The minimum, maximum, and preferred size of aButtonVBT are all equal to the
minimum size of its child, in each axis.

INTERFACE ButtonVBT;

IMPORT VBT, Filter, PackSplit, PaintOp;

TYPE
T <: Public;
Public = Filter.T OBJECT ( * CONST* )

action: Proc
METHODS

<* LL.sup = VBT.mu * >
pre();
post();
cancel();
<* LL.sup <= VBT.mu * >
init(ch: VBT.T;

action: Proc;
ref: REFANY := NIL): T;

END;

Proc =
PROCEDURE(self: T; READONLY cd: VBT.MouseRec);
<* LL.sup = VBT.mu * >

The callv.init(...) initializesv with child ch and action procaction and adds
ref to v ’s property set if it is notNIL . The action procedure can accessref (if it is
notNIL ) by callingVBT.GetProp .

The mouse and position methods of aButtonVBT.T call thepre method on a
down click, and then call thecancel method if the user chords by clicking another
mouse button or if the user moves the mouse out of the button. Otherwise they call the
action procedureproc if the user releases the mouse button.

The defaultpre method highlights the button, the defaultpost and cancel
methods unhighlight it. Consequently there should be aHighlightVBT somewhere
above the button. SinceTrestle.Install automatically inserts aHighlightVBT ,
you usually don’t have to worry about this.

The action procedure is a field rather than a method in order to allow buttons with
different action procedures to share their method suites.

PROCEDURE New(
ch: VBT.T;
action: Proc;
ref: REFANY := NIL): T; < * LL.sup = VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE MenuBar(
ch0, ch1, ch2, ch3, ch4, ch5,

ch6, ch7, ch8, ch9: VBT.T := NIL;
op: PaintOp.T := PaintOp.Bg)
: PackSplit.T; < * LL.sup = VBT.mu * >

Return aPackSplit with the given children, left-justified, and with its
background painted withop .

MenuBar is convenient for building a horizontal row of buttons. If the row fills up,
the extra buttons will wrap to the next line.

END ButtonVBT.

5.7 Quick buttons

A QuickBtnVBT.T is a button that activates immediately on down-clicks. Quick
buttons are useful for boolean toggles and radio buttons.

A QuickBtnVBT has itspre , action , andpost methods called on every mouse
click of typeFirstDown in its domain. Itscancel method is never called. Its default
pre andpost methods are no-ops.

INTERFACE QuickBtnVBT;

IMPORT ButtonVBT, VBT;

TYPE T <: ButtonVBT.T;

The callv.init(ch, action, ref) initializesv as a quick button with childch
and action procedureaction , and addsref to v ’s property set if it is notNIL .

PROCEDURE New(
ch: VBT.T;
action: ButtonVBT.Proc;
ref: REFANY := NIL): T; < * LL.sup = VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

END QuickBtnVBT.

5.8 Menu Buttons

A MenuBtnVBT.T is a button suitable for the items in pop-up and pull-down menus.
When the cursor rolls into a menu button, thepre method is called and the button

is readied. If it receives a mouse transition of type LastUp while it is readied, the
action andpost methods are called. Thecancel method is called if the cursor
leaves the button or the user chords with the mouse while the button is readied .

INTERFACE MenuBtnVBT;

IMPORT ButtonVBT, VBT;

TYPE T <: ButtonVBT.T;

The callv.init(ch, action, ref) initializesv as a menu button with childch
and action procedureaction , and addsref to v ’s property set if it is notNIL .

PROCEDURE New(
ch: VBT.T; action: ButtonVBT.Proc;
ref: REFANY := NIL): T; < * LL.sup = VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE TextItem(
name: TEXT; action: ButtonVBT.Proc;
ref: REFANY := NIL): T; < * LL.sup = VBT.mu * >

Return a menu button that displays the textname.

TextItem is a convenience procedure for making a menu button with aTextVBT
child. The borders are initialized to make the button suitable for stacking into a menu
using a verticalHVSplit . More precisely,TextItem is equivalent to:

New(TextVBT.New(name, 0.0, 0.5, 3.0, 0.5),
action, ref)

END MenuBtnVBT.

5.9 Anchor Buttons

An AnchorBtnVBT.T is a button that activates a pull-down menu when you click on
it or roll into it from another anchor button.

Associated with each anchor buttonb is

� b.menu , the menu to be activated,

� b.hfudge and b.vfudge , dimensions in millimeters that control where the
menu is popped up,

� b.n , a count of the number ofZSplit ancestors ofb to skip when looking for
theZSplit to insert the menu into.

A down click on an anchor buttonb activatesit by:

� calling the methodb.pre() , and then

� inserting the windowb.menu so that its northwest corner isb.hfudge millime-
ters to the right andb.vfudge millimeters below the southwest corner ofb. The
menu will be inserted into the (b.n )th ZSplit ancestor ofb (counting the first
ZSplit ancestor as zero), or as an undecorated top-level window ifb has at most
b.n ZSplit ancestors.

The anchor button will be deactivated when it gets another mouse transition or when
the user rolls the mouse over a sibling anchor button, in which case the sibling will
be activated. Two anchor buttons are siblings if they have the same “anchor parent”.
The anchor parent is specified when the anchor button is created; if it isNIL , then the
normal parent is used as the anchor parent. When an anchor button is deactivated, its
cancel method is called and its menu is deleted from itsZSplit .

The defaultpre method highlights the anchor button; the defaultcancel method
unhighlights it.

In the common case in which the user down-clicks on the anchor, rolls over the
menu, and up-clicks on one of the items, the upclick will be delivered to the item
first, which will invoke the appropriate action, and then will be delivered to the anchor
button (since the anchor button has the mouse focus), which will delete the menu.

A HighlightVBT is automatically inserted over the menu when it is inserted,
and discarded when the menu is deleted. This allows the menu items to highlight
themselves without interfering with the highlighting of the anchor button.

The action procedure andpost method of an anchor button are never called.
Thepre andcancel methods can be overridden; for example, thepre method could

prepare the menu before it is inserted. This is the reason the menu field is revealed in
the type declaration.

The same menu can be associated with several anchor buttons, provided that only
one of them is active at a time.

INTERFACE AnchorBtnVBT;

IMPORT ButtonVBT, VBT, ZSplit, Point;

TYPE
T <: Public;
Public = ButtonVBT.T OBJECT

menu: VBT.T
METHODS <* LL.sup <= VBT.mu * >

init(ch: VBT.T;
menu: VBT.T;
n: CARDINAL := 0;
anchorParent: VBT.T := NIL;
hfudge, vfudge := 0.0;
ref: REFANY := NIL): T

END;

The callv.init(...) initializes the button with the given attributes, and addsref
to v ’s property set if it is notNIL . This includes a call toButtonVBT.T.init(v,
ch) .

You must not change the menu while theAnchorBtnVBT is active.

PROCEDURE New(
ch: VBT.T;
menu: VBT.T;
n: CARDINAL := 0;
anchorParent: VBT.T := NIL;
hfudge, vfudge := 0.0;
ref: REFANY := NIL): T; < * LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE SetParent(v: T; p: VBT.T);
<* LL.sup = VBT.mu * >

Set the anchor parent ofv to bep. If v is active, this is a checked runtime error.

PROCEDURE GetParent(v: T): VBT.T; < * LL.sup = VBT.mu * >

Return the anchor parent ofv .

PROCEDURE Set(v: T; n: CARDINAL; hfudge, vfudge: REAL);
<* LL.sup = VBT.mu * >

Set the attributes ofv . If v is active, this is a checked runtime error.

PROCEDURE Get(v: T; VAR n: CARDINAL;
VAR hfudge, vfudge: REAL); < * LL.sup = VBT.mu * >

Fetch the attributes ofv .

PROCEDURE IsActive(v: T): BOOLEAN; < * LL.sup = VBT.mu * >

ReturnTRUEif and only if v is active.

END AnchorBtnVBT.

6 Some useful Leaf VBTs

6.1 The TextVBT interface

A TextVBT.T is aVBT that displays a text string.
The minimum size of aTextVBT is just large enough to display its text (surrounded

by any margins that were supplied when theTextVBT was created), except that if its
text is empty its minimum size is just large enough to display the text “X”. Its preferred
size is the same as its minimum size, and its maximum size is very large.

INTERFACE TextVBT;

IMPORT VBT, Font, PaintOp, Rect;

TYPE
T <: Public;
Public = VBT.Leaf OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(txt: TEXT;

halign, valign: REAL := 0.5;
hmargin: REAL := 0.5;
vmargin: REAL := 0.0;
fnt: Font.T := Font.BuiltIn;
bgFg: PaintOp.ColorQuad := NIL): T

END;

The callv.init(...) initializesv as aTextVBT that displays the texttxt in the
font fnt , and returnsv .

The text will be painted withbgFg ’s foreground; the background will be painted
with bgFg ’s background. If bgFg is NIL these default toPaintOp.Fg and
PaintOp.Bg . The text should not contain any newline characters: it will be treated
as a single line. Ifhalign = 0.0 , the west boundary of the text will be indented by
the givenhmargin (in millimeters) from the west boundary of theVBT; if halign
= 1.0 , the east boundary of the text will be inside the east boundary of theVBT by
the givenhmargin ; for other values ofhalign , the horizontal position of the text
is computed by linear interpolation. In particular,halign = 0.5 centers the text
horizontally. The vertical position is determined byvmargin andvalign in a similar
way.

Control-left-click in the text sets the source selection to be a readonly version of
the text. Thus you can copy the text out of anyTextVBT .

PROCEDURE New(
txt: TEXT;
halign, valign: REAL := 0.5;
hmargin: REAL := 0.5;
vmargin: REAL := 0.0;
fnt: Font.T := Font.BuiltIn;

bgFg: PaintOp.ColorQuad := NIL) : T;
<* LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE Put(v: T; txt: TEXT); < * LL.sup < v * >

Change the text displayed byv to betxt and markv for redisplay.

PROCEDURE Get(v: T): TEXT; < * LL.sup < v * >

Return the text displayed by v.

PROCEDURE SetFont(
v: T;
fnt: Font.T;
bgFg : PaintOp.ColorQuad := NIL);

<* LL.sup = VBT.mu * >

Setv ’s font andbgFg to the given values and markv for redisplay. IfbgFg is
defaulted,PaintOp.bgFg is used.

PROCEDURE GetFont(v: T): Font.T; < * LL.sup = VBT.mu * >

Returnv ’s font.

PROCEDURE GetQuad(v: T): PaintOp.ColorQuad;
<* LL.sup = VBT.mu * >

Returnv ’s color quad.

PROCEDURE GetTextRect(v: T): Rect.T;
<* LL.sup = VBT.mu * >

Return the current bounding rectangle ofv ’s text.

END TextVBT.

6.2 The TextureVBT interface

A TextureVBT.T is aVBTthat displays a texture, possibly colored. Its preferred and
minimum sizes are zero and its maximum size is very large, in each axis.

INTERFACE TextureVBT;

IMPORT VBT, PaintOp, Pixmap;

TYPE
T <: Public;
Public = VBT.Leaf OBJECT METHODS

<* LL.sup <= VBT.mu * >
init(op: PaintOp.T := PaintOp.BgFg;

txt: Pixmap.T := Pixmap.Solid;
nwAlign: BOOLEAN := FALSE): T

END;

The call v.init(...) initializes v as aTextureVBT displaying txt with the
painting operationop .

The domain ofv will be painted using the painting operationop and the texture
txt+delta , wheredelta is the origin unlessnwAlign is set toTRUE, in which case
delta will be set to the northwest corner ofv .

PROCEDURE New(
op: PaintOp.T := PaintOp.BgFg;
txt: Pixmap.T := Pixmap.Solid;
nwAlign: BOOLEAN := FALSE): T; < * LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

PROCEDURE Set(
v: T;
op: PaintOp.T := PaintOp.BgFg;
txt: Pixmap.T := Pixmap.Solid;
nwAlign: BOOLEAN := FALSE);

<* LL.sup = VBT.mu * >

Changev ’s texture and mark it for redisplay.

PROCEDURE Get(
v: T;
VAR op: PaintOp.T;
VAR txt: Pixmap.T;
VAR nwAlign: BOOLEAN); < * LL.sup = VBT.mu * >

Fetchv ’s texture.

END TextureVBT.

6.3 The HVBar interface

An HVBar.T is an adjustable bar that allows a user to adjust the division of space
between the children of anHVSplit .

An HVBar must be a child of anHVSplit . When the user pushes a mouse button
over the bar, the cursor changes shape and the outline of the bar is highlighted. The
highlight follows the cursor as long as the button is down. When the button comes up,
the bar callsHVSplit.Adjust to move the bar to the currently highlighted position.
If the user tries to move the bar outside the range of positions that are consistent with
the size constraints of the children of the parentHVSplit , the highlighted bar will not
follow the cursor. If the user chords while dragging, then adjusting mode is cancelled.

The bar has methods that you can override that are calledeach time the bar is
moved, or continuously during adjustment.

In order for the bar to highlight correctly, some ancestor of theHVSplit on which
it is installed must be aHighlightVBT . SinceTrestle.Install automatically
inserts aHighlightVBT over top-level windows, you usually don’t have to worry
about this.

INTERFACE HVBar;

IMPORT VBT, PaintOp, Pixmap, TextureVBT;

TYPE
T <: Public;
Public = TextureVBT.T OBJECT METHODS

<* LL = VBT.mu * >
pre(READONLY cd: VBT.MouseRec);
post(READONLY cd: VBT.MouseRec);
during(n: INTEGER);
<* LL <= VBT.mu * >
init(size: REAL := DefaultSize;

op: PaintOp.T := PaintOp.BgFg;
txt: Pixmap.T := Pixmap.Gray): T

END;

The callv.init(...) initializesv as anHVBar with the given properties and returns
v . This includes callingTextureVBT.T.init(v, op, txt) .

The argumentsize gives the number of millimeters that the bar will occupy in the
parentHVSplit .

An adjusting barb calls b.pre(cd) when it begins adjusting in response to a
mouse clickcd . It callsb.during(k) each time the mouse moves during dragging,
wherek is the coordinate that thelo (i.e., west or north) edge of the bar would move
to if dragging were stopped at that instant. Finally, the bar callsb.post(cd) when it
stops adjusting in response to an upclick or chordcd . TheHVSplit will be adjusted
(but not redisplayed) beforeb.post(cd) is called.

The defaultpre andduring methods highlight the position the bar would move
to if dragging were stopped. The defaultpost method removes the highlighting.

CONST
DefaultSize = 2.5;

PROCEDURE New(
size := DefaultSize;
op := PaintOp.BgFg;
txt := Pixmap.Gray): T; < * LL.sup <= VBT.mu * >

New(...) is equivalent toNEW(T).init(...) .

END HVBar.

7 Resources

In this section we introduce resources (painting operations, cursors, pixmaps, and
fonts). We will introduce the screen-independent forms first, then the screen-dependent
forms.

A screen-independent resource is represented as an integer, which is simply an
index into a table called a “palette”. Unless you are re-implementing Trestle over
a new window system, you can ignore the palette and treat the integers as opaque
values that serve only to distinguish one resource from another. To prevent one kind
of screen-independent resource from being confused with another, the integers are
wrapped into one-componenet records.

A few screen-independent resources arepredefined, which means that constant
integers are assigned to them in the public interface. Each interface that defines
a screen-independent resource declares a subrange typePredefined that contains
the integers that are predefined. These types will be handy when we get to the
screen-dependent resources; until then you can ignore them.

7.1 The PaintOp interface

A PaintOp.T is a screen-independent painting operation.
A painting operationop takes a source pixels and a destination pixeld and

produces a new valueop(d, s) for the destination pixel.
A painting operation that ignores the source pixel is called atint. If op is a tint, we

just writeop(d) instead ofop(d, s) . If the effect of a tint is to set the destination
pixel to some fixed value independent of its initial value, then the tint is said to be
opaque.

The locking level isLL.sup <= VBT.mu for all of the procedures in this interface.

INTERFACE PaintOp;

TYPE
T = RECORD op:INTEGER END; Predefined = [0..16];

CONST
Bg = T{0};
Fg = T{1};
Transparent = T{2};
Swap = T{3};

Copy = T{4};

Bg, Fg, Transparent , andSwapare Trestle’s four basic tints.
Bg sets the destination pixel to the screen’s background color;Fg sets it to

the screen’s foreground color;Transparent is the identity function;Swap is a
self-inverting operation that exchanges the foreground and background pixels. More



  

76 7 RESOURCES

precisely, consider a particular screentype and letbgpix andfgpix be the foreground
and background pixel for that screentype. Then for any pixeld,

Bg(d) = bgpix
Fg(d) = fgpix

Transparent(d) = d

Swap(bgpix) = fgpix
Swap(fgpix) = bgpix
Swap(Swap(d)) = d
Swap(d) # d

The operationCopy copies source to destination:

Copy(d, s) = s

Copy is not a tint, and should be used only when the source pixels are of the same
screentype as the destination pixels (for example, withVBT.Scroll , or when painting
a pixmap of the same type as the screen).

CONST
BgBg = Bg;
BgFg = T{5};
BgTransparent = T{6};
BgSwap = T{7};

FgFg = Fg;
FgBg = T{8};
FgTransparent = T{9};
FgSwap = T{10};

TransparentTransparent = Transparent;
TransparentBg = T{11};
TransparentFg = T{12};
TransparentSwap = T{13};

SwapSwap = Swap;
SwapBg = T{14};
SwapFg = T{15};
SwapTransparent = T{16};

The sixteen operations above all have names of the formXY, whereX andY are one of
the four basic tints. They are defined by the rule:

XY(dest, source) =
IF source = 0 THEN X(dest) ELSE Y(dest) END

For example,BgFg can be used to paint a one bit deep source interpreting zeros as
background and ones as foreground.

Obviously these sixteen painting operations should be used only with one-bit deep
sources. However, not all one-bit deep sources are of the same screentype: for
example, different screentypes might have different rules for representing bitmaps. To
accomodate this unfortunate fact of life, we associate with every screentypest another
screentypest.bits , which is the type of bitmap sources appropriate forst . The
depth ofst.bits is always one. If the depth ofst is one, then it is possible (but not
certain) thatst.bits = st . When using one of sixteen operations above on aVBT
with screentypest , the source must have typest.bits . You will be happy to recall
that this will be taken care of automatically if you use screen-independent bitmaps and
fonts.

Next there is a procedure for generating colored painting operations.

TYPE
Mode = {Stable, Normal, Accurate};
BW = {UseBg, UseFg, UseIntensity};

PROCEDURE FromRGB(
r, g, b: REAL;
mode := Mode.Normal;
gray := -1.0;
bw := BW.UseIntensity): T;

Return a tint that will set a pixel to the color(r,g,b) .

The valuesr , g, andb should be in the range0.0 to 1.0 ; they represent the fractions
of red, green, and blue in the desired color.

Thegray argument controls what the tint will do on a gray-scale display. Ifgray
is between zero and one, it specifies the intensity of the tint. Ifgray is defaulted to
-1 , then the tint will use the intensity of the color(r,g,b) .

Thebw argument controls what the tint will be on a monochrome display. Ifbw is
UseBg or UseFg, then the tint will beBg orFg, respectively. Ifbw isUseIntensity ,
then the tint will beFg if r , g, andb are all zero (that is, if the color is black), andBg
otherwise.

The mode argument is relevant on color and gray-scale displays. When the total
number of pixel colors desired by all of the applications that are running exceeds the
number of available colors, then some applications’ colors will change (usually in an
unpleasantly random way).

To reduce the likelihood that your color will change randomly (at the cost of
fidelity), setmode to Stable . To increase the fidelity of the pixel to the specified
intensities (at the cost of increased danger of random change), setmode to Accurate .
For example, an icon window should use stable colors; a color editor should use
accurate colors.

PROCEDURE Pair(op0, op1: T): T;

Return an operationop such thatop(d,0) = op0(d) and op(d,1) =
op1(d) .

For example,

Pair(FromRGB(1.0,1.0,1.0), FromRGB(1.0,0.0,0.0))

will paint a bitmap with zeros as white and ones as red.

PROCEDURE SwapPair(op0, op1: T): T;

Return an operation that swaps the pixels painted byop0 andop1 .

SwapPair requires thatop0 andop1 be opaque, that is, they must set the destination
to particular pixels (say,pix0 andpix1 ). Then the tintop returned bySwapPair
satisfies:

op(pix0) = pix1

op(pix1) = pix0

op(op(p)) = p for any pixel p

For example,Swap = SwapPair(Bg, Fg) .

Sometimes it is handy to collect several related painting operations into a single object:

TYPE
ColorQuad = OBJECT

bg, fg, bgFg, transparentFg: T
END;

PROCEDURE MakeColorQuad(bg, fg: T): ColorQuad;

ReturnColorQuad{bg,fg,Pair(bg,fg),Pair(Transparent,fg)} .

TYPE
ColorScheme = ColorQuad OBJECT

swap, bgTransparent, bgSwap, fgBg, fgTransparent,
fgSwap, transparentBg, transparentSwap,
swapBg, swapFg, swapTransparent: T;

END;

PROCEDURE MakeColorScheme(bg, fg: T): ColorScheme;

Return the fifteen painting operations other thanTransparent that can be
made by combiningbg , fg , andTransparent , usingSwapPair andPair .

In MakeColorQuad andMakeColorScheme , bg andfg should be tints.

VAR (* CONST* ) bgFg: ColorScheme;

This “variable” is really a constant forMakeColorScheme(Bg, Fg) .

END PaintOp.

7.2 The Cursor interface

A Cursor.T is a screen-independent specification of a cursor shape. The call
VBT.SetCursor(v, cs) sets the cursor ofv to becs .

The locking level isLL.sup <= VBT.mu for all of the procedures in this interface.

INTERFACE Cursor;

IMPORT Pixmap, Point;

TYPE T = RECORD cs: INTEGER END; Predefined = [0..2];

CONST
DontCare = T{0};
TextPointer = T{1};
NotReady = T{2};

You should setCursor.DontCare when you don’t care about the cursor shape;
Cursor.TextPointer when the cursor is to be used for editing text, andCur-
sor.NotReady to indicate that the application is not receptive to user input.

TYPE Raw = RECORD
plane1, plane2: Pixmap.Raw;
hotspot: Point.T;
color1, color2, color3: RGB;

END;
BW = {UseBg, UseFg, UseIntensity};
RGB = RECORD

r, g, b: REAL;
gray := -1.0;
bw := BW.UseIntensity

END;

A Rawrepresents a cursor with explicit offset, bitmaps, and colors.
Theplane1 andplane2 are depth-1 pixmaps. They must have the same bounding

rectangle, and the hotspot must lie within the bounding rectangle or on its east or south
edge. If the hotspot is illegal, it will be moved to the closest legal position.

The cursor’s hotspot is kept on top of the mouse’s location on the screen. The
cursor’s image tracks the mouse relative to the hotspot. For example, if the hotspot is
(0, 0), the (0, 0) bit of the cursor’s image will be located over the mouse’s location.
The remainder of the cursor will appear to the south and east.

The color of each pixel in the cursor’s image is determined from the corresponding
bits inplane1 andplane2 (p1 andp2):

p1 = 0, p2 = 0 => transparent
p1 = 0, p2 = 1 => color1
p1 = 1, p2 = 0 => color2
p1 = 1, p2 = 1 => color3

The colors for the cursor are matched as closely as possible to the selection of cursor
colors that the screentype supports. If the screentype allows only two colors for the
cursor, then the pixels that would have beencolor3 will be color1 . Thegray and
bw values control the color on gray-scale and monochrome displays,according to the
same rule used inPaintOp.FromRGB .

PROCEDURE FromRaw(READONLY r: Raw): T;

Return a cursor that looks liker on all screens.

If the screentype does not supportr ’s colors or size,FromRaw will clip or convert
colors as necessary. On a screentype that does not allow user-defined cursors, the
cursor returned byFromRawwill behave likeDontCare .

PROCEDURE FromName(READONLY names: ARRAY OF TEXT): T;

Return the first available cursor of those named in the arraynames.

The entries ofnames are cursor names as specified in theScrnCursor interface,
possibly containing wild card characters. On any particular screentype,From-
Name(names) iterates throughnames in order and returns an arbitrary match from
the first name that matches anything. If no name has any matches, it returnsDontCare .

Standard X screentypes support the cursors named inX WindowSystemby Scheifler
et. al. [5] Appendix B. Therefore, for example,

FromName(ARRAY OF TEXT{"XC_Arrow"})

returns a cursor that behaves like the X arrow cursor on X screentypes, and like
DontCare on screentypes that have no cursor namedXC_Arrow .

END Cursor.

7.3 The Pixmap interface

A Pixmap.T is a screen-independent specification of a pixmap. Many procedures
interpretPixmap.Ts as textures, by tiling the plane with translated copies of the
pixmap. There are three predefined pixmaps:

The locking level isLL.sup <= VBT.mu for all of the procedures in this interface.

INTERFACE Pixmap;

TYPE T = RECORD pm: INTEGER END; Predefined = [0..2];

CONST
Solid = T{0};
Empty = T{1};
Gray = T{2};

Solid represents a pixmap of all ones.Empty represents a pixmap of all zeros.Gray
represents a checkerboard of ones and zeros.

The domains of these pixmaps may vary from screentype to screentype, but they
will always be non-empty.

When used on a screentypest , they will have typest.bits (see thePaintOp
interface).

TYPE Raw <: ROOT;

A Pixmap.Raw represents a pixmap as a packed array of pixels. TheScrnPixmap
interface reveals the representation.

PROCEDURE FromBitmap(bits: Raw): T;

Return a pixmap that looks likebits on all screens.

FromBitmap causes a checked runtime error if the depth ofbits is not one. On a
screentypest , it will have typest.bits .

END Pixmap.

7.4 The Font interface

A Font.T is a screen-independent specification of a typeface. There is one predefined
Font.T , which yields the built-in font of the screentype.

The locking level isLL.sup <= VBT.mu for all of the procedures in this interface.

INTERFACE Font;

TYPE T = RECORD fnt: INTEGER END; Predefined = [0..0];

CONST BuiltIn = T{0};

PROCEDURE FromName(READONLY names: ARRAY OF TEXT): T;

Return the first available font of those named in the arraynames.

The entries ofnames are font names as specified in theScrnFont interface, possibly
containing wild card characters. On any particular screentype,FromName(names)
iterates throughnames in order and returns an arbitrary match from the first name that
matches anything. If no name has any matches, it returns the built-in font.

Standard X screentypes give fonts long “names” that encode their properties, so
with X it is almost always desirable to include wild-card characters in the names. For
example,

FromName(
ARRAY OF TEXT{"-* -times-medium-r- * - * - * -10?- * "})

will return a font that, on an X server containing the standard fonts, is some Times
Roman medium-weight unslanted font sized 10 to 10.9 points, and behaves like
Font.BuiltIn on any screentye that doesn’t have a font whose name matches the
pattern.

END Font.

7.5 The Palette interface

The Palette interface allows you to implement your own screen-independent re-
sources by registering a closure to produce an appropriate screen-dependent resource
for any given screentype.

INTERFACE Palette;

IMPORT VBT, ScreenType, PaintOp, Cursor, Pixmap, Font,
ScrnPaintOp, ScrnCursor, ScrnPixmap, ScrnFont;

Translating a screen-independent resource into its screen-dependent form is called
resolvingthe resource. Here are the closure types for resolving resources:

TYPE
OpClosure = OBJECT METHODS

<* LL.sup <= VBT.mu * >
apply(st: VBT.ScreenType): ScrnPaintOp.T;

END;

CursorClosure = OBJECT METHODS
<* LL.sup <= VBT.mu * >
apply(st: VBT.ScreenType): ScrnCursor.T;

END;

PixmapClosure = OBJECT METHODS
<* LL.sup <= VBT.mu * >
apply(st: VBT.ScreenType): ScrnPixmap.T;

END;

FontClosure = OBJECT METHODS
<* LL.sup <= VBT.mu * >
apply(st: VBT.ScreenType): ScrnFont.T;

END;

When anapply method is called,st # NIL . If the method returnsNIL , then some
default screen-dependent resource will be used; for example, the built-in font or the
transparent painting operation.

The following procedures produce screen-independent resources from closures:

PROCEDURE FromOpClosure(cl: OpClosure): PaintOp.T;
<* LL.sup <= VBT.mu * >

Return aPaintOp.T that behaves likecl.apply(st) onst .

PROCEDURE FromCursorClosure
(cl: CursorClosure): Cursor.T; < * LL.sup <= VBT.mu * >

Return aCursor.T that behaves likecl.apply(st) onst .

PROCEDURE FromPixmapClosure

(cl: PixmapClosure): Pixmap.T; < * LL.sup <= VBT.mu * >

Return aPixmap.T that behaves likecl.apply(st) onst .

PROCEDURE FromFontClosure(cl: FontClosure): Font.T;
<* LL.sup <= VBT.mu * >

Return aFont.T that behaves likecl.apply(st) onst .

If your apply method that resolves a resource needs to resolve some other resource, you
should use one of the following procedures to do so. In all cases,st must be non-NIL .

PROCEDURE ResolveOp(st: VBT.ScreenType; op: PaintOp.T)
: ScrnPaintOp.T;

Resolveop for st .

PROCEDURE ResolveCursor(st: VBT.ScreenType;
cursor: Cursor.T): ScrnCursor.T;

Resolvecursor for st .

PROCEDURE ResolvePixmap(st: VBT.ScreenType;
pixmap: Pixmap.T): ScrnPixmap.T;

Resolvepixmap for st .

PROCEDURE ResolveFont(st: VBT.ScreenType; font: Font.T)
: ScrnFont.T;

Resolvefont for st .

If you create a cycle of screen-independent resources each of which tries to resolve the
next resource in the cycle, then the program will deadlock.

To implement screen-independent resources, every screentype includes apalette,
which is a table of screen-dependent resources appropriate for that screentype. Most
clients don’t need to worry about the palette, but if you are implementing aVBTclass
that translates to some other window system—like X or Microsoft Windows—here is
the procedure for building the palette in the screentype for a top-level window:

PROCEDURE Init(st: VBT.ScreenType);
<* LL.sup = VBT.mu.v * >

Initialize st ’s palette, if it is not already initialized, by resolving all screen-
independent resources forst and storing the results.

END Palette.

7.6 The ScreenType interface

A ScreenType.T represents a class of screens that have a common pixel depth, a
common set of operations on the pixels, and common repositories for cursors, pixmaps,
and fonts.

When the screentype of aVBTchanges, any screen-dependent resources for the old
screentype become useless. The application must use the new screentype’soracles
to look up resources that are valid for the new screentype. This is all handled
automatically if you use screen-independent resources that are predefined or defined
by somebody else. But you will need to use this interface if you are implementing your
own screen-independent resources.

INTERFACE ScreenType;

IMPORT ScrnCursor, VBT, ScrnColorMap, ScrnFont,
ScrnPaintOp, ScrnPixmap;

TYPE T = VBT.ScreenType;

REVEAL VBT.ScreenType <: Public;

TYPE
Public = VBT.ScreenTypePublic OBJECT ( * CONST* )

bg, fg: ScrnPaintOp.Pixel;
bits: T;
op: ScrnPaintOp.Oracle;
cursor: ScrnCursor.Oracle;
pixmap: ScrnPixmap.Oracle;
font: ScrnFont.Oracle;
cmap: ScrnColorMap.Oracle;

END;

For a screentypest , the valuesst.bg andst.fg are the pixel values that represent
the user’s default background and foreground colors onst . If the screen is color-
mapped, these are appropriate for the default colormap. For applications doing simple
painting,bg is logical white andfg is logical black. Depending on the screen and user
preferences, the actual colors that the user sees might be different.

The screentypest.bits is the screentype for 1-bit deep pixmap sources for
painting on screens of typest . It is guaranteed thatst.bits.bits=st.bits ,
st.bits.fg=1 , andst.bits.bg=0 .

The oraclesst.op , st.font , st.cursor , andst.pixmap contains methods
that provide screen-dependent resources appropriate forst —for example,st.font
has a method that will look up fonts by name.

If st.cmap # NIL , st is a color-mapped screen, which means that the color of
a pixel is determined by looking up its value in a table. The color map can be either
readonly or writable.

END ScreenType.

7.7 Screen-dependent painting operations

INTERFACE ScrnPaintOp;

IMPORT TrestleComm, PaintOp;

A ScrnPaintOp.T is a painting operation that is valid for some particular screentype.
If op is aScrnPaintOp.T valid for screentypest , thenop maps a source pixel

s and destination pixeld to a result pixelop(d, s) . It will be used in a painting
operation that setsd := op(d, s) . Both d andop(d, s) have typest , ands
either has typest or st.bits . (The typest.bits is the screentype for one-bit deep
sources that can be used withst .) For example, in a copy operation,s has typest ,
while in painting a bitmap,s has typest.bits .

A ScrnPaintOp.Oracle is meaningful only as theop field of some screentype
st . It provides methods to generateScreenPaintOp.T s that are valid forst .

A tint is a paintop that is independent ofs . If op is a tint, we writeop(d) instead
of op(d, s) . (Even in the case of a tint, the type ofs must best.bits ; otherwise
the result of applying the tint is undefined.)

7.7.1 Obtaining handles from the oracle

TYPE
Pixel = INTEGER;
Oracle = Private OBJECT

METHODS
<* LL.sup <= VBT.mu * >
opaque(pix: Pixel): T

RAISES {Failure, TrestleComm.Failure};
bgfg(bg, fg: T): T

RAISES {Failure, TrestleComm.Failure};
swap(p,q: Pixel): T

RAISES {Failure, TrestleComm.Failure};
transparent(): T

RAISES {Failure, TrestleComm.Failure};
copy(): T

RAISES {Failure, TrestleComm.Failure};
builtIn(op: PaintOp.Predefined): T;

END;
Private <: ROOT;

EXCEPTION Failure;

For a screentypest , the fieldst.op is anOracle whose methods satisfy the following
specifications:

The method call

op := st.op.opaque(pix)

setsop to a tint such thatop(p) = pix for anyp. The method call

op := st.op.bgfg(bg, fg)

setsop to a tint such thatop(p, 0) = bg(p) andop(p, 1) = fg(p) , for any
p, if bg andfg are tints. The method call

op := st.op.swap(p, q)

setsop to a tint such thatop(p)=q , op(q)=p , and for anyx , op(op(x))=x . The
method call

op := st.op.transparent()

setsop to a tint such thatop(p) = p for anyp. The method call

op := st.op.copy()

setsop to a painting operation such thatop(p, q) = q for anyp andq. The method
call

st.op.builtIn(op)

returns the operation valid forst that corresponds to the predefined screen-independent
operationPaintOp.T{op} .

The exceptionFailure is raised if the screentype cannot provide the requested
painting operation. For all the methods,LL.sup <= VBT.mu .

TYPE
PlaneWiseOracle = Oracle OBJECT

METHODS <* LL.sup <= VBT.mu * >
planewise(

READONLY mask: ARRAY OF BOOLEAN;
op1, op2: T): T

RAISES {Failure, TrestleComm.Failure};
END;

If a screentype’sop oracle is aPlaneWiseOracle (which you can test with
TYPECASE), then you can use itsplanewise method to define painting operations
by their effects on each bit position of the destination pixel. Letp[i] denote biti of
pixel p. AssumingNUMBER(mask) = st.depth , the method call

op := st.op.planewise(mask, bitOps)

setsop so that ford ands of screentypest andi in [0..st.depth-1] ,

IF mask[i] THEN
op(d, s)[i] = op1(d[i], s[i])

ELSE
op(d, s)[i] = op2(d[i], s[i])

END

The method may raiseFailure if it does not support a particular combination ofop1 ,
op2 , andmask.

The convenience procedureConstructPlanewiseOp can be used to construct a
painting operation from an array of boolean functions represented by the enumeration
by BitOp :

TYPE
BitOp = {

Zero, ( * 0 * )
And, ( * dest AND src * )
NotAnd, ( * (NOT dest) AND src * )
Src, ( * src * )
AndNot, ( * dest and (NOT src) * )
Dest, ( * dest * )
Xor, ( * dest XOR src * )
Or, ( * dest OR src * )
Nor, ( * (NOT dest) AND (NOT src) * )
Equal, ( * dest XOR (NOT src) * )
Invert, ( * NOT dest * )
NotOr, ( * (NOT dest) OR src * )
NotSrc, ( * NOT src * )
OrNot, ( * dest OR (NOT src) * )
Nand, ( * (NOT dest) OR (NOT src) * )
One}; ( * 1 * )

PROCEDURE ConstructPlanewiseOp(
pwo: PlaneWiseOracle;
READONLY bitOps: ARRAY OF BitOp): T

RAISES {Failure, TrestleComm.Failure};
<* LL.sup <= VBT.mu * >

Return the painting operation that appliesbitOp[i] to planei of the source
and destination.

If NUMBER(bitOps) = st.depth then ConstructPlanewiseOp usespwo to
construct and return an operationop such that fors andd of screentypest and i in
[0 .. st.depth-1] ,

op(d, s)[i] = bitOps[i](d[i], s[i])

The procedure may raiseFailure if the screentype does not support a particular array
bitOps .

7.7.2 The handle object

TYPE
T <: Public;
Public = OBJECT id: INTEGER; pix: Pixel := -1 END;

If p is aT, thenp.id is an identifier whose interpretation depends on the screentype.
If p was created by a callst.op.opaque(pix) , thenp.pix = pix ; otherwise
p.pix = -1 .

END ScrnPaintOp.

7.8 Screen-dependent cursors

A ScrnCursor.T is a handle on a cursor shape that is valid for some particular
screentype, called theowner of the handle. Some handles have names; others are
anonymous. A named handle is valid forever. The cursor referenced by an anonymous
handle will be garbage-collected when all handles to it have been dropped.

INTERFACE ScrnCursor;

IMPORT Point, ScrnPixmap, TrestleComm, Cursor;

EXCEPTION Failure;

VAR DontCare: T;

TYPE Raw = Cursor.Raw;

See theCursor interface for the raw representation of a cursor shape as a pair of
bitmaps, color information, and hotspot offset.

7.8.1 Obtaining handles from the oracle

TYPE
Oracle = Private OBJECT ( * CONST* )

width, height: INTEGER;
METHODS

<* LL.sup <= VBT.mu * >
load(READONLY r: Raw; nm: TEXT := NIL): T

RAISES {TrestleComm.Failure};
list(pat: TEXT; maxResults: CARDINAL := 1)

: REF ARRAY OF TEXT
RAISES {TrestleComm.Failure};

lookup(name: TEXT): T RAISES {TrestleComm.Failure};
builtIn(cs: Cursor.Predefined): T;

END;
Private <: ROOT;

For a screentypest , the fieldst.cursor is anOracle that produces cursors owned
by st :

The integersst.cursor.width andst.cursor.height are the dimensions
in pixels of the largest cursor image that the screentypest supports. Larger images
will be cropped; smaller images will be padded.

The method call

st.cursor.load(r, nm)

allocates and returns a cursor handlec owned byst whose contents are equal to
r . If nm # NIL , c receives the namenm, and any cursor handle owned byst that
previously had the namenmbecomes anonymous.

The method call

st.cursor.list(pat, maxResults)

returns the names of all cursors owned byst that match the patternpat . The list
of results may be truncated to lengthmaxResults . A * matches any number of
characters and a? matches a single character.

The method call

st.cursor.lookup(name)

return the cursor handle owned byst with the given name, orNIL if no cursor has this
name.

The method call

st.cursor.builtIn(cs)

returns the screen-dependent cursor valid forst that corresponds to the predefined
screen-independent cursorCursor.T{cs} .

The locking level for all methods isLL.sup <= VBT.mu .

7.8.2 The handle object

TYPE
T <: Public;
Public = OBJECT ( * CONST* )

id: INTEGER
METHODS

<* LL.sup <= VBT.mu * >
localize(): Raw

RAISES {TrestleComm.Failure, Failure};
unload() RAISES {TrestleComm.Failure};

END;

If cs is aScrnCursor.T , thencs.id is an identifier whose interpretation depends
on the screentype that ownscs . The method callcs.localize() returns a raw
cursor equal to the one on whichcs is a handle, and the method callcs.unload()
causescs to become anonymous.

END ScrnCursor.

7.9 Screen-dependent pixmaps

A ScrnPixmap.T is a handle on a rectangular array of pixels that is valid for use on
a particular screentype, called theownerof the handle. Some handles have names;
others are anonymous. A named handle is valid forever; the pixmap referenced by an
anonymous handle will be garbage-collected when all handles to it have been dropped.

INTERFACE ScrnPixmap;

IMPORT Point, Rect, Word, TrestleComm, Pixmap;

EXCEPTION Failure;

TYPE Raw = Pixmap.Raw;

The raw representation of a pixmap is revealed at the end of this interface.

7.9.1 Obtaining handles from the oracle

TYPE
Oracle = Private OBJECT
METHODS

<* LL.sup <= VBT.mu * >
load(READONLY r: Raw; nm: TEXT := NIL): T

RAISES {TrestleComm.Failure};
list(pat: TEXT; maxResults: CARDINAL := 1)

: REF ARRAY OF TEXT RAISES {TrestleComm.Failure};
lookup(name: TEXT): T RAISES {TrestleComm.Failure};
builtIn(pm: Pixmap.Predefined): T;

END;
Private <: ROOT;

For a screentypest , the fieldst.pixmap is anOracle that produces pixmaps owned
by st .

The method callst.pixmap.load(r, nm) allocates and returns a pixmap
handlep owned byst whose contents are equal tor . The depth ofr must either be
1 or st.depth , otherwise there is a checked runtime error. Ifnm # NIL , p receives
the namenm, and any pixmap handle owned byst that previously had the namenm
becomes anonymous.

The method callst.pixmap.list(pat, maxResults) returns the names of
all pixmaps owned byst that match the patternpat . The list of results may be
truncated to lengthmaxResults . A * matches any number of characters and a?
matches any single character.

The method callst.pixmap.lookup(name) return the pixmap with the given
name, orNIL if no pixmap has this name.

Themethod callst.pixmap.builtIn(pm) returns thescreen-dependent pixmap
valid for st that corresponds to the predefined screen-independentPixmap.T{pm} .

The locking level for all methods isLL.sup <= VBT.mu .

7.9.2 The handle object

TYPE
T <: Public;
Public = OBJECT ( * CONST* )

id: INTEGER;
depth: INTEGER;
bounds: Rect.T

METHODS
<* LL.sup <= VBT.mu * >
localize(READONLY rect: Rect.T): Raw

RAISES {TrestleComm.Failure};
unload() RAISES {TrestleComm.Failure};
free() RAISES {TrestleComm.Failure}

END;

If pm is aScrnPixmap.T , thenpm.id is an identifier whose interpretation depends
on the screentype that ownspm. The fieldpm.depth is the number of bits in each
pixel of pm, andpm.bounds is the rectangular extent ofpm.

Themethod callpm.localize(rect) returns a raw pixmap equal to a rectangualr
subpixmap of the one on whichpmis a handle. The bounds of the raw pixmap returned
by localize is Rect.Meet(rect, pm.bounds) .

The method callpm.unload() causespmto become anonymous.
Pixmaps consume large amounts of memory. The method callpm.free() releases

the memory associated with the pixmap. You must make sure that allVBTs usingpm
have finished painting before you free it. After a call tofree , the pixmap bounds and
contents are arbitrary.

7.9.3 The raw representation

A raw pixmap allows the client to directly locate and modify the bits of the pixmap.
The following procedure produces a new raw pixmap:

PROCEDURE NewRaw(dpth: INTEGER;
READONLY bnds: Rect.T): Raw;

<* LL arbitrary * >

Allocate and return a raw pixmap with the given depth and bounds.

The initial contents of the pixmap returned byNewRaware undefined.
Here is the representation of a raw pixmap:

REVEAL Pixmap.Raw <: Raw_Public;

TYPE
Raw_Public = OBJECT

depth: INTEGER;

bounds: Rect.T;
pixels: REF ARRAY OF Word.T;
offset: INTEGER;
bitsPerPixel: INTEGER;
wordsPerRow: INTEGER;
pixelOrder: ByteOrder;
westRounded: INTEGER;

METHODS
get(READONLY pt: Point.T): Pixel;
set(READONLY pt: Point.T; pix: Pixel);
sub(READONLY rect: Rect.T): Raw;

END;

Pixel = Word.T;
ByteOrder = {MSBFirst, LSBFirst};

The methods provide the easiest way to operate on a raw pixmap, and we will explain
them first. Letpmbe aScrnPixmap.Raw , then:

The method call

pm.get(pt)

returns the pixel value at the pointpt in the pixmap. The result is undefined ifpt is
not inpm.bounds .

The method call

pm.set(pt, pix)

sets the pixel value at the pointpt of the pixmappm to the valuepix . It is a noop if
pt is not inpm.bounds .

The method call

pm.sub(rect)

returns a pixmap whose bounds areRect.Meet(rect, pm.bounds) and whose
contents are shared withpm’s.

It is also possible to bypass the methods andaccess the data in the raw pixmap
directly. Here is the specification for the internal layout of pixels in a raw pixmap:

A valuepmof typePixmap.Raw is a rectangular subregion of a larger rectangular
pixmap, which we shall call thesurround. The surround is a word-aligned pixmap,
stored in raster-scan order by rows. Pixels do not cross word boundaries. More
precisely, the westmost pixel in each row of the surround is always a pixel whose
h-coordinate is a multiple ofpixelsPerWord (which is equal toWord.Size DIV
pm.bitsPerPixel ). The eastmost pixel in each row of the surround is always a pixel
whoseh-coordinate modulopixelsPerWord is congruent topixelsPerWord-
1. Hence, the number of pixels in each row of the surround is a multiple of
pixelsPerWord . The valuepm.wordsPerRow is the number of words that are
needed to store one row of the surround.

The valuepm.bitsPerPixel might be greater thanpm.depth ; for example, a
twelve-bit deep pixmap might be stored with sixteen bits per pixel.

The pixels of the surround are stored in the arraypm.pixels . Each row is
represented inpm.wordsPerRow adjacent words; the first of these words stores the
westmostpixelsPerWord pixels of the row, the following word stores the adjacent
pixelsPerWord pixels, and so on until the last word, which stores the eastmost
pixelsPerWord pixels.

The order in which pixels are packed into words is indicated bypm.pixelOrder .
In this discussion, bit0 is the least significant bit and bitWord.Size - 1 is the most
significant bit of a word.

If pm.pixelOrder = LSBFirst , the bits of the pixels are as follows (where
bpp is pm.bitsPerPixel ):

pixel 0: bits 0..bpp-1
pixel 1: bits bpp..2 * bpp-1
...
pixel i : bits i * bpp..(i+1) *bpp-1

If pm.pixelOrder = MSBFirst , the pixels are stored in reverse order, so that pixel
i occupies the same bits as pixelpixelsPerWord-i-1 occupies forLSBFirst .

A Word.Extract of the bits indicated above, from the correct word, gives the
pixel’s value. If the word size does not contain an integral number of pixels, the unused
bits in the word have undefined values.

The pixmappm itself is a rectangular region selected from the surround; the value
pm.bounds , of typeRect.T , specifies the domain ofpm. The valuepm.offset
specifies where inpm.pixels the words containing the pixels ofpmcan be found. In
particular, the northwestern-most bit ofpm, the bit with coordinates

h = pm.bounds.west and v = pm.bounds.north,

is stored in wordpm.pixels[pm.offset] . The pixel is the(pm.bounds.west
MOD pixelsPerWord) ’th pixel of the word. Its bits can be found by the earlier
formulas.

The general formula for the word containing the pixel with positionh, v is

pm.pixels[
(v - pm.bounds.north) * pm.wordsPerRow +
(h - pm.westRounded) DIV pixelsPerWord) + pm.offset].

Here is another useful formula. The surround rectangle must be at least wide enough
to contain the subrectanglepm.bounds , even after we have rounded the west edge
of pm.bounds westward to the next word boundary and rounded the east edge of
pm.bounds eastward to the next word boundary. As a result, we have the inequality:

pm.wordsPerRow >=
((pm.bounds.east - 1) DIV pixelsPerWord) -
(pm.bounds.west DIV pixelsPerWord) + 1

Finally, the valuepm.westRounded is provided for convenience; it is equal to

bounds.west - (bounds.west MOD pixelsPerWord),

that is, the western boundary moved west to the nearest word boundary.

END ScrnPixmap.

7.10 Screen-dependent fonts

A ScrnFont.T is a handle on a typeface that is valid for some particular screen-
type, called theowner of the handle. All handles have names, which are highly
conventionalized strings encoding the size, style, and other properties of the typeface.

INTERFACE ScrnFont;

IMPORT ScrnPixmap, Rect, TrestleComm, Font;

EXCEPTION Failure;

7.10.1 Obtaining handles from the oracle

TYPE
Oracle = Private OBJECT

METHODS
<* LL.sup <= VBT.mu * >
list(pat: TEXT; maxResults := 1):

REF ARRAY OF TEXT RAISES {TrestleComm.Failure};
match(

family: TEXT;
pointSize: INTEGER := 120;
slant: Slant := Slant.Roman;
maxResults: CARDINAL := 1;
weightName: TEXT := AnyMatch;
version: TEXT := "";
foundry: TEXT := AnyMatch;
width: TEXT := AnyMatch;
pixelsize: INTEGER := AnyValue;
hres, vres: INTEGER := ScreenTypeResolution;
spacing: Spacing := Spacing.Any;
averageWidth: INTEGER := AnyValue;
charsetRegistry: TEXT := "ISO8859";
charsetEncoding: TEXT := "1")

: REF ARRAY OF TEXT RAISES {TrestleComm.Failure};
lookup(name: TEXT): T

RAISES {Failure, TrestleComm.Failure};
builtIn(f: Font.Predefined): T;

END;
Private <: ROOT;

For a screentypest , the field st.font is an Oracle that produces font handles
owned byst .

The method call

st.font.list(pat, maxResults)

returns the names of all fonts owned byst that match the patternpat . The list
of results may be truncated to lengthmaxResults . A * matches any number of
characters and a? matches a single character.

The arguments to thematch method specify various font attributes, as explained
below. The method call

st.font.match(...)

returns the names of all font handles owned byst that match the specifications. The
list of results may be truncated to the lengthmaxResults . If no fonts match the
specifications, the result will be eitherNIL or an empty array. PassingAnyMatch for a
text attribute, orAnyValue for an integer attribute, allows any value for that attribute.
For text attributes, partial text matches are also possible: a* matches any number of
characters and? matches a single character.

The method call

st.font.lookup(name)

returns the font handle owned byst with the given name. Generallyname should be
one of the names returned by thelist or match method.

The method call

st.font.builtIn(f)

returns the screen-dependent font valid forst that corresponds to the predefined
screen-independent fontFont.T{f} .

The locking level for all methods isLL.sup <= VBT.mu .

7.10.2 Font attributes

The arguments to a font oracle list method specify font attributes whose full speci-
fications are the “X Logical Font Description Conventions Version 1.3”, an MIT X
Consortium Standard which can be found in Part IV ofX Window Systemby Scheifler
and Gettys [5]. Here they are described in brief.

The argumentfamily specifies the family of the typeface. To find out what fonts
your X server has, run thexlsfonts program. Most servers support the families
Courier , Helvetica , andTimes , among others.

The argumentpointsize is ten times the font’s size in points; e.g., 120 for a
standard 12-point font.

The argumentslant is an element of the following enumeration type:

TYPE
Slant = {Roman, Italic, Oblique, ReverseItalic,

ReverseOblique, Other, Any};

whose elements have the following interpretations:

Roman: Upright letters in a roman style.

Italic : Clockwise slanted letters in an italic style.

Oblique : Clockwise slanted letters in a roman style.

ReverseItalic : Counter clockwise slanted letters in an italic style.

ReverseOblique : Counter clockwise slanted letters in a roman style.

Other : None of the above

Any: Any of the above (includingOther ).

The argumentweightName is the foundry’s name for the font’s weight; e.g.,
Bold , DemiBold , or Medium.

The argumentversion specifies the version of theX Logical Font Description
Conventionsthat describes the format of a font’s name. If the argument is omitted,
Version 1.3 is assumed. (Version 1.3 is the only version as these words are written.)

The argumentfoundry specifies the X registered name for the font’s foundry, e.g.,
Adobe , B&H, Bitstream ,DEC.

The argumentwidth specifies the foundry’s name for the font’s width; e.g.,
Normal or Condensed .

The argumentpixelsize specifies the size of the font in pixels. The size in points
depends on the vertical resolution of the device: A pixelsize of 20 could represent a
20-point font at 75 pixels per inch or a 10-point font at 150 pixels per inch.

The argumentshres andvres specify the horizontal and vertical screen resolution
for which the font is designed, in pixels per inch.

The argumentspacing is an element of the following enumeration:

TYPE Spacing =
{Proportional, Monospaced, CharCell, Any};

whose elements have the following meaning:

Proportional : Character widths vary.

Monospaced : Character widths are constant.

CharCell : Font is self-clearing, as defined in theVBT interface.

Any: Any of the above.

The argumentaverageWidth specifies the un-weighted arithmetic mean of the
widths of all glyphs in the font, measured in tenths of a pixel.

The argumentscharsetRegistry andcharsetEncoding are the X names of
the font’s character set and encoding scheme; e.g.,ISO8859 and1 for ISO Latin-1
fonts. See Appendix G of [5].

CONST
AnyMatch = " * ";
AnyValue = -1;
ScreenTypeResolution = -2;

PassingAnyMatch as an argument to thelist method matches any text value
for the corresponding attribute, andAnyValue matches any integer value. Passing
ScreenTypeResolution for hres or vres matches fonts whose horizontal and
vertical resolutions agree with the screentype that owns the font.

7.10.3 Registering fonts

Some screentypes allow the client to register fonts. The client registers the font’s strike
(bits) and metrics (description) with theStrikeOracle . The name of the font is
implied by the attributes in the metrics, so thelist and lookup methods will find
client-registered fonts.

TYPE
StrikeOracle = Oracle OBJECT

METHODS
<* LL.sup <= VBT.mu * >
load(strike: Strike; metrics: Metrics): T

RAISES {Failure, TrestleComm.Failure};
END;

The method callst.font.load(strike, metrics) creates a font owned byst
with the given strike and metrics and returns a handle to it.

Themetrics argument must define all of the initial fields of the font metrics record:
family , pointSize , ..., isAscii , anddefaultChar . The valuesminBounds
andmaxBounds must be provided ifcharMetrics isNIL ; otherwise ifprintWidth
is AnyValue , the load method will compute them fromcharMetrics . If any of
the remaining fields have the valueAnyValue , theload method will compute them.

7.10.4 The handle object

TYPE
T <: Public;
Public = OBJECT ( * CONST* )

id: INTEGER;
metrics: Metrics

END;



   

98 7 RESOURCES

TYPE StrikeFont = T OBJECT
METHODS <* LL.sup <= VBT.mu * >

strike(): Strike RAISES {TrestleComm.Failure}
END;

TYPE Strike = OBJECT
METHODS <* LL.sup <= VBT.mu * >

glyph(ch: INTEGER): ScrnPixmap.T;
END;

If f is aScrnFont.T , thenf.id is an identifier whose interpretation depends on the
screentype that ownsf and f.metrics are the metrics forf . If in addition f is a
StrikeFont , then f.strike() returnsf ’s strike. The screentype of the strike’s
pixmaps will be the screentype that ownsf .

If str is aStrike , thenstr.glyph(ch) is the pixmap for the characterch . This
will be empty except for characters in the range[m.firstChar..m.lastChar] ,
wheremis the metrics (see below) for the font of whichstr is the strike.

PROCEDURE BoundingBox(txt: TEXT; fnt: T): Rect.T;
<* LL arbitrary * >

Return the smallest rectangle that contains the bounding boxes of the characters
of txt if txt were painted in the fontfnt with txt ’s reference point at the
origin.

PROCEDURE BoundingBoxSub(
READONLY txt: ARRAY OF CHAR;
fnt: T): Rect.T;

<* LL arbitrary * >

Like BoundingBox but takes an array instead of aTEXT.

PROCEDURE TextWidth(txt: TEXT; fnt: T): INTEGER;
<* LL arbitrary * >

Return the sum of the printing widths of the characters intxt in the fontfnt .

7.10.5 The raw representation

TYPE
CharMetric = RECORD

printWidth: INTEGER;
boundingBox: Rect.T;

END;
CharMetrics = REF ARRAY OF CharMetric;

TheprintWidth of a character is the displacement to the next character’s reference
point.

The boundingBox of a character is the smallest rectangle with sides parallel to
the axes that contains the glyph of the character placed with its reference point at (0,0).

TYPE
Metrics = OBJECT ( * CONST* )

family: TEXT;
pointSize: INTEGER;
slant: Slant;
weightName: TEXT;
version: TEXT;
foundry: TEXT;
width: TEXT;
pixelsize: INTEGER;
hres, vres: INTEGER;
spacing: Spacing;
averageWidth: INTEGER;
charsetRegistry: TEXT;
charsetEncoding: TEXT;
firstChar, lastChar: INTEGER;
charMetrics: CharMetrics;
selfClearing: BOOLEAN;
rightKerning, leftKerning: BOOLEAN;
isAscii: BOOLEAN;
defaultChar: INTEGER;
minBounds, maxBounds: CharMetric;

METHODS <* LL arbitrary * >
intProp(name: TEXT; ch: INTEGER := -1): INTEGER

RAISES {Failure};
textProp(name: TEXT; ch: INTEGER := -1): TEXT

RAISES {Failure};
END;

The fields fromfamily to charSetEncoding in theMetrics object specify the
attributes that were defined for thelookup method. A value of* or Any in one of
these fields means that the corresponding attribute is unknown.

The integersfirstChar and lastChar are the indices of the first and last
characters defined in the font.

The arraycharMetrics specifies the metrics of the individual characters. The
metrics for characterch are incharMetrics[ch-firstChar] . If all characters
have the sameprintWidth and boundingBox , then these values are stored in
minBounds andmaxBounds and thecharMetrics field isNIL .

The flagselfClearing indicates whether the font is self-clearing, as defined
in the VBT interface, and the two kerning flags indicate the present of right and left
kerning in the font.

The flag isAscii indicates that character codes 32-126 (base 10) have their
normal ASCII meanings.

The integerdefaultChar is the code for the recommended character to display
in the place of a character that isn’t defined for the font.

The rectanglesminBounds.boundingBox and maxBounds.boundingBox
contain the meet and join, respectively, of the bounding boxes of all characters in
the font when they are positioned with their reference points at (0, 0). The values
minBounds.printWidth and maxBounds.printWidth are the minimum and
maximum printing widths for all characters in the font.

The method callm.intProp(nm) returns the integer value of the font attribute
namednm, or raisesFailure if this attribute is not defined form. The method call
m.intProp(nm, ORD(ch)) returns the integer value of the font attribute namednm
for the characterch , or raisesFailure if this attribute is not defined for(m, ch) .
The textProp method is similar.

The set of attributes returned by the metrics methods depend on the font. Fonts
that are owned by X screentypes support the attributes defined in Part IV ofX Window
System(op. cit.); we recommend that other fonts support them too. (To read an X font
attribute whose type is an X atom, use thetextProp method, which returns the name
of the atom.)

END ScrnFont.

7.11 Color maps

A ScrnColorMap.T is a handle on a colormap that is valid for some particular
screentype, called theowner of the handle. Some handles have names; others are
anonymous. A named handle is valid forever. The colormap referenced by an
anonymous handle will be garbage-collected when all handles to it have been dropped.

Every colormap has adepth; the pixel values defined by the color map are in
the range[0..(2^depth)-1] . Every color-mapped screentype defines a set of
preferredcolors that cover the spectrum reasonably densely. Some preferred colors are
designated asstable.

Clients can allocate pixels out of a color map as read-only shared entries or as
writable exclusive entries. The implementation maintains reference counts on the
read-only entries so that an entry can be freed when it is no longer allocated to any
client.

INTERFACE ScrnColorMap;

IMPORT TrestleComm;

7.11.1 Obtaining handles from the oracle

TYPE
Oracle = Private OBJECT

METHODS
<* LL.sup <= VBT.mu * >
standard(): T RAISES {TrestleComm.Failure};

new(name: TEXT := NIL; preLoaded := TRUE): T
RAISES {TrestleComm.Failure, Failure};

lookup(name: TEXT): T
RAISES {TrestleComm.Failure};

list(pat: TEXT; maxResults: CARDINAL := 1)
: REF ARRAY OF TEXT RAISES {TrestleComm.Failure}

END;
Private <: ROOT;

EXCEPTION Failure;

Every color-mapped screentypest contains a fieldst.cmap of typeOracle , which
hands out colormaps owned byst :

The method call

st.cmap.standard()

returns the default colormap owned byst . This is the colormap that a top-level
window will initially have when it is rescreened tost . Initially, the stable colors are
allocated read-only with a reference count of one.

The method call

st.cmap.new(name, preLoaded)

creates and returns a new colormap owned byst with the given name. IfpreLoaded
is true, the stable colors are initially allocated read-only; otherwise nothing is allocated
initially.

The method call

st.cmap.lookup(name)

returns the colormap owned byst with the given name, orNIL if no colormap has this
name.

The method call

st.cmap.list(pat, maxResults)

returns the names of colormaps owned byst that match the patternpat . The list
of results may be truncated to lengthmaxResults . A * matches any number of
characters and a? matches any single character.

7.11.2 The handle object

TYPE
T <: Public;
Public = OBJECT ( * CONST* )

depth: INTEGER;
readOnly: BOOLEAN;
ramp: Ramp;

METHODS

<* LL.sup <= VBT.mu * >
fromRGB(rgb: RGB; mode := Mode.Normal): Pixel

RAISES {Failure, TrestleComm.Failure};
read(VAR res: ARRAY OF Entry)

RAISES {TrestleComm.Failure};
write(READONLY new: ARRAY OF Entry)

RAISES {Failure, TrestleComm.Failure};
new(dim: CARDINAL): Cube RAISES

{Failure, TrestleComm.Failure};
free(READONLY cb: Cube)

RAISES {TrestleComm.Failure};
END;

Mode = {Stable, Normal, Accurate};
Ramp = RECORD

base: INTEGER;
last, mult: ARRAY Primary OF INTEGER;

END;
Primary = {Red, Green, Blue};
Cube = RECORD lo, hi: Pixel END;
Pixel = INTEGER;
RGB = RECORD r, g, b: REAL END;
Entry = RECORD pix: Pixel; rgb: RGB END;

The fieldcm.depth is the depth ofcm, andcm.readOnly is TRUEif cm cannot be
written. The fieldcm.ramp defines a three dimensional lattice of colors preallocated
in cm, as follows.

If cm.ramp.base is -1 , the lattice of preallocated colors is empty.
If cm.ramp.base is not-1 , then the pixel value

base + r * mult[Red] + g * mult[Green] + b * mult[Blue]

represents the color(r/last[Red], g/last[Green], b/last[Blue]) , for r
in the range[0..last[Red]] , g in the range[0..last[Green]] , andb in the
range[0..last[Blue]] .

An RGBrepresents the color with the given blend of red, green, and blue. Each
of the numbers is in the range[0.0..1.0] ; thus the triple(0.0, 0.0, 0.0)
specifies black. In case of a gray scale display, only ther component is relevant.

The method call

cm.fromRGB(rgb, mode)

extends the read-only portion ofcm with a new entry whose value is nearrgb and
returns the pixel of the new entry. If the read-only portion ofcm already contains an
entry whose value is nearrgb , that entry’s pixel is returned. Themode argument
controls how near the new entry’s value will be torgb , as follows. Ifmode is Stable ,
the new entry’s color is the nearest stable color torgb . If mode is Normal , the new
entry’s color is the nearest preferred color torgb . If mode is Accurate , the new

entry’s color is the nearest color torgb that the hardware supports. The method raises
Failure if a new entry is required but the colormap is full.

For each entrye in the arrayres , the method call

cm.read(res)

setse.rgb to the color incmof the pixele.pixel .
The method call

cm.write(new)

changes the value ofcm at p to be rgb , for each pair(p, rgb) in the arraynew,
assuming all these pixels are writable. Otherwise the method raisesFailure . The
arraynew must be sorted.

The method call

cm.new(dim)

extends the writable portion ofcm with a set of2dim new entries whose pixels form
a cube, and returns the cube. The method raisesFailure if the free entries of the
colormap do not contain a cube of the given dimension.

A Cube cb represents a set of pixels by the following rule: a pixelp is in cb if
Word.And(lo, pix) = lo andWord.Or(hi, pix) = hi .

The method callcm.free(cb) deallocates from the writable portion ofcm each
entry whose pixel is in the cubecb , assuming all of these pixels are allocated.

END ScrnColorMap.

8 Geometry interfaces

Most programs that use windows need to perform geometric calculations with integer
lattice points. Such calculations can easily become obscure and error-prone. This
section provides a set of geometry interfaces that help make them easier to read and
write.

The interfaces are namedAxis , Point , Interval , Rect , Region , Path , and
Trapezoid . The locking level is arbitrary for all procedures in these interfaces.

8.1 The Axis Interface

Axis.T.Hor and Axis.T.Ver are Trestle’s names for the horizontal and vertical
axes.Axis.Other exchangesHor andVer .

INTERFACE Axis;

TYPE T = {Hor, Ver};

CONST Other = ARRAY T OF T {T.Ver, T.Hor};

END Axis.

8.2 The Point interface

A Point.T is a pair of integers representing a position in the plane. Ifpt is a point,
thenpt.h is the distance ofpt to the right of the coordinate origin, andpt.v is the
distance ofpt below the coordinate origin. That is, thehv coordinate system is related
to the Cartesian coordinate system by the equation(h, v) = (x, -y) .

INTERFACE Point; IMPORT Axis;

TYPE T = RECORD h, v: INTEGER END;

CONST Origin = T{0, 0};

PROCEDURE Add(READONLY p, q: T): T;

ReturnT{p.h + q.h, p.v + q.v} .

PROCEDURE Sub(READONLY p, q: T): T;

ReturnT{p.h - q.h, p.v - q.v} .

PROCEDURE Minus (READONLY p: T): T;

ReturnT{-p.h, -p.v}

PROCEDURE Mul(READONLY p: T; n: INTEGER): T;

ReturnT{p.h * n, p.v * n} .

PROCEDURE Div(READONLY p: T; n: INTEGER): T;

ReturnT{p.h DIV n, p.v DIV n} .

PROCEDURE Mod(READONLY p: T; n: INTEGER): T;

ReturnT{p.h MOD n, p.v MOD n} .

PROCEDURE Scale(READONLY p: T; num, den: INTEGER): T;

ReturnDiv(Mul(p, num), den) .

PROCEDURE Min(READONLY p, q: T): T;

ReturnT{MIN(p.h, q.h), MIN(p.v, q.v)} .

PROCEDURE Max(READONLY p, q: T): T;

ReturnT{MAX(p.h, q.h), MAX(p.v, q.v)} .

PROCEDURE MoveH(READONLY p: T; dh: INTEGER): T;

ReturnT{p.h+dh, p.v} .

PROCEDURE MoveV(READONLY p: T; dv: INTEGER): T;

ReturnT{p.h, p.v+dv} .

PROCEDURE MoveHV(READONLY p: T; dh, dv: INTEGER): T;

ReturnT{p.h+dh, p.v+dv} .

PROCEDURE Transpose(READONLY p: T; ax := Axis.T.Ver): T;

If ax = Hor then returnp else returnT{p.v, p.h} .

For example,Point.Transpose(pt, ax).h is theax component ofpt .

PROCEDURE DistSquare(READONLY p, q: T): INTEGER;

Return the square of the Euclidean distance betweenp andq.

END Point.

8.3 The Interval interface

An Interval.T is a contiguous set of integers. An intervala contains an integern if

a.lo <= n AND n < a.hi

We impose the restriction that if an interval contains no integers, then it must be equal
as a record toInterval.Empty .

INTERFACE Interval;

TYPE T = RECORD lo, hi: INTEGER END;

CONST
Empty = T{0, 0};
Full = T{FIRST(INTEGER), LAST(INTEGER)};

PROCEDURE FromBounds(lo, hi: INTEGER): T;

If lo >= hi then returnEmpty , else returnT{lo, hi} .

PROCEDURE FromAbsBounds(n, m: INTEGER): T;

ReturnFromBounds(MIN(n,m), MAX(n,m)) .

PROCEDURE FromBound(lo: INTEGER; s: CARDINAL): T;

ReturnFromBounds(lo, lo+s) .

PROCEDURE FromSize(s: CARDINAL): T;

ReturnFromBounds(0, s) .

PROCEDURE Move(READONLY a: T; n: INTEGER): T;

ReturnFromBounds(a.lo+n, a.hi+n) .

PROCEDURE Inset(READONLY a: T; n: INTEGER): T;

If a is empty then returnEmpty , else returnFromBounds(a.lo + n, a.hi
- n) .

PROCEDURE Change(READONLY a: T; dlo, dhi: INTEGER): T;

If a is empty then returnEmpty , else returnFromBounds(a.lo + dlo,
a.hi + dhi) .

PROCEDURE Join(READONLY a, b: T): T;

Return the smallest interval containing botha andb.

PROCEDURE Meet(READONLY a, b: T): T;

Return the largest interval contained in both ofa andb.

PROCEDURE Project(READONLY a: T; n: INTEGER): INTEGER;

Return the element ofa that is closest ton. This is a checked runtime error ifa
is empty.

PROCEDURE Mod(n: INTEGER; READONLY a: T): INTEGER;

Return the member ofa whose distance fromn is a multiple ofSize(a) . This
is a checked runtime error ifa is empty.

PROCEDURE Size(READONLY a: T): CARDINAL;

Returna.hi - a.lo .

PROCEDURE Middle(READONLY a: T): INTEGER;

Return(a.hi + a.lo) DIV 2 .

PROCEDURE Center(READONLY a: T; n: INTEGER): T;

If a is empty then returnEmpty , else returnb such thatSize(b) = Size(a)
andMiddle(b) = n .

PROCEDURE IsEmpty(READONLY a: T): BOOLEAN;

Return whethera is empty.

PROCEDURE Member(n: INTEGER; READONLY a: T): BOOLEAN;

Return whethern is in a.

PROCEDURE Overlap(READONLY a, b: T): BOOLEAN;

Return whethera andb have any element in common.

PROCEDURE Subset(READONLY a, b: T): BOOLEAN;

Return whethera is contained inb.

END Interval.

8.4 The Rect interface

A Rect.T is a set of points lying in a rectangle with its sides parallel to the coordinate
axes. The directions of the screen are named after the compass points, with north at
the top. A rectanglerect contains a pointpt if

pt.h is in [rect.west .. rect.east - 1] AND
pt.v is in [rect.north .. rect.south - 1]

We impose the restriction that if a rectangle contains no points, then it must be equal
as a record toRect.Empty .

INTERFACE Rect;

IMPORT Axis, Interval, Point;

TYPE T = RECORD west, east, north, south: INTEGER END;

CONST
Empty = T{0,0,0,0};
Full = T{FIRST(INTEGER), LAST(INTEGER),

FIRST(INTEGER), LAST(INTEGER)};

PROCEDURE FromEdges(w, e, n, s: INTEGER): T;

If w >= e or n >= s returnEmpty , else returnT{w,e,n,s} .

PROCEDURE FromAbsEdges(h1, h2, v1, v2: INTEGER): T;

Return

FromEdges(MIN(h1,h2), MAX(h1,h2),
MIN(v1,v2), MAX(v1,v2))

PROCEDURE FromCorners(READONLY p, q: Point.T): T;

ReturnFromAbsEdges(p.h, q.h, p.v, q.v) .

PROCEDURE FromCorner(
READONLY p: Point.T;
hor, ver: CARDINAL): T;

ReturnFromEdges(p.h, p.h+hor, p.v, p.v+ver) .

PROCEDURE FromIntervals
(READONLY hor, ver: Interval.T): T;

ReturnFromEdges(hor.lo, hor.hi, ver.lo, ver.hi) .

PROCEDURE FromPoint(READONLY p: Point.T): T;

Return the rectangle whose only element isp.

PROCEDURE FromSize(hor, ver: CARDINAL): T;

ReturnFromCorner(Point.Origin, hor, ver) .

PROCEDURE Add(READONLY r: T; READONLY p: Point.T): T;

Return

FromEdges(r.west+p.h, r.east+p.h,
r.north+p.v, r.south+p.v)

PROCEDURE Sub(READONLY r: T; READONLY p: Point.T): T;

ReturnAdd(r, Point.Minus(p)) .

PROCEDURE Change
(READONLY r: T; dw,de,dn,ds: INTEGER): T;

If r is empty returnEmpty , else return the rectangleFromEdges(r.west+dw,
r.east+de, r.north+dn, r.south+ds) .

PROCEDURE Inset(READONLY r: T; n: INTEGER): T;

ReturnChange(r, n, -n, n, -n) .

PROCEDURE Transpose(READONLY r: T; ax := Axis.T.Ver): T;

If r is empty or ifax = Axis.Hor , then returnr , else returnT{r.north,
r.south, r.west, r.east} .

PROCEDURE Join(READONLY r, s: T): T;

Return the smallest rectangle containing bothr ands .

PROCEDURE Meet(READONLY r, s: T): T;

Return the largest rectangle contained in bothr ands .

PROCEDURE HorSize(READONLY r: T): CARDINAL;

Returnr.east - r.west .

PROCEDURE VerSize(READONLY r: T): CARDINAL;

Returnr.south - r.north .

PROCEDURE Middle(READONLY r: T): Point.T;

Return Point.T{(r.west+r.east) DIV 2, (r.north+r.south)
DIV 2} .

PROCEDURE Center(READONLY r: T; READONLY p: Point.T): T;

If r is empty then returnEmpty else return a rectangles such that
Congruent(r, s) andMiddle(s) = p .

PROCEDURE NorthWest(READONLY r: T): Point.T;

ReturnPoint.T{r.west,r.north} .

PROCEDURE NorthEast(READONLY r: T): Point.T;

ReturnPoint.T{r.east,r.north} .

PROCEDURE SouthWest(READONLY r: T): Point.T;

ReturnPoint.T{r.west,r.south} .

PROCEDURE SouthEast(READONLY r: T): Point.T;

ReturnPoint.T{r.east,r.south} .

PROCEDURE Project(READONLY r: T;
READONLY p: Point.T): Point.T;

Return the element ofr that is closest top. This is a checked runtime error ifr
is empty.

TYPE Partition = ARRAY [0..4] OF T;

PROCEDURE Factor(
READONLY r, s: T;
VAR (* out * ) f: Partition;
dh, dv: INTEGER) ;

Partitionr into 5 piecesf[0]..f[4] wheref[2] = Meet(r,s) , and the
other rectangles inf partition the set differencer-s .

The order off is such that ifi<j then f[i] translated by any positive multiple of
(dh,dv) doesn’t intersectf[j] . (Only the signs ofdh anddv affect the order, not
their magnitude.)

PROCEDURE Mod(READONLY p: Point.T;
READONLY r: T): Point.T;

Return the element ofr whose distance fromp in each axis is a multiple of the
size ofr in that axis. This is a checked runtime error ifr is empty.

PROCEDURE IsEmpty(READONLY r: T): BOOLEAN;

Return whetherr is empty.

PROCEDURE Member(READONLY p: Point.T;
READONLY r: T): BOOLEAN;

Return whether p is in r.

PROCEDURE Overlap(READONLY r, s: T): BOOLEAN;

Return whetherr ands have any element in common.

PROCEDURE Subset(READONLY r, s: T): BOOLEAN;

Return whetherr is contained ins .

PROCEDURE Congruent(READONLY r, s: T): BOOLEAN;

Return whetherr ands are congruent, that is, whether they have the same height
and width.

END Rect.

8.5 The Region interface

A Region.T represents a set of integer lattice points.

INTERFACE Region;

IMPORT Rect, Point, Axis;

TYPE
T = RECORD r: Rect.T; p: P := NIL END;

P <: REFANY;

If rg is a region, thenrg.r is the smallest rectangle containing all points inrg , and
rg.p is the private representation of the region as a sorted array of disjoint rectangles.

CONST
Empty = T{Rect.Empty, NIL};

Full = T{Rect.Full, NIL};

PROCEDURE FromRect(READONLY r: Rect.T): T;

Return the region containing the same points asr .

PROCEDURE FromRects(READONLY ra: ARRAY OF Rect.T): T;

Return the region containing all points in any rectangle ofra .

PROCEDURE ToRects(READONLY rg: T): REF ARRAY OF Rect.T;

Returns a list of disjoint rectangles that partitionrg .

The callToRects(Empty) produces an array of length zero.

PROCEDURE FromPoint(READONLY p: Point.T): T;

Return the region containing exactly the pointp.

PROCEDURE BoundingBox(READONLY rg: T): Rect.T;

Return the smallest rectangle containing all the points ofrg ; this is equivalent
to rg.r .

PROCEDURE Add(READONLY rg: T; READONLY p: Point.T): T;

Return the translation ofrg by p.

That is,Add(rg, p) containspt if and only if rg containsPoint.Sub(pt, p) .

PROCEDURE Sub(READONLY rg: T; READONLY p: Point.T): T;

ReturnAdd(rg, Point.Minus(p)) .

PROCEDURE AddHV(READONLY rg: T; dh, dv: INTEGER): T;

ReturnAdd(rg, Point.T{dh,dv}) .

PROCEDURE Inset(READONLY rg: T; n: INTEGER): T;

Return the region inset intorg by n.

That is, if n is non-negative,Inset(rg, n) contains a pointpt if all points within
distancen of pt are contained inrg . If n is non-positive,Inset(rg, n) contains
a point pt if some point within distance-n of pt is in rg . For the purposes of
this definition, pointsp and q are “within distancen” if both ABS(p.h-q.h) and
ABS(p.v-q.v) are at mostn. (If n is zero, both definitions giveInset(rg, n) =
rg .)

PROCEDURE PlaceAxis(READONLY rg: T;
n: INTEGER; hv: Axis.T): T;

Return the retraction ofrg by n along thehv axis.

That is, letrect equalRect.FromSize(1, ABS(n)) if hv is Axis.T.Ver or
Rect.FromSize(ABS(n), 1) if hv is Axis.T.Hor . If n is non-negative, then

PlaceAxis(rg, n, hv) contains a pointpt if the rectangleRect.Add(pt,
rect) is contained inrg . If n is negative, thenPlaceAxis(rg, n, hv) contains
a pointpt if Rect.Add(pt, rect) contains some point inrg .

PROCEDURE Place(READONLY rg: T; h, v: INTEGER): T;

Return the retraction ofrg by h along the horizontal axis and byv along the
vertical axis.

More precisely,Place(rg, h, v) is defined by the expression

PlaceAxis(PlaceAxis(rg, h, Axis.T.Hor), v, Axis.T.Ver)

PROCEDURE Join(READONLY rg, rgP: T): T;

Return the union of the points inrg andrgP .

PROCEDURE JoinRect(READONLY r: Rect.T;
READONLY rg: T): T;

Return the union of the points inr andrg .

PROCEDURE JoinRegions(READONLY rg: REF ARRAY OF T): T;

Return the union of all the regions inrg .

PROCEDURE Meet(READONLY rg, rgP: T): T;

Return the intersection ofrg andrgP .

PROCEDURE MeetRect(READONLY r: Rect.T;
READONLY rg: T): T;

Return the intersection of the points inr andrg .

PROCEDURE Difference(READONLY rg, rgP: T): T;

Return the set of points inrg and not inrgP .

PROCEDURE SymmetricDifference(READONLY rg, rgP: T): T;

Return the set of points in exactly one ofrg andrgP .

PROCEDURE MaxSubset(READONLY r: Rect.T;
READONLY rg: T): Rect.T;

Return a large rectangular subset ofrg containingr , or returnEmpty if r is not
a subset ofrg .

PROCEDURE Equal(READONLY rg, rgP: T): BOOLEAN;

Return whetherrg andrgP contain the same points.

PROCEDURE IsEmpty(READONLY rg: T): BOOLEAN;

Return whetherrg is empty.

PROCEDURE IsRect(READONLY rg: T): BOOLEAN;

Return whetherrg is a rectangle, that is, whether it contains all the points in its
bounding box.

PROCEDURE Member(READONLY p: Point.T;
READONLY rg: T): BOOLEAN;

Return whetherp is in rg .

PROCEDURE SubsetRect(READONLY r: Rect.T;
READONLY rg: T): BOOLEAN;

Return whetherr is contained inrg .

PROCEDURE Subset(READONLY rg, rgP: T): BOOLEAN;

Return whetherrg is contained inrgP .

PROCEDURE OverlapRect(READONLY r: Rect.T;
READONLY rg: T): BOOLEAN;

Return whetherr andrg have any point in common.

PROCEDURE Overlap(READONLY rg, rgP: T): BOOLEAN;

Return whetherrg andrgP have any point in common.

END Region.

8.6 The Path interface

A Path.T is a sequence of straight and curved line segments, suitable for stroking or
filling.

A segmentis a directed arc in the Cartesian plane determined by two cubic
polynomialsh(t) , v(t) , wheret ranges over the interval of real numbers[0, 1] .
The segment is said tostart at (h(0), v(0)) andendat (h(1), v(1)) . If h and
v are linear functions oft , then the segment islinear: it consists of a line segment. If
h andv are constant functions oft , then the segment isdegenerate: it consists of a
single point.

The segments of a path are grouped into contiguoussubpaths, which can beopen
or closed. Within a subpath,each segment starts where the previous segment ends. In
a closed subpath, the last segment ends where the first segment starts. (This may also
happen for an open subpath, but this coincidence does not make the subpath closed.)

Thecurrent pointof a path is the endpoint of the last segment of its last subpath,
assuming this subpath is open. If the path is empty or if the last subpath is closed, the
current point is undefined.

INTERFACE Path;

IMPORT Point;

TYPE T <: ROOT;

The callNEW(Path.T) creates an empty path.

PROCEDURE Reset(path: T);

Setpath to be empty.

PROCEDURE MoveTo(path: T; READONLY p: Point.T);

Extendpath with a new degenerate segment that starts and ends atp. This
begins a new subpath.

PROCEDURE LineTo(path: T; READONLY p: Point.T);

Extendpath with a linear segment that starts at its current point and ends atp.

PROCEDURE CurveTo(path: T; READONLY q, r, s: Point.T);

Extendpath with a curved segment that starts at its current point and ends ats .

CurveTo adds a curve that starts from the current point ofpath in the direction of
q, and ends ats coming from the direction ofr . More precisely, letp be the current
point ofpath and leth(t) andv(t) be the cubic polynomials such that

(h(0), v(0)) = p
(h(1), v(1)) = s
(h’(0), v’(0)) = 3 * (q - p)
(h’(1), v’(1)) = 3 * (s - r)

(where the primes denote differentiation with respect tot ). ThenCurveTo adds the
segment(h(t), v(t)) for t between zero and one. This is called theBezierarc
determined byp, q, r , ands .

PROCEDURE Close(path: T);

Add a linear segment to create a closed loop inpath .

More precisely, letp be the current point ofpath , and letq be last point ofpath that
was added by a call toMoveTo (thusq is the startpoint of the first segment of the last
subpath ofpath ). Close adds a linear segment fromp to q and marks the sequence
of segments fromq to the end of the path as a closed subpath.

PROCEDURE IsEmpty(p: T): BOOLEAN;

ReturnsTRUEif p is empty.

PROCEDURE IsClosed(p: T): BOOLEAN;

ReturnsTRUEif p is empty or the last subpath ofp is closed.

PROCEDURE CurrentPoint(p: T): Point.T;

Returns the current point ofp.

LineTo , CurveTo , Close , andCurrentPoint are checked runtime errors if the
path has no current point.

EXCEPTION Malformed;

TheMalformed exception is raised when a procedure detects a malformed path.

PROCEDURE Translate(p: T; READONLY delta: Point.T): T
RAISES {Malformed};

The result of translatingp by delta .

TYPE
MapObject = OBJECT METHODS

move(READONLY pt: Point.T);
line(READONLY pt1, pt2: Point.T);
close(READONLY pt1, pt2: Point.T);
curve(READONLY pt1, pt2, pt3, pt4: Point.T)

END;

PROCEDURE Map(path: T; map: MapObject)
RAISES {Malformed};

Apply the appropriate method ofmap to each segment ofpath .

That is, for each segments of path , in order,Mapexcecutes the following:

IF s is a linear segment(p, q) THEN
IF s was generated byMoveTo THEN

( * p = q * )
map.move(p)

ELSIF s was generated byLineTo THEN
map.line(p, q)

ELSE (* s was generated byClose * )
map.close(p, q)

END
ELSE (* s is a curved segment(p, q, r, s) * )

map.curve(p, q, r, s)
END

Map raises the exception if it is passed a malformed path.

PROCEDURE Copy(p: T): T;

Returns a newly allocated path with the same contents asp.

PROCEDURE Flatten(p: T): T RAISES {Malformed};

Return a path likep but with curved segments replaced by polygonal
approximations.

END Path.

8.7 The Trapezoid interface

A Trapezoid.T represents a set of points lying in a quadrilateral whose north and
south edges are horizontal and whose west and east edges have arbitrary non-horizontal
slopes. For example, a diagonal line can be represented as a tall skinny trapezoid.

INTERFACE Trapezoid;

IMPORT Point;

TYPE
T = RECORD

vlo, vhi: INTEGER;
m1, m2: Rational;
p1, p2: Point.T;

END;
Rational = RECORD n, d: INTEGER END;

For a trapezoidtr ,

� tr.vlo and tr.vhi are the v coordinates of its north and south edges,
respectively;

� tr.m1 and tr.m2 are the slopes of its west and east edges, respectively, as
(delta v) / (delta h) . A denominator of zero represents an infinite
slope; i.e., a vertical edge. A numerator of zero is illegal.

� tr.p1 and tr.p2 are points on the infinite lines that extend the west and east
edges, respectively.

Trapezoids are closed on the north and west edges, open on the south and east
edges, closed on the northwest corner, and open on the other corners.

A Rational q represents the rational numberq.n/q.d .

END Trapezoid.

9 Implementing your own splits

This section defines the information needed to implement newVBTclasses, especially
split classes and filter classes. Most VBT leaf classes can get by with the information
in theVBT interface.

Events that flow down the tree ofVBTs, like mouse clicks and repaint events,
are relayed via the methods described in theVBT interface. To relay the event, the
parent method recursively activates the appropriate child method. However, the parent
should not activate the child method directly; it should use one of the procedures in
this interface to activate the child method indirectly.

A typical down method of aVBT v has the formv.method(args) and has
locking levelVBT.mu or VBT.mu.v , as explained in theVBT interface.

Information also flows up the tree ofVBTs; for example, painting commands
and commands to set the cursor shape and cage. This information is also relayed
via methods, which we call “up” methods. For example, when a childch of
a parentp changes its cursor, Trestle notifies the parent by calling the method
p.setcursor(ch) . This method is expected to read the child’s cursor and take
appropriate action based on the class of the split.

A typical up method call has the formparent.method(child, args) and
has locking levelLL.sup = child .

Notice that the up methods come from the parent, not the child. This is convenient,
since it is the parent that defines the class of split. However, it means that if aVBT’s
parent isNIL , then there are no up methods, so that painting on it (for example) is a
noop. This produces a wrinkle at the address space boundary: theVBT that we call the
root of the tree actually has a parent whose only purpose is to supply up methods for
communicating across the address space boundary.

The procedures in the split interface for inserting, deleting, and enumerating
children are also implemented via methods, which we call “split” methods. For
example,Split.Succ(v, ch) is implemented by callingv.succ(ch) .

9.1 The VBTClass interface

TheVBTClass interface specifies the up methods, the split methods, and the wrapper
procedures by which a parent activates a child’s down methods.

In general, to implement a split or filter you override the down methods, up
methods, and split methods of the parent. However, usually you will be able to inherit
the majority of the methods from existing classes, and only have to override a few
of them. We mention several groups of methods that in most cases you will want to
inherit rather than reimplement.

The two down methods

VBT.Split.mouse
VBT.Split.position

together with the two up methods

VBT.Split.setcage
VBT.Split.setcursor

conspire to implement the mouse-cage semantics described in theVBT interface for
delivering mouse clicks and cursor positions and for setting the cursor shape. They
work for anyVBT.Split , and there is almost never any reason to override them. As
a far-fetched example of when you would override them, imagine a filter that converts
shifted left button clicks to right button clicks.

Although you probably won’t want to override these methods, you will have to help
them a bit. They cache the results of thelocate method, and therefore require that
you callVBTClass.LocateChanged whenever the geometry of your split changes
in a way that affects the locate method.

The up methods

VBT.Split.acquire
VBT.Split.release
VBT.Split.put
VBT.Split.forge
VBT.Split.readUp
VBT.Split.writeUp

implement the event-time semantics described in theVBT interface. They simply
recurse up the tree ofVBTs. At the root the recursive calls reach aVBT in which
these methods are overridden to make the appropriate X calls. There is rarely any
reason to override these methods. As an example of when you might want to override
them, imagine keeping track of whichVBT in your application last held the keyboard
focus. You could do this by introducing a filter whoseacquire method recorded the
information before recursing on the parent.

Keystrokes and miscellaneous codes can skip levels of the tree when they are
delivered. For example, associated with each top-level window is a filter much like
the one just described, which keeps track of which of its decendants are selection
owners. This filter forwards keystrokes and lost codes directly to the appropriate
owner, bypassing the intermediate windows in the tree.

The up methods

VBT.Split.paintbatch
VBT.Split.capture
VBT.Split.sync

implement painting, painting synchronization, and screen capture. Thesync and
capture methods recurse up the tree in the obvious way. Thepaintbatch method
also recurses up the tree, but in a less obvious way.

It would be too inefficient to call a method for every painting command; therefore
the class-independent painting code groups painting commands into batches and hands
them to the method a batch at a time. For example, thepaintbatch method of
a ZSplit clips the batch of painting commands to the visible portion of the child’s
domain and then executes the clipped operations on itself.

Painting on the vast majority ofVBTs can be implemented simply by clipping to
their domain and then relaying the painting to their parent. To speed up this common
case, everyVBT has ashort-circuitbit. If this bit is set then Trestle doesn’t call the
VBT’s paintbatch method at all; it just clips to theVBT’s domain and paints on its
parent. Typically the onlyVBTs whose short-circuit bits are not set are the rootVBT
and thoseZSplit children that are overlapped by other children or that extend outside
the parent’s domain.

If the short-circuit bits are set on all theVBTs from v to the root, then the class-
independent painting code will relay batches of painting commands fromv to the root
without activating any methods. Thepaintbatch method at the root translates the
batch of painting commands into the appropriate X operations.

The default methodVBT.Split.paintbatch sets the short-circuit bit and
recurses on the parent. In the unlikely event that you want to override this method,
the interfacesBatch , BatchUtil , and PaintPrivate define the representation
of painting commands in batches. You could for example overriding the paintbatch
method to implement a class ofVBT that paints into a raw pixmap in your address
space.

To speed up painting, Trestle does not rely on garbage collection for paintbatches:
you must free them explicitly.

You almost never need to implement the split methodssucc , pred , move, nth ,
index , and locate ; on the other hand you must be careful to inherit them from
the right place. There are two main subtypes ofVBT.Split , filters and “proper”
splits, and they have different suites of split methods. The implementations of the split
methods for filters are

Filter.T.succ

Filter.T.pred

Filter.T.move

Filter.T.nth

Filter.T.index

Filter.T.locate

These are all quite trivial procedures, since a filter has at most one child. If you declare
a split as a subtype ofFilter.T , you inherit these methods automatically.

Most proper splits are subtypes ofProperSplit.T ,which keeps the children in a
doubly-linked list. For example,ZSplits , HVSplits ,TSplits , andPackSplits
are all subtypes ofProperSplit.T . The methods

ProperSplit.T.succ

ProperSplit.T.pred

ProperSplit.T.move

ProperSplit.T.nth

ProperSplit.T.index

ProperSplit.T.locate

implement the split methods using the doubly-linked list. If you declare a split as a
subtype ofProperSplit.T , you inherit these methods automatically.

INTERFACE VBTClass;

IMPORT VBT, Trestle, Axis, Point, Rect, Region,
ScrnCursor, ScrnPixmap, Cursor, Batch;

Before we get to the up methods and the split methods, there is more to be revealed
aboutVBTs in general:

REVEAL
VBT.Prefix <: Prefix;

TYPE Prefix =
MUTEX OBJECT <* LL >= {VBT.mu, SELF} * >

parent: VBT.Split := NIL;
upRef: ROOT := NIL;
domain: Rect.T := Rect.Empty;
st: VBT.ScreenType := NIL;

METHODS <* LL.sup = SELF * >
getcursor(): ScrnCursor.T;
<* LL.sup = VBT.mu * >
axisOrder(): Axis.T;

END;

From VBT.Prefix <: Prefix it follows VBT.T <: Prefix ; hence everyVBT
is aMUTEXobject, and has the above fields and methods. The complete revelation for
the typeVBT.T is private to Trestle.

The fieldsv.parent , v.domain , and v.st record v ’s parent, domain, and
screentype.

The objectv.upRef is used by the methods ofv.parent to store information
specific to the childv . For example, ifv.parent is a ZSplit , then v.upRef
contains a region representing the visible part ofv , pointers to the children before and
afterv , and other information. In a filter,v.upRef is usuallyNIL , since when there
is only one child, all the state can be stored in data fields directly in the parent object.

If v.parent is NIL , then so isv.upRef .
The locking level comment on the data fields means that in order to write one of the

fieldsv.parent , v.upRef , v.domain , or v.st , a thread must have bothVBT.mu
and v locked. Consequently, in order to read one of the fields, a thread must have
eitherVBT.mu (or a share ofVBT.mu) or v locked. Thus the fields can be read either
by up methods or by down methods.

The callv.getcursor() returns the cursor that should be displayed overv ; that
is, the cursor that was calledGetCursor(v) in theVBT interface. It is almost never
necessary to override thegetcursor method, since leaves and splits have suitable
default methods.

TheaxisOrder method determines whether it is preferable to fix aVBT’s height
first or its width first. For example, a horizontal packsplit would rather have its width
fixed before its range of heights is queried, since its height depends on its width.
In general, ifv ’s size range in axisax affects its size range in the other axis (and
not vice-versa), thenv.axisOrder() should returnax . The default is to return
Axis.T.Hor .

Next we come to the specifications of the split methods and the up methods:

REVEAL VBT.Split <: Public;

TYPE Public = VBT.Leaf OBJECT
METHODS

( * The split methods * )

<* LL >= {VBT.mu, SELF, ch} * >
beChild(ch: VBT.T);
<* LL.sup = VBT.mu * >
replace(ch, new: VBT.T);
insert(pred, new: VBT.T);
move(pred, ch: VBT.T);
locate(READONLY pt: Point.T;

VAR (* OUT* ) r: Rect.T): VBT.T;
<* LL >= {VBT.mu} * >
succ(ch: VBT.T): VBT.T;
pred(ch: VBT.T): VBT.T;
nth(n: CARDINAL): VBT.T;
index(ch: VBT.T): CARDINAL;

( * The up methods * )

<* LL.sup = ch * >
setcage(ch: VBT.T);
setcursor(ch: VBT.T);
paintbatch(ch: VBT.T; b: Batch.T);
sync(ch: VBT.T);
capture(ch: VBT.T; READONLY rect: Rect.T;

VAR (* out * ) br: Region.T) : ScrnPixmap.T;
screenOf(ch: VBT.T; READONLY pt: Point.T)

: Trestle.ScreenOfRec;
<* LL.sup < SELF AND LL >= {ch, VBT.mu.ch} * >
newShape(ch: VBT.T);
<* LL.sup = ch * >
acquire(ch: VBT.T; w: VBT.T; s: VBT.Selection;

ts: VBT.TimeStamp) RAISES {VBT.Error};
release(ch: VBT.T; w: VBT.T; s: VBT.Selection);



   

122 9 IMPLEMENTING YOUR OWN SPLITS

put(ch: VBT.T; w: VBT.T; s: VBT.Selection;
ts: VBT.TimeStamp; type: VBT.MiscCodeType;
READONLY detail := VBT.NullDetail)
RAISES {VBT.Error};

forge(ch: VBT.T; w: VBT.T; type: VBT.MiscCodeType;
READONLY detail := VBT.NullDetail)
RAISES {VBT.Error};

<* LL.sup <= VBT.mu * >
readUp(ch: VBT.T; w: VBT.T; s: VBT.Selection;

ts: VBT.TimeStamp; tc: CARDINAL) : VBT.Value
RAISES {VBT.Error};

writeUp(ch: VBT.T; w: VBT.T; s: VBT.Selection;
ts: VBT.TimeStamp; val: VBT.Value; tc: CARDINAL)
RAISES {VBT.Error};

END;

Notice that aVBT.Split is a subtype of aVBT.Leaf . That is, everyVBT.Split
is also aVBT.Leaf , and therefore the painting operations in theVBT interface can be
applied to splits. This fact is revealed here rather than in theVBT interface to prevent
clients of VBT from accidentally painting on splits. To do so is almost certainly a
mistake—it is the responsibility of the split’s implementation to paint on the parent as
necessary to keep its screen up to date.

9.1.1 Specifications of the split methods

The first group of methods implement the behavior in theSplit interface:
The method callv.beChild(ch) initializesch.upRef as appropriate for a child

of v . The method can assume thatch is non-nil and has the same screentype asv .
When the method is called,LL >= {VBT.mu, v, ch} .

When declaring a subtypeST of a split typeS, thebeChild method forST will
ordinarily callS.beChild(v, ch) , which in turn will callS’s supertype’sbeChild
method, and so on. Only one of the methods should allocate theupRef , but all of
them may initialize different parts of it. Two rules make this work. First, the type of
theupRef for children ofST splits should be a subtype of the type of theupRef for
children ofS splits. Second, if abeChild method findsch.upRef is NIL andNIL
is not appropriate for the type, the method should allocatech.upRef ; otherwise it
should narrowch.upRef to the appropriate type and initialize it.

For example,HVSplit.T is a subtype ofProperSplit.T . Hidden in the
HVSplit module is a typeHVSplit.Child , which represents the per-child in-
formation needed by anHVSplit . The typeHVSplit.Child is a subtype of
ProperSplit.Child . The methodHVSplit.beChild(hv, ch) allocates a
newHVSplit.Child , stores it inch.upRef , initializes the part of it that is specific
to HVSplit , and then callsProperSplit.beChild(hv, ch) , which initializes

the part ofch.upRef that is common to all proper splits, and then calls its supertype’s
beChild method, and so on.

The chain of calls eventually ends with a call toVBT.Split.beChild , which
causes an error ifch is not detached or ifch ’s screentype differs fromv , and otherwise
setsch.parent to v and marksv for redisplay.

The method callv.replace(ch, new) simply implements the operation
Split.Replace(v, ch, new) , and the callv.replace(ch, NIL) implements
Split.Delete(v, ch) . Before calling the method, the generic code inSplit
marksv for redisplay, checks thatch is a child ofv and thatnew is detached, and
rescreensnew to the screentype ofv .

Similarly, the method callv.insert(pred, new) implements the operation
Split.Insert(v, pred, new) . Before calling the method, the generic code in
Split marksv for redisplay, checks thatpred is NIL or a child ofv and thatnew
is detached, and rescreensnew to the screentype ofv . A split that can only contain
a limited number of children may detach and discard the previous child to implement
insert .

The callv.move(pred, ch) implementsSplit.Move(v, pred, ch) . Be-
fore calling the method, the generic code verifies thatpred andch are children ofv
(or NIL , in the case ofpred ), and avoids the call ifpred = ch or v.succ(pred)
= ch .

When thereplace , insert , or move method is called,LL.sup = VBT.mu .
The default methods are equal toNIL ; so every split class must arrange to override
these methods, usually by inheriting them fromFilter or fromProperSplit .

The method callsv.succ(ch) , v.pred(ch) , v.nth(n) , andv.index(ch)
implement the corresponding operations in theSplit interface. In all cases,LL >=
{VBT.mu} .

The default methodVBT.Split.succ is NIL ; so every split class must ar-
range to override the method, usually by inheriting them fromFilter or from
ProperSplit . The default methodsVBT.Split.pred , VBT.Split.nth , and
VBT.Split.index are implemented by repeatedly calling thesucc method.

The method callv.locate(pt, r) returns the child ofv that controls theposition
pt , or NIL if there is no such child. The method also setsr to a rectangle containing
pt such that for all pointsq in the meet ofr anddomain(v) , v.locate(q, ...)
would return the same result asv.locate(pt, ...) . The split implementation
is expected to maker as large as possible, so that clients can avoid callinglocate
unnecessarily. When the method is called,pt will be in domain(v) . When the locate
method is called,LL.sup = VBT.mu .

If v inherits themouse, position , setcursor , or setcage methods from
VBT.Split , then you must callLocateChanged(v) whenever any operation on
the split invalidates a rectangle-child pair returned previously byv.locate :

PROCEDURE LocateChanged(v: VBT.Split);
<* LL.sup = VBT.mu * >

Clear any cached results of thelocate method.

The default methodVBT.Split.locate(v, pt, r) enumeratesv ’s children in
succ order and returns the first childch whose domain containspt . It setsr to a
maximal rectangle that lies inside the domain ofch and outside the domains of all
preceding children. If no child containspt , it returnsNIL and setsr to a maximal
rectangle that lies inside the domain ofv and outside the domains of all its children.
This is suitable if the children don’t overlap or if whenever two children overlap, the
top one appears earlier insucc order.

9.1.2 Specifications of the up methods

So much for the split methods; here now are the specifications of the up methods. In
all cases,ch is a child ofv .

The method callv.setcage(ch) is called by the system wheneverch ’s cage
is changed. It is called withLL.sup = ch . The default method implements the
behavior described in theVBT interface.

The method callv.setcursor(ch) is called by the system whenever the result
of ch.getcursor() might have changed. It is called withLL.sup = ch . The
default method implements the behavior described in theVBT interface.

The method callv.paintbatch(ch, b) is called to paint the batchb of painting
commands onv ’s child ch . The procedure can assume that the batch is not empty and
that its clipping rectangle is a subset ofch ’s domain. It is responsible for ensuring that
b is eventually freed, which can be achieved by calling passingb to Batch.Free or
by passingb to another paintbatch method, which will inherit the obligation to free the
batch. Apaintbatch method is allowed to modify the batch. The default method
clips the batch toch ’s domain, paints the batch on the parent, and setsch ’s shortcircuit
bit. The method is called withLL.sup = ch .

The method callv.sync(ch) implementsVBT.Sync(ch) . When the method
is called, ch ’s batch will have been forced. The default method simply applies
VBT.Sync to the parent. When the method is called,ch ’s batch isNIL andLL.sup
= ch .

The method callv.capture(ch, r, br) implementsVBT.Capture(ch,
r, br) . The default method recurses on the parent. When the method is called,ch ’s
batch isNIL , r is a subset ofch ’s domain, andLL.sup = ch .

Themethod callv.screenOf(ch, pt) implementsTrestle.ScreenOf(ch,
pt) . The default method recurses on the parent. When the method is called,LL.sup
= ch .

The method callv.newShape(ch) signals thatch ’s size range, preferred size, or
axis order may have changed. The default recurses on the parent. When the method is
called,LL.sup < v AND LL >= {ch, VBT.mu.ch} .

The remaining methods implement event-time operations for a descendent (not
necessarily a direct child) of the windowv . In all cases,ch is a child ofv andw is a
descendant ofch .

Theacquire , release , put , andforge methods implement the corresponding
procedures from theVBTinterface. For example,v.put(ch, w, s, ts, cd) im-

plementsVBT.Put(w, s, ts, cd.type, cd.detail) . When these methods
are called,LL.sup = ch .

Similarly, thereadUp andwriteUp methods implement theproceduresVBT.Read
andVBT.Write . When these methods are called,LL.sup <= VBT.mu .

9.1.3 Getting and setting the state of a VBT

PROCEDURE Cage(v: VBT.T): VBT.Cage; < * LL >= {v} * >

Return v’s cage.

TYPE
VBTCageType = {Gone, Everywhere, Rectangle};

PROCEDURE CageType(v: VBT.T): VBTCageType;
<* LL >= {v} * >

Returnv ’s cage’s type.

CageType(v) returns Gone if Cage(v) = VBT.GoneCage , Everywhere if
Cage(v) = VBT.EverywhereCage , andRectangle otherwise. It is more effi-
cient thanCage.

PROCEDURE GetCursor(v: VBT.T): Cursor.T;
<* LL >= {v} * >

Returncursor(v) .

PROCEDURE SetShortCircuit(v: VBT.T); < * LL >= {v} * >

Set the short-circuit property ofv .

PROCEDURE ClearShortCircuit(v: VBT.T); < * LL >= {v} * >

Clear the short-ciruit propery ofv .

If v ’s short-circuit property is on, painting onv will be implemented by clipping to its
domain and painting on its parent.

The next three procedures are equivalent to the corresponding procedures inVBT,
except they have a different locking level:

PROCEDURE PutProp(v: VBT.T; ref: REFANY);
<* LL >= {v} * >

PROCEDURE GetProp(v: VBT.T; tc: INTEGER): REFANY;
<* LL >= {v} * >

PROCEDURE RemProp(v: VBT.T; tc: INTEGER);
<* LL >= {v} * >

In implementing a split it is sometimes necessary to read a child’s bad region; in which
case the following procedure is useful:

PROCEDURE GetBadRegion(v: VBT.T): Region.T;
<* LL >= {v} * >

Return v’s bad region; that is, the join ofbad(v) andexposed(v) .

For the convenience of split implementors, everyVBThas a “newshape” bit which is
set by a call toVBT.NewShape . For example, the redisplay or shape method of a split
can test these bits to determine which of its children have new shapes.

PROCEDURE HasNewShape(v: VBT.T): BOOLEAN;
<* LL.sup < v * >

Return the value ofv ’s newshape bit.

PROCEDURE ClearNewShape(v: VBT.T); < * LL.sup < v * >

Clearv ’s newshape bit.

9.1.4 Procedures for activating the down methods of a VBT

PROCEDURE Reshape(
v: VBT.T;
READONLY new, saved: Rect.T);

<* LL.sup >= VBT.mu.v AND LL.sup <= VBT.mu * >

Prepare for and callv ’s reshape method.

That is,Reshape changesv.domain and then schedules a call to

v.reshape(VBT.ReshapeRec{v.domain, new, saved})

It should always be called instead of a direct call to the method, since it establishes
essential internal invariants before calling the method. The bits in thesaved argument
must remain valid until the method returns. It is all right forsaved to be larger than
v ’s old domain;Reshape will clip it to v ’s old domain before calling the method. It
is illegal to reshape a detachedVBT to have a non-empty domain.

For example, thereshape method ofBorderedVBT usesVBTClass.Reshape
to reshape its child.

PROCEDURE Rescreen(v: VBT.T; st: VBT.ScreenType);
<* LL.sup >= VBT.mu.v AND LL.sup <= VBT.mu * >

Prepare for and callv ’s rescreen method.

That is,Rescreen executes

prev := v.domain;
v.domain := Rect.Empty;
v.st := st;
v.rescreen(VBT.RescreenRec{prev, st}).

For example, to determine how large a menum would be if it were inserted into a
ZSplit z , you can’t simply callGetShapes(m) , since in general the screentype ofm

could be different from the screentype ofz , and the shape can depend on the screentype.
But you can callVBTClass.Rescreen(m, z.st) followed byGetShapes(m) .

PROCEDURE Repaint(v: VBT.T; READONLY badR: Region.T);
<* LL.sup >= VBT.mu.v AND LL.sup <= VBT.mu * >

JoinbadR into v ’s bad region and then prepare for and callv ’s repaint method.

PROCEDURE Position(v: VBT.T;
READONLY cd: VBT.PositionRec);

<* LL.sup = VBT.mu * >

Prepare for and callv ’s position method.

PROCEDURE Key(v: VBT.T; READONLY cd: VBT.KeyRec);
<* LL.sup = VBT.mu * >

Prepare for and callv ’s key method.

PROCEDURE Mouse(v: VBT.T; READONLY cd: VBT.MouseRec);
<* LL.sup = VBT.mu * >

Prepare for and callv ’s mouse method.

PROCEDURE Misc(v: VBT.T; READONLY cd: VBT.MiscRec);
<* LL.sup = VBT.mu * >

Prepare for and callv ’s misc method.

The following two procedures schedule calls to the down methods without making the
calls synchronously. They are useful when you hold too many locks to call a down
method directly. For example, when aZSplit child scrolls bits that are obscured,
the locking level of thepaintbatch method precludes calling therepaint method
directly; but a call can be scheduled withForceRepaint .

PROCEDURE ForceEscape(v: VBT.T); < * LL.sup >= {v} * >

Enqueue a cage escape togone for delivery tov .

PROCEDURE ForceRepaint(v: VBT.T;
READONLY rgn: Region.T; deliver := TRUE);

<* LL.sup >= {v} * >

Joinrgn into v ’s bad region, and possibly schedule a call tov ’s repaint method.

VBTClass.ForceRepaint is like VBT.ForceRepaint , except that it has a differ-
ent locking level, and ifdeliver is FALSE then no thread will be forked to deliver
the bad region—in this case the caller has the obligation to deliver the bad region soon,
either by callingForceRepaint with deliver = TRUE , or by callingRepaint .

PROCEDURE Redisplay(v: VBT.T); < * LL.sup = VBT.mu * >

If v is marked for redisplay, then unmark it and prepare for and call
v.redisplay() .

PROCEDURE GetShape(v: VBT.T; ax: Axis.T; n: CARDINAL;
clearNewShape := TRUE): VBT.SizeRange;

<* LL.sup >= VBT.mu.v AND LL.sup <= VBT.mu * >

Prepare for and callv ’s shape method.

GetShape causes a checked runtime error if the result of the shape method is invalid.
If clearNewShape is TRUE, GetShape calls ClearNewShape(v) before it calls
the method.

PROCEDURE GetShapes(v: VBT.T; clearNewShape := TRUE):
ARRAY Axis.T OF VBT.SizeRange;

<* LL.sup >= VBT.mu.v AND LL.sup <= VBT.mu * >

Return the shapes ofv in both axes.

GetShapes calls the shape method ofv in each axis, using the order determined by
v.axisOrder() , and returns the array of the resulting size ranges. IfclearNew-
Shape is TRUE, GetShapes callsClearNewShape(v) before it calls the method.

GetShapes is convenient if both the height and width preferences of the child can
be accomodated—for example, when inserting a top level window orZSplit child.

PROCEDURE Detach(v: VBT.T); < * LL.sup = VBT.mu * >

Set v.parent and v.upRef to NIL ; set v ’s domain to empty, enqueue a
reshape to empty, and clearv ’s shortcircuit bit.

9.1.5 Procedures for activating the up methods of a VBT

The following six procedures are like the corresponding procedures in theVBTinterface,
except that they have a different locking level:

PROCEDURE SetCage(v: VBT.T; READONLY cg: VBT.Cage);
<* LL.sup = v * >

PROCEDURE SetCursor(v: VBT.T; cs: Cursor.T);
<* LL.sup = v * >

PROCEDURE Acquire(
v: VBT.T;
s: VBT.Selection;
t: VBT.TimeStamp)

RAISES {VBT.Error}; < * LL.sup = v * >

PROCEDURE Release(v: VBT.T; s: VBT.Selection);
<* LL.sup = v * >

PROCEDURE Put(
v: VBT.T;
s: VBT.Selection;
t: VBT.TimeStamp;

type: VBT.MiscCodeType;
READONLY detail := VBT.NullDetail)

RAISES {VBT.Error};
<* LL.sup = v * >

PROCEDURE Forge(
v: VBT.T;
type: VBT.MiscCodeType;
READONLY detail := VBT.NullDetail)

RAISES {VBT.Error};
<* LL.sup = v * >

Finally, here is a procedure for executing a batch of painting commands on aVBT:

PROCEDURE PaintBatch(v: VBT.T; VAR b: Batch.T);
<* LL.sup < v * >

Execute the batchb of painting commands onv , freeb, and setb to NIL .

The interpretation ofb is described in theBatch andPaintPrivate interfaces. If
b.clipped is erroneously set toTRUE, thenPaintBatch may execute the batched
painting commands without clipping them tob.clip , but it will not paint outsidev ’s
domain.

END VBTClass.

9.2 The FilterClass interface

The FilterClass interface reveals the representation of a filter. If you are
implementing a subtype ofFilter.T , you can importFilterClass to gain access
to the child field.

INTERFACE FilterClass;

IMPORT Filter, Split, VBT;

REVEAL Filter.T <: Public;

TYPE Public =
Filter.Public OBJECT < * LL >= {SELF, VBT.mu} * >

ch: VBT.T
END;

A filter f is a split with the single childf.ch , or with no children iff.ch=NIL .
The beChild method initializesch and callsSplit.T.beChild . Thesucc ,

pred , nth , index , andlocate methods use thech field in the obvious way. The
misc , key , read , write , reshape , shape , andaxisOrder methods forward to
the child.

END FilterClass.

9.3 The ProperSplit interface

A ProperSplit.T is a type ofVBT.Split that contains a circularly-linked list
of its children. All of Trestle’s built-in splits that are not filters are subclasses of
ProperSplit .

INTERFACE ProperSplit;

IMPORT VBT, VBTClass, Split;

TYPE
T <: Public;
Public = VBT.Split OBJECT

<* LL >= {SELF, VBT.mu} * >
lastChild: Child := NIL

END;
Child = OBJECT

<* LL >= {SELF.ch.parent, VBT.mu} * >
pred, succ: Child := NIL;
ch: VBT.T

END;

If ch is a child of aProperSplit.T , then ch.upRef must be of typeProp-
erSplit.Child , andch.upRef.ch must equalch . The succ and pred links
represent a doubly-linked list of the children. Thesucc links are circular; thepred
links are linear. The parent’slastChild field is is NIL if there are no children;
otherwise it points to the last child insucc order.

The locking level comments imply that to write any of the links, a thread must have
bothVBT.mu and the parent locked.

If v is a T, the call v.beChild(ch) setsch.upref to NEW(Child) if it is
NIL . In any case it setsch.upref.ch := ch and callsVBT.Split.beChild(v,
ch) .

The following procedures are useful for implementing subtypes ofProperSplit.T :

PROCEDURE Insert(v: T; pred: Child; newch: VBT.T);
<* LL >= {VBT.mu, v, newch} * >

Insertnewch as a new child afterpred , and markv for redisplay.

The childnewch must be detached and of the appropriate screentype. It can beNIL to
indicate insertion at the head of the list.Insert calls thebeChild method ofnewCh.

PROCEDURE PreInsert(v: T; pred, ch: VBT.T): Child
RAISES {Split.NotAChild}; < * LL.sup = VBT.mu * >

Rescreench to havev ’s screentype (if necessary), cause a checked runtime error
if ch is attached, raiseSplit.NotAChild if pred is non-nil and not a child
of v , and finally returnpred.upRef , or NIL if pred is NIL .

PROCEDURE Move(v: T; pred, ch: Child);
<* LL >= {VBT.mu, v} * >

Movech in the list of children so that it followspred and markv for redisplay.

PROCEDURE Delete(v: T; ch: Child);
<* LL >= {VBT.mu} AND LL.sup < v * >

Removech from the list of children, detachch.ch , and markv for redisplay.

END ProperSplit.

10 Implementing your own painting procedures

10.1 The Batch interface

A Batch.T is a data structure containing a sequence ofVBT painting commands.
Batches are untraced: they must be explicitly allocated and freed using the procedures
in this interface.

INTERFACE Batch;

IMPORT Word;

TYPE T <: ADDRESS;

PROCEDURE New(len: INTEGER := -1): T;

Allocate a batch containing at leastlen Word.Ts .

If len = -1 , the number ofWord.T s in the result will beVBTTuning.BatchSize .
Initially the clip and scroll source areRect.Empty .

PROCEDURE Free(VAR ba: T);

Returnba to the free list and setba := NIL .

Free(ba) is a checked runtime error ifba is NIL .

END Batch.

10.2 The BatchUtil interface

This interface provides operations to clip and translate a batch of painting commands.
It is useful to those who are implementing window classes with customized painting
behavior.

Don’t apply these procedures to a batch whose contents are concurrently being read
or written.

INTERFACE BatchUtil;

IMPORT Batch, Rect, Point, PaintPrivate;

PROCEDURE GetLength(ba: Batch.T): CARDINAL;

Return the number ofWord.Ts in use inba .

PROCEDURE Copy(ba: Batch.T): Batch.T;

Allocate and return a new batch initialized with a copy ofba .

Every entry in a batch has a clipping rectangle; there is also a clipping rectangle for
the batch as a whole. The effective clipping rectangle for a painting operation is the
intersection of its clipping rectangle with its batch’s clipping rectangle.

PROCEDURE GetClip(ba: Batch.T): Rect.T;

Returnba ’s clipping rectangle.

TYPE ClipState = {Unclipped, Clipped, Tight};

PROCEDURE GetClipState(ba: Batch.T): ClipState;

Returnba ’s clipping state.

If GetClipState(ba) is Clipped then the clipping rectangle of every painting
operation inba is a subset ofGetClip(ba) . If GetClipState(ba) is Tight then
GetClip(ba) is equal to the join of the clipping rectangles of the painting operations
in ba . If GetClipState(ba) is Unclipped , there is no particular relationship
betweenba ’s clipping rectangle and the clipping rectangles of the entries inba .

PROCEDURE Meet(ba: Batch.T; READONLY clip: Rect.T);

Setba ’s clipping rectangle toRect.Meet(GetClip(ba), clip) .

If the assignment is non-trivial, this will change the clip state ofba to beUnclipped .

PROCEDURE Clip(ba: Batch.T);

Apply ba ’s clipping rectangle to each operation.

That is, if GetClipState(ba) is Unclipped , then for each painting operation in
ba , Clip replaces the clipping rectangle of the operation with the meet of the rectangle
andGetClip(ba) , and sets the clipstate ofba to Clipped .

PROCEDURE Tighten(ba: Batch.T);

Achieveba.clipped = Tight without changing the effect ofba .

That is, Tighten(ba) is equivalent toClip(ba) followed by assigning toba ’s
clipping rectangle the join of the resulting clipping rectangles of the entries inba .

PROCEDURE Translate(ba: Batch.T;
READONLY delta: Point.T);

Translateba by delta .

That is, for each painting operation inba , translate the target of the painting operation
by delta . This always involves translating the clipping rectangle of the operation
by delta . It also addsdelta to the delta components of all textures and to
the reference point ofTextComs . It adjusts thep1 , p2 , vlo , and vhi fields of
TrapComs . The relative displacement of a scrolling command is not affected; that is,
both the source and target of the scroll are translated bydelta . The clipping rectangle
of the batch is also translated.

PROCEDURE ByteSwap(ba: Batch.T);

Convert all text painting operations inba to have the same byteorder as
PaintPrivate.HostByteOrder .

PROCEDURE Succ(ba: Batch.T;
cptr: PaintPrivate.CommandPtr)
: PaintPrivate.CommandPtr;

Return the pointer to the entry inba that follows the one pointed to bycptr .

Succ(ba, NIL) returns the first entry inba ; Succ(ba, cptr) = NIL when
cptr is the last entry inba . To visit each entry in the batchba , use a loop like this:

cptr := BatchUtil.Succ(ba, NIL);
WHILE cptr # NIL DO

CASE cptr.command OF ... END;
cptr := BatchUtil.Succ(ba, cptr)

END

ThePaintPrivate interface explains the format of the entries.

END BatchUtil.

10.3 The PaintPrivate interface

This interface defines the layout of entries in paint batches.

INTERFACE PaintPrivate;

IMPORT Rect, Point, Trapezoid, Word;

TYPE
PaintOp = INTEGER;
Pixmap = INTEGER;
Font = INTEGER;

In a paint batch,PaintOps , Pixmaps , andFonts are represented by integers in a
screentype-dependent way. During rescreening an old batch might find its way to a
screen of the wrong type, causing garbage to be painted; but the garbage will be painted
over with the correct pixels promptly.

TYPE
PaintCommand = {RepeatCom, TintCom, TextureCom,

PixmapCom, ScrollCom, TrapCom, TextCom,
ExtensionCom};

PackedCommand = BITS 32 FOR PaintCommand;
FixedSzCommand =

[PaintCommand.RepeatCom..PaintCommand.TrapCom];
ByteOrder = {MSBFirst, LSBFirst};
PackedByteOrder = BITS 32 FOR ByteOrder;

VAR (* CONST* )
HostByteOrder: ByteOrder;



  

10.3 The PaintPrivate interface 135

There are eight types of entries; each of which begins with a word containing a
PaintCommand that indicates which type of entry it is.

Entries of typeTintCom , TextureCom , PixmapCom, ScrollCom , TrapCom,
andTextCom are used to implement theVBT operationsPaintTint , PaintTex-
ture , PaintPixmap , Scroll , PaintTrapezoid , andPaintText/PaintSub .

A RepeatCom entry in a batch indicates that thepreceding entry is to be re-executed
with its clipping rectangle changed to that of theRepeatCom entry. For example, these
are used for implementingPolyTint , PolyTexture , andPaintRegion . There
are some restrictions on whereRepeatCom entries can occur.

ExtensionCom entries can be used to implement additional painting operations
beyond those that are built into Trestle.

Some of the entries are fixed size; that is, the size of the entry is determined by
their type. The following array gives the sizes of the fixed-size commands:

CONST
WS = BYTESIZE(Word.T);
ComSize =

ARRAY FixedSzCommand OF INTEGER
{(BYTESIZE(CommandRec) + WS-1) DIV WS,

(BYTESIZE(TintRec) + WS-1) DIV WS,
(BYTESIZE(PixmapRec) + WS-1) DIV WS,
(BYTESIZE(PixmapRec) + WS-1) DIV WS,
(BYTESIZE(ScrollRec) + WS-1) DIV WS,
(BYTESIZE(TrapRec) + WS-1) DIV WS};

ComSize[c] equals the size inWord.T s of a paint batch entry for the commandc .

TYPE
CommandRec =

RECORD command: PackedCommand; clip: Rect.T END;
CommandPtr = UNTRACED REF CommandRec;
RepeatPtr = CommandPtr;

We define aRec and aPtr type for each kind of batch entry.
Every batch entry is a “pseudo-subtype” of aCommand, in the sense that its record

type hasCommandRecas a prefix.
A repeat command has no other fields besides the command identifier itself and the

clipping rectangle. Hence aRepeatPtr is simply a pointer to aCommandRec.
All of the batch entries that are not repeat commands contain aPaintOp . They

are all pseudo-subtypes of the followingRec andPtr types:

PaintRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp

END;



  

136 10 IMPLEMENTING YOUR OWN PAINTING PROCEDURES

PaintPtr = UNTRACED REF PaintRec;

The following four entry types correspond toPaintTint , PaintPixmap , Scroll ,
andPaintTrapezoid operations.

TintRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp

END;
TintPtr = UNTRACED REF TintRec;

PixmapRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp;
delta: Point.T;
pm: Pixmap

END;
PixmapPtr = UNTRACED REF PixmapRec;
TexturePtr = PixmapPtr;

ScrollRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp;
delta: Point.T;

END;
ScrollPtr = UNTRACED REF ScrollRec;

It is illegal for aScrollRec to be directly followed in a batch by aRepeat command.

TrapRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp;
delta: Point.T;
pm: Pixmap;
p1, p2: Point.T;
m1, m2: Trapezoid.Rational;

END;
TrapPtr = UNTRACED REF TrapRec;

If tr is aTrapRec , thentr.p1 and tr.p2 are points that are on the extensions of
the west and east edges of the trapezoid, andtr.m1 andtr.m2 are the slopes of the
west and east edges. The slopes are given as(delta v) / (delta h) . A zero
denominator represents an infinite slope; i.e., a vertical edge. A zero numerator is
illegal.

The entries that are not fixed-size are pseudo-subtypes ofVarSzRec , which contains
a size field with the number ofWord.T ’s in the entire entry.

VarSzRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp;
szOfRec: INTEGER;

END;
VarSzPtr = UNTRACED REF VarSzRec;

PaintText and PaintSub operations result in the following entry type, in which
commandwill equalTextCom :

TextRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp;
szOfRec: INTEGER;
byteOrder: PackedByteOrder;
clipped: BITS BITSIZE(Word.T) FOR BOOLEAN;
refpt: Point.T;
fnt: Font;
txtsz, dlsz: INTEGER;

( * dl: ARRAY [0..dlsz-1] OF VBT.Displacement * )
( * chars: ARRAY [0..txtsz-1] OF CHAR * )

END;
TextPtr = UNTRACED REF TextRec;

In a TextRec , the booleanclipped must be set ifboundingbox(text) is not
a subset of the batch’sclip . A TextRec can be directly followed in a batch by a
Repeat only if clipped isTRUE. Thedl andchars fields are declared in comments
since Modula-3 does not allow a record to contain a variable-sized array; they must
be accessed using address arithmetic. Thechars field will be padded out so that the
TextRec ends on a word boundary.

ThebyteOrder field defines the byteorder of the characters. (Since paint batches can
be transported across address spaces and merged, the byte order could be different for
different records in a paint batch.)

ExtensionRec = RECORD
command: PackedCommand;
clip: Rect.T;
op: PaintOp;
szOfRec: INTEGER;

delta: Point.T;
pm: Pixmap;
fnt: Font;
subCommand: INTEGER;

( * extensionData: ARRAY OF CHAR * )

END;
ExtensionPtr = UNTRACED REF ExtensionRec;

An ExtensionRec can beused to implement paintingoperations that exploit rendering
primitives that may be available on some particular implementation. Extension
commands get aPaintOp , a delta , a pm, and afnt “for free”; they can also put
whatever data they need into the rest of the extension data part of the record. The field
szOfRec is the number ofWord.Ts in the extension record, including the extension
data. When anExtensionRec is translated, it’sclip anddelta fields are translated
automatically; its extension data is unaffected.

PROCEDURE CommandLength(p: CommandPtr): INTEGER;

Return the length in words of the command entryp.

END PaintPrivate.

11 Miscellaneous interfaces

11.1 The VBTTuning interface

This interface defines values that can be changed to maximize Trestle’s performance
on particular systems.

INTERFACE VBTTuning;

IMPORT Word;

CONST
BatchSize: CARDINAL = 325;
BatchLatency: CARDINAL = 50000;
HVParlim: CARDINAL = 100000;
ZParlim: CARDINAL = 100000;
ResumeLength: CARDINAL = 1;
CombineLimit: CARDINAL

= (BatchSize * ADRSIZE(Word.T)) DIV 2;

The valueBatchSize is the number ofWord.T ’s in a standard painting batch.
The valueBatchLatency is the number of microseconds before a paint batch is

automatically forced.
The valuesHVParlim and ZParlim are the default minimum child areas (in

pixels) for which ZSplit and HVSplit will fork separate repaint or reshaping
threads.

ResumeLength is the size that a queue of paint batches must shrink to before a
cross-address space filter will unblock a thread that painted into an overfull queue. It
must be at least 1.

The valueCombineLimit is the number of addressable units (e.g., bytes) in a
batch beyond which Trestle will not consider combining another batch into it.

END VBTTuning.

11.2 The TrestleComm interface

INTERFACE TrestleComm;

EXCEPTION Failure;

Raised when communication to the window server fails.

END TrestleComm.

12 History and Acknowledgments

“There are lots of interesting problems in window systems”, said Butler Lampson to
Greg Nelson in April, 1984; and he was right. Nelson was enticed into the design
meetings for the new window system for the Firefly multiprocessor at SRC. In 1984
most of the discussions were about what came to be called the event-time protocol, and
besides Lampson and Nelson the main participants were Mark R. Brown, Jim Horning,
and Lyle Ramshaw. Mark Brown and Greg Nelson wrote the first version of the VBT
interface.

Mark Manasse joined SRC in 1985, and he and Nelson finished the design and
implementation of the first version of Trestle (then called Trellis), which they shipped
for use at SRC on December 31st,1985.

Trestle evolved for five years, improving under feedback from the projects that built
upon it, notably Luca Cardelli’s Dialog Editor, Mark R. Brown’s Ivy text editor, Marc
H. Brown’s FormsVBT system, Patrick Chan’s session manager Rooms, and a number
of applications built by Andrew Birrell. Bob Ayers’s Facade system spurred the Trestle
team into performance work that otherwise might never have been undertaken.

In 1990 and 1991, Steve Glassman, Mark Manasse, and Greg Nelson overhauled
Trestle to make it into the portable Modula-3 X toolkit described in this reference
manual. We are grateful to the Modula-3 export sites that used the alpha-test version
of the system released in January 1991; special thanks for the helpful feedback from
Dave Goldberg, Norman Ramsey, Jim Meehan, and Marc H. Brown. Finally, we thank
Patrick Chan, James Mason, and Jim Horning, who carefully read the entire reference
manual and made many helpful suggestions.

